


MSX



Vince Apps



VINCE APPS

MSX PELIT



Englanninkielisestä alkuteoksesta
The MSX Program Book

Suomentanut
Juha Salomäki

Kansi ja kuvat
Tadeusz Piskorski

© Amer-yhtymä Oy Amersoft

Amer-yhtymä Oy Weilin+Göösin kirjapaino
Espoo 1985

ISBN 951-35-3633-5

SISÄLLYS

Tärkeitä huomautuksia	4
1. Muukalaiset.	5
2. Kielletty kaupunki	15
3. Kolmiulotteinen sokkelo	33
4. Sähkötyt.	39
5. Faaraon kirous	45
6. Piilosana	61
7. Pilotti.	67
8. Kodin tietorekisterit.	85
9. Peikkoluolat	93
10. Pingviini.	103
11. Tilikirja.	111
12. Pikari.	121
13. Pelastus	127
14. Muukalaisrynnäkö.	133
15. Urut.	141
16. Pakoilu	143

TÄRKEITÄ HUOMAUTUKSIA

Jos olet aloittelija tietokoneasioissa ja olet tottumaton ohjelmalistauksien kirjoittamiseen, muista seurata ohjeita jokaista yksityiskohtaa myöten, sillä mitään kunnollista oikotietä ei ole olemassa.

Tietokoneesi on hyvin tarkka kieliopin ja välimerkkien suhteen eikä toimi kunnolla, jos pienikin yksityiskohta puuttuu.


Ole erittäin tarkkana numeron 0 ja kirjaimen O kohdalla ja huomaa niiden ero. Kaikki tämän kirjan ohjelmalistaukset on tulostettu suoraan MSX-tietokoneesta, joten niissä ei ole virheitä.

Jos aiot tallentaa jotkin näistä ohjelmista nauhalle välttääksesi niiden uudelleenkirjoittamisen, varmista aina, että nauhuri on hyväksynyt tallennetun tiedon, ennen kuin jatkat eteenpäin.

Huomaat, että kärsivällisyys palkitaan tässä korkean teknologian maailmassa.

Ohjelmalistauksia kirjoittaessa kannattaa huomata, että kuva-ruudussa oleva teksti ei välttämättä näytä samanlaiselta kuin kirjassa oleva ohjelmalistaus. Kirjassa rivien pituus on erilainen.

MUUKALAISET



Syvältä avaruudesta tulevat muukalaiset ovat aloittaneet hyökkäyksensä. Näiden muukalaisten olemassaolo on aina tiedetty, mutta ne eivät ole koskaan ennen tehneet mitään pahaa.

Imperiumin puolustajana sinun on tuhottava kahdenlaisia pahalaisia.

Kuinka peliä pelataan

Liiku käyttäen kohdistinta vasemmalle ja oikealle siirtäviä näppäimiä sekä tulita välilyöntinäppäimellä.

Pienet vihreät hyökkääjät ovat kymmenen pisteen arvoisia, ja paljon ilkeämmät punaiset ovat kahdenkymmenen pisteen arvoisia.

Peli vaikeutuu, kun pistesaaliisi kasvaa. Tällöin hyökkääjät alkavat tulittaa takaisin entistä nopeammin ja aina epäonnistutuksi väistössä tuhoutuu yksi kolmesta aluksestasi.

Ole varovainen liikkeissäsi ja tähtää aina huolellisesti, sillä sinulla on vain yksi laukaus kutakin muukalaista kohti.

Tietoa ohjelmasta

Rivin n:o

200–300	Määritellään spritet muukalaisia, laser-asetta ja alustasi varten.
390–450	Asetetaan alkuarvot riveille ja sarakkeille, joilla muukalaiset ja aluksesi esiintyvät.
470–650	Piirretään tausta.
660–680	Tulostetaan otsikot.
690	Käynnistää törmäyksen tunnistimen.
720–740	Liikuttaa alustasi oikealle.
750–760	Liikuttaa alustasi vasemmalle.
780–790	Testaa, onko laser-ase ampumavalmiina.
810–870	Laserin laukaisu
900–940	Muukalaisten liikkeen päätyminen
950	Päätös; ampuuko 1. muukalainen.
980–1020	1. muukalainen ampuu.
1030–1060	Liikuttaa 1. muukalaista.
1100–1160	2. muukalaisen alkuasema
1230	Päätös; ampuuko 2. muukalainen.
1260–1290	2. muukalainen ampuu.
1300–1390	Liikuttaa 2. muukalaista.
1420–1480	Pääohjelman silmukka aliohjelmien kutsumiseksi
1500–1510	Peli ohi
1580–1680	Laskee parhaat tulokset.
1700–1730	Tulostaa parhaat tulokset.
1750–1850	Kysyy uutta peliä.
1870–2050	Osuma muukalaiseen
2080–2190	Muukalaiset osuvat alukseesi.
2210–2320	Äänitehosteet

2330-2520 Räjähdyksen ääni
 2540-2670 Alkuotsikko ja ohjeet

Ohjelma: MUUK-JS

```

100 REM *****
110 REM *           *
120 REM * MUUKALAISET *
130 REM *           *
140 REM *****
150 GOSUB 2540
160 R=RND(-5)
170 SCREEN 2
180 OPEN "GRP:" AS 1
190 ON SPRITE GOSUB 1870
200 REM MÄÄRITETÄÄN SPRITET
210 FOR K=1 TO 8
220 P$=""
230 FOR J=0 TO 7
240 READ D$
250 P$=P$+CHR$(VAL("&H"+D$))
260 NEXT J
270 SPRITE$(K)=P$
280 NEXT K
290 REM SPRITE DATA
300 DATA 24,01,80,01,80,01,80,2A
310 DATA C3,FF,DB,FF,C3,BD,18,66
320 DATA 18,18,FF,FF,FF,FF,FF,FF
330 DATA 00,08,08,08,08,08,08,00
340 DATA 08,08,08,08,08,08,08,08
350 DATA 24,01,80,01,80,01,80,2A
360 DATA 3C,7E,BD,BD,FF,3C,42,81
370 DATA 3C,7E,BD,BD,FF,3C,18,18
380 REM ALKUARVOT
390 CL=8
400 OC=8
410 BS=3
420 AC=32
430 AR=10
440 BC=3
450 BR=10

```

```

460 REM TAUSTAN PIIRTÄMINEN
470 COLOR 1,1,1
480 CLS
490 REM PLANEETAT
500 CIRCLE(190,22),13,11,,,1.3
510 PAINT(190,22),11
520 CIRCLE(190,22),16,6,5.5,3.85,.1
530 CIRCLE(190,23),16,13,5.5,3.85,.1
540 REM TÄHDET
550 FOR J=1 TO 15
560 R1=INT(RND(1)*240)+10
570 R2=INT(RND(1)*90)+10
580 CIRCLE(R1,R2),1,15,,,1.3
590 NEXT J
600 CIRCLE(50,70),16,9,,,1.3
610 PAINT(50,70),9
620 PRESET(0,120),5
630 REM VUORET
640 DRAW "E8F16E12F20R2E30R3F22R2E17F
40R10E33R9F32"
650 PAINT(1,161),5
660 PRESET(30,0)
670 COLOR 15,1
680 PRINT#1,"ALUKSIA: 3      TULOS: 0"
690 SPRITE ON
700 GOTO 1420
710 REM ODOTTAA LIIKE- TAI AMPUMAKASKYÄ
NÄPPÄIMISTÖLTÄ
720 K$=INKEY$
730 IF K$<>CHR$(28) OR CL>27 THEN 750
740 CL=CL+1
750 IF K$<>CHR$(29) OR CL<8 THEN 770
760 CL=CL-1
770 PUT SPRITE 3,(CL*8,180),6,3
780 IF K$<>CHR$(32) THEN 880
790 IF LS=0 THEN 880
800 REM AMPUMINENR
810 LS=LS-1
820 FZ=1
830 GOSUB 2220
840 FOR J=169 TO 1 STEP -7
850 PUT SPRITE 5,(CL*8,J),15,5
860 NEXT J
870 PUT SPRITE 5,(0,209),,5
880 RETURN

```



```
890 REM MUUKALAISTEN LIIKE PÄÄTTY
900 IF BC>3 THEN 950
910 AA=0
920 BC=32+INT(RND(1)*2)
930 PUT SPRITE 2,(0,209),,2
940 RETURN
950 IF RND(1)>.2+SC/150 THEN 1030
960 IF BC<7 OR BC>28 THEN 1030
970 REM MUUKALAISET AMPUVAT
980 FZ=2
990 FOR J=BR*8+8 TO 190 STEP 8
1000 PUT SPRITE 4,(BC*8,J),11,4
1010 NEXT J
1020 PUT SPRITE 4,(0,209),,4
1030 BC=BC-2
1040 SOUND 7,56
1050 PLAY "04L60A"
1060 PUT SPRITE 2,(BC*8,BR*8),6,2
1070 RETURN
1080 IF AC<>32 THEN 1230
1090 REM MUUKALAISTEN LIIKE
1100 IF RND(1)>.3 THEN 1180
1110 PUT SPRITE 7,(0,209),,7
1120 PUT SPRITE 8,(0,209),,8
1130 AA=1
1140 BR=INT(RND(1)*18)+2
1150 BC=31+INT(RND(1)*2)
1160 LS=1
1170 RETURN
1180 PUT SPRITE 7,(0,209),,7
1190 PUT SPRITE 8,(0,209),,8
1200 AR=INT(RND(1)*18)+2
1210 LS=1
1220 AC=1
1230 IF RND(1)>.01+SC/250 THEN 1300
1240 FZ=2
1250 GOSUB 2220
1260 FOR J=AR*8+8 TO 190 STEP 8
1270 PUT SPRITE 5,(AC*8,J),11,5
1280 NEXT J
1290 PUT SPRITE 5,(0,209),,5
1300 AC=AC+1
1310 IF LF/2=INT(LF/2) THEN 1370
1320 SOUND 7,56
1330 PLAY "02L40D"
```

```

1340 PUT SPRITE 7,(0,209),,7
1350 PUT SPRITE 8,(AC*8,AR*8),12,8
1360 GOTO 1390
1370 PUT SPRITE 8,(0,209),,8
1380 PUT SPRITE 7,(AC*8,AR*8),12,7
1390 LF=LF+1
1400 RETURN
1410 REM ALIOHJELMIEN KUTSUMINEN
1420 GOSUB 720
1430 IF AR<>1 THEN 1460
1440 GOSUB 900
1450 GOTO 1470
1460 GOSUB 1080
1470 GOSUB 720
1480 GOTO 1420
1490 REM PELIN LOPPUMINEN
1500 SOUND 7,56
1510 PLAY "O5BAFEDCO4BAFEDCO3BAFEDC"
1520 PRESET(50,90)
1530 COLOR 3
1540 PRINT#1," P E L I O H I"
1550 FOR DE=1 TO 2500:NEXT DE
1560 SCREEN 0
1570 COLOR 15,1
1580 PRINT "SAIT";SC;"PISTETTÄ":PRINT
1590 INPUT "MIKÄ ON NIMESI";NM$
1600 IF LEN(NM$)>15 THEN PRINT "NIMI ONLIIAN
    PITKÄ!":PRINT:GOTO 1590
1610 REM PARHAAT TULOKSET TAULUKKONA
1620 S(10)=SC
1630 N$(10)=NM$
1640 FOR J=10 TO 2 STEP-1
1650 IF S(J)<S(J-1) THEN 1680
1660 SWAP S(J),S(J-1)
1670 SWAP N$(J),N$(J-1)
1680 NEXT J
1690 CLS
1700 PRINT TAB(8);"PARAS TULOS":PRINT:PRINT
1710 FOR M=1 TO 9
1720 PRINT " ";M;" ";S(M);" ";N$(M)
1730 NEXT M
1740 LOCATE 2,19
1750 PRINT "UUSI PELI? PAINA 'P'"
1760 PRINT " LOPETUS? PAINA 'L'"

```

```
1770 PRINT
1780 K#=INKEY$:IF K#="" THEN 1780
1790 PLAY "05C"
1800 IF K#<>"P" THEN 1840
1810 SC=0
1820 SCREEN 2
1830 GOTO 390
1840 IF K#<>"L" THEN 1780
1850 GOTO 2680
1860 REM SPRITE-TÖRMÄYS
1870 SPRITE OFF
1880 IF FZ=2 THEN 2060
1890 GOSUB 2330
1900 IF CL=AC THEN SC=SC+10 ELSE SC=SC+20
1910 PUT SPRITE 6,(CL*8,J+8),6,6
1920 PUT SPRITE 5,(0,209),,5
1930 PUT SPRITE 7,(0,209),,7
1940 PUT SPRITE 8,(0,209),,8
1950 PUT SPRITE 2,(0,209),,2
1960 PRESET(190,0)
1970 COLOR 1
1980 PRINT#1,STRING$(5,219)
1990 COLOR 15
2000 PRESET(190,0)
2010 PRINT#1,SC
2020 PUT SPRITE 6,(0,209),,6
2030 J=10
2040 AC=32:BC=3
2050 GOTO 2190
2060 BS=BS-1
2070 GOSUB 2330
2080 PUT SPRITE 6,(CL*8,J-8),6,6
2090 IF BS=0 THEN 1500
2100 CL=8
2110 PRESET(90,0)
2120 COLOR 1
2130 PRINT#1,STRING$(4,219)
2140 COLOR 15
2150 PRESET(90,0)
2160 PRINT#1,BS
2170 PUT SPRITE 6,(0,209),,6
2180 J=190
2190 SPRITE ON
2200 RETURN
```



```

2210 REM ÄÄNITEHOSTEET
2220 SOUND 6,15
2230 SOUND 7,0
2240 SOUND 8,16
2250 SOUND 9,16
2260 SOUND 10,16
2270 SOUND 11,0
2280 SOUND 12,16
2290 SOUND 13,0
2300 SOUND 7,56
2310 FOR DE=1 TO 30:NEXT DE
2320 RETURN
2330 SOUND 0,0
2340 SOUND 1,5
2350 SOUND 2,0
2360 SOUND 3,13
2370 SOUND 4,255
2380 SOUND 5,15
2390 SOUND 6,30
2400 SOUND 7,0
2410 SOUND 8,16
2420 SOUND 9,16
2430 SOUND 10,16
2440 SOUND 11,0
2450 SOUND 12,5
2460 SOUND 13,0
2470 FOR DE=1 TO 30:NEXT DE
2480 SOUND 12,56
2490 SOUND 13,0
2500 SOUND 6,0
2510 FOR DE=1 TO 30:NEXT DE
2520 RETURN
2530 REM OTSIKKO
2540 COLOR 15,1
2550 CLS
2560 LOCATE 10,5
2570 PRINT "M U U K A L A I S E T"
2580 LOCATE 0,8
2590 PRINT "Aluksesi liikkuu oikealle ja
vasemmallekohdistinnäppäimistä":PRINT
2600 PRINT "Ämpuminen välilyöntinäppäimellä"
:PRINT
2610 PRINT "Tahtaa hyvin, sinulla on vain
" yksi lau-kaus yhtä muukalaista kohti

```

```
2620 LOCATE 0,18
2630 PRINT "Peli alkaa, kun painat jotain
näppäintä"
2640 K$=INKEY$
2650 IF K$="" THEN 2640
2660 PLAY "05L30C"
2670 RETURN
2680 END
```

1. The first part of the paper is devoted to the study of the properties of the function $f(x)$ defined by the equation $f(x) = \int_0^x f(t) dt$. It is shown that $f(x)$ is a constant function.

2. In the second part, we consider the function $g(x)$ defined by the equation $g(x) = \int_0^x g(t) dt$. It is shown that $g(x)$ is a constant function.

3. The third part of the paper is devoted to the study of the properties of the function $h(x)$ defined by the equation $h(x) = \int_0^x h(t) dt$. It is shown that $h(x)$ is a constant function.

4. In the fourth part, we consider the function $k(x)$ defined by the equation $k(x) = \int_0^x k(t) dt$. It is shown that $k(x)$ is a constant function.

5. The fifth part of the paper is devoted to the study of the properties of the function $l(x)$ defined by the equation $l(x) = \int_0^x l(t) dt$. It is shown that $l(x)$ is a constant function.

6. In the sixth part, we consider the function $m(x)$ defined by the equation $m(x) = \int_0^x m(t) dt$. It is shown that $m(x)$ is a constant function.

7. The seventh part of the paper is devoted to the study of the properties of the function $n(x)$ defined by the equation $n(x) = \int_0^x n(t) dt$. It is shown that $n(x)$ is a constant function.

8. In the eighth part, we consider the function $o(x)$ defined by the equation $o(x) = \int_0^x o(t) dt$. It is shown that $o(x)$ is a constant function.

9. The ninth part of the paper is devoted to the study of the properties of the function $p(x)$ defined by the equation $p(x) = \int_0^x p(t) dt$. It is shown that $p(x)$ is a constant function.

10. In the tenth part, we consider the function $q(x)$ defined by the equation $q(x) = \int_0^x q(t) dt$. It is shown that $q(x)$ is a constant function.

KIELLETTY KAUPUNKI



Kerrotaan tarinaa hylätystä, kaukaisella planeetalla sijaitsevasta kaupungista, jota voimakentät, aistiharhoja aiheuttavat kaasut ja vieraat elämänmuodot vartioivat, sillä kaupungissa on kätkeytyneenä tarunomainen aarre. Onko sinulla rohkeutta lähteä etsimään aarretta?

Kaupunki on aivan suuren aarniometsän laidassa, lähellä kimmeltäviä järviä. Pahaa-aavistamattoman vieraan valittavana saattaa olla hopealusikoita, sinisiä juomia, metallikiekkoja ja mahdollisesti jopa kuolema. Huomaat kuitenkin, että kaiken käydessä oikein vaikeaksi voit aina pysähtyä vaikkapa juomaan vähän limonaadia.

Kuinka peliä pelataan

Anna syöttötietoina verbi ja sen jälkeen jokin substantiivi, esimerkiksi OTA LUSIKKA, OTA LAATIKKO, OTA MUISTI-VIHKO jne. Ilmansuuntia voidaan merkitä niiden alkukirjaimilla eli p:llä, e:llä, i:llä ja l:llä.

Kaikki syöttötiedot on annettava pienaakkosin.

Seikkailupelin kirjoittaminen on hyvin aikaavievää puuhaa, sillä kirjoittaessa täytyy olla huolellinen, varsinkin DATA-lausekkeissa, koska muutoin ohjelman looginen rakenne kärsii ja peli ei toimi oikein. Tästä syystä onkin suositeltavaa, että et kirjoita ohjelmaa sisään yhdellä kertaa, vaan kirjoitat vain sen verran kuin hyvin jaksat ja tallennat ohjelman nauhalle vähän kerrallaan. Seuraavalla kerralla lataat listauksen taas tietokoneeseen ja jatkat ohjelman kirjoittamista, kunnes saat sen valmiiksi.

Pelissä on myös *tallenna*-toiminto, jolla voit tallentaa pelitilanteesi nauhalle, ellet voi pelata kerralla loppuun asti. Vastaa-vasti voit myöhemmin jatkaa jo saavuttamastasi tilanteesta käyttämällä *lataa*-käskyä.

Tietoa ohjelmasta

Rivin n:o

200	Tilavaraukset taulukoille, joissa ovat mahdolliset suunnat, esineiden sijainnit ja kuvaukset sekä huoneiden kuvaukset.
230-360	Suunnat, esineiden sijainnit sekä kuvaukset luetaan DATA-lausekkeista.
400-640	Kirjoittaa peliohjeet.
690	Tyhjentää näytön yläosan.
710	Sijaintipaikan kuvaus
730-760	Tutkii, onko tässä paikassa esineitä.
780-820	Ilmoittaa näkyvillä olevat esineet.
840-910	Ilmoittaa mahdolliset suunnat.
930	Piirtää viivan kuvauksen alle.
980-1000	Seuraavan ohjeen syöttö
1010-1070	Erikoistapauksia
1090	Etsii ohjeessa olevan välilyönnin.
1110-1150	Erottelee verbin ja substantiivin sekä antaa niille numeroarvot.

1160-1180	Tunnistamaton verbi tai substantiivi
1200	Hyppää oikeaan kohtaan verbiä varten.
1220-1270	Verbi Paina
1290-1400	Juo
1420-1640	Laita
1660-1830	Lue
1850-2090	Vedä
2130-2260	Käytä
2270-2340	Pudota
2360-2540	Mene, liiku
2560-2740	Ota
2760-3100	Tutki
3120-3380	Avaa
3390-3400	Verbien ja substantiivien DATA-tiedot
3410-4160	Suuntien ja huoneiden DATA-tiedot
4170-4390	Autaa
4400-4420	Viive
4460-4660	Tallenna pelitilanne nauhalle.
4680-4880	Lataa vanha peli nauhalta.
4890-4970	Epäonnistuit.
4980-5020	Onnistuit.

Ohjelma: KIEL-JS

```

100 REM *****
110 REM *
120 REM * KIELLETTY KAUPUNKI *
130 REM *
140 REM *****
150 CLS
160 CLEAR 2000
170 COLOR 1,15,4
180 LOCATE 8,10
190 PRINT "KIELLETTY KAUPUNKI"
200 DIM M(60,4),OB(30),OJ$(30),R$(60),
DR$(4),OL(30)
210 GOTO 3390
220 REM LUKEE ALKUARVOT JA ESINEIDE
N KUVAUKSET
230 READ V$,N$,RN,NN,CP
240 FOR J=1 TO RN
250 FOR K=1 TO 4
260 READ M(J,K)

```

```

270 NEXT K,J
280 FOR J=1 TO NN
290 READ OL(J)
300 NEXT J
310 FOR J=1 TO NN
320 READ OJ$(J)
330 NEXT J
340 FOR J=1 TO 4
350 READ DR$(J)
360 NEXT J
370 REM TYHJENTÄÄ NÄYTÖN JA KIRJOITTA
A      OHJEET
380 CLS
390 LOCATE 0,6
400 PRINT "Kaukaisella planeetalla on vanha,
hy-   lätty kaupunki, jonne on tarun
mukaan kätkeytyä valtava jalokiviasarre."
410 PRINT:PRINT
420 PRINT "Monet ovat yrittäneet löytää
tämän aar-teen, mutta yksikään heistä ei
ole pa- lannut takaisin."
430 PRINT:PRINT:PRINT
440 PRINT "Paina jotain näppäintä..."
450 NO$=INKEY$
460 IF NO$="" THEN 450
470 CLS
480 LOCATE 0,4
490 PRINT "Voimme yhdessä yrittää löytää
tämän   aarteen. Kirjoita kaksi sanaa
ohjeeksi,esim.'mene länteen, tutki seinää,
käytäväävainta, paina nappia."
500 PRINT:PRINT
510 PRINT "On vielä monia muita ohjeita,
joita   voit käyttää, mutta en halua p
ilata   iloasi kertomalla niitä sinulle!"
520 PRINT:PRINT
530 PRINT "Äärimmäisessä hätässä voit
käyttää   sanaa 'auta'."
540 PRINT:PRINT
550 PRINT "Paina jotain näppäintä..."
560 NO$=INKEY$
570 IF NO$="" THEN 560
580 CLS
590 LOCATE 0,6
600 PRINT "Pelitilanne voidaan myös

```

```

tallentaa      muistiin. Tällöin käskysanaksi
riittää 'tallenna'. Kirjoita sana 'lataa'
halu-tessasi jatkaa vanhaa peliä."
610 PRINT:PRINT:PRINT
620 PRINT "Peli alkaa. Paina jotain
näppäintä..."
630 NO$=INKEY$
640 IF NO$="" THEN 630
650 CLS
660 GOTO 970
670 REM PAIKAN KUVAUS, NÄKYVILLÄ OLEVA
T      ESINEET JA MAHDOLLISET SUUNNAT
680 LOCATE 0,0
690 PRINT SPACE$(255);SPACE$(255)
700 LOCATE 0,2
710 PRINT "Olet ";R$(CP)
720 REM TARKISTAA ONKO TÄSSÄ
PAIKASSA      ESINEITÄ
730 FOR J=1 TO NN
740 IF OL(J)=CP THEN 780
750 NEXT J
760 GOTO 840
770 REM KIRJOITTAA PAIKASSA
OLEVAT      ESINEET
780 PRINT:PRINT "NÄKYVILLÄ: ";
790 FOR J=1 TO NN
800 IF OL(J)<>CP THEN 820
810 PRINT OJ$(J)
820 NEXT J
830 REM MAHDOLLISET ILMANSUUNNAT
840 PRINT:PRINT "SUUNNAT: ";
850 FOR J=1 TO 4
860 IF M(CP,J)=0 OR M(CP,J)=88 THEN 890
870 PRINT DR$(J);
880 DF=1
890 NEXT J
900 IF DF=0 THEN PRINT "??"
910 DF=0
920 PRINT:PRINT
930 PRINT STRING$(37,219)
940 IF CP=46 OR CP=25 OR CP=14 OR CP=5 OR
CP=10 THEN 4430
950 RETURN
960 PRINT
970 GOSUB 670

```



```

980 LOCATE 0,23:PRINT
990 INPUT "Anna ohjeet";IP$
1000 PRINT
1010 IF IP$="auta" THEN 4170
1020 IF IP$="tallenna" THEN 4450
1030 IF IP$="lataa" THEN 4670
1040 IF IP$="p" THEN IP$="men poh"
1050 IF IP$="e" THEN IP$="men ete"
1060 IF IP$="i" THEN IP$="men itä"
1070 IF IP$="l" THEN IP$="men län"
1080 REM ETSII VALILYÖNNIN SYÖTETY
STA TIEDOSTA
1090 X=INSTR(1,IP$," ")
1100 REM EROTTAA VERBIN
1110 VB$=LEFT$(IP$,3)
1120 NT$=MID$(IP$,X+1,LEN(IP$)-X)
1130 REM EROTTAA SUBSTANTIIVIN
1140 NO$=MID$(IP$,X+1,3)
1150 VB=(INSTR(1,VB$,VB$)+2)/3
1160 IF VB<1 THEN PRINT "En ymmärrä
verbiä.":GOTO 970
1170 NO=(INSTR(1,N$,NO$)+2)/3
1180 IF NO<1 OR (INT(NO)<>NO) THEN PRINT
"Tuntematon sana verbin jälkeen.":GOTO
970
1190 REM VALITSEE VERBIA VASTAAVA
N OHJELMAKOHDAN
1200 ON VB GOTO 1290,2750,2350,1660,1850,
1220,2100,2560,2560,2250,2250,2750,3110
,2350,1850,1420,1220
1210 REM PAINA
1220 IF NO=24 AND OB(1)=1 AND CP=44 THEN
ELSE 1270
1230 PRINT "Voimakenttä katosi!"
1240 M(44,1)=11
1250 R$(44)="läpikulikutunnelissa."
1260 GOTO 970
1270 GOTO 2080
1280 REM JUO
1290 IF OB(7)=1 AND NO=21 THEN 1300 ELSE
1390
1300 PRINT "Jotain todella outoa tapahtuu!"
1310 GOSUB 4400
1320 PRINT "Tuntuu, että tulen painavamma
ksi!"
1330 GOSUB 4400

```

```

1340 PRINT "Lattia liikahti!"
1350 GOSUB 4400
1360 PRINT "OOH! Lattia tuntuu sortuvan!"
1370 GOSUB 4400:GOSUB 4400
1380 GOTO 4890
1390 PRINT "Olipa outo ehdotus! En usko,
      että voin tehdä niin."
1400 GOTO 970
1410 REM LAITA
1420 IF NO=2 AND CP=13 AND OB(2)=1 THEN
1430 ELSE 1570
1430 PRINT "Kuuluu kummallinen huriseva
      ääni!"
1440 PRINT
1450 GOSUB 4400
1460 PRINT "Voisiko, tämä olla lopun alkua?"
1470 PRINT
1480 GOSUB 4400
1490 PRINT "Jotain tulee ulos koneesta!"
1500 PRINT
1510 GOSUB 4400
1520 PRINT "Se on ohut metallitanko;
      suunnilleen avaimen kokoinen."
1530 GOSUB 4400
1540 OB(2)=0
1550 OL(3)=13
1560 GOTO 970
1570 IF NO=3 AND CP=39 AND OB(3)=1 THEN 1580
ELSE 1620
1580 PRINT "Salaseinä siirtyi sivuun ja
      paljasti kapean oviaukon!"
1590 M(39,1)=16
1600 R$(39)="oviaukossa."
1610 GOTO 970
1620 IF NO=5 AND CP=2 AND OB(5)=1 THEN 2140
1630 PRINT "Mitaan ei tapahdu. Yritä jotain
      eri- laista."
1640 GOTO 970
1650 REM LUE, TUTKI
1660 IF (NO=8 OR NO=19) AND (OB(8)=1 OR
      OL(8)=CP) THEN 1670 ELSE 1690
1670 PRINT "Siinä lukee: aptelo änis
      ömläh. En ymmärrä alkuunkaan. Ehkä
      se on muu-kalaisten kieltä."
1680 GOTO 970

```

```

1690 IF (NO=10 OR NO=19) AND (OB(10)=1 OR
    OL(10)=CP) THEN 1700 ELSE 1720
1700 PRINT "Siinä lukee: 211154 2511419.
    Arvaan, että tuo on jonkun puhelinnum
    ero."
1710 GOTO 970
1720 IF NO=19 AND CP=16 THEN 1730 ELSE 1780
1730 PRINT "Siinä lukee: Varo sumua! Se
    sekoittaa aistit."
1740 PRINT
1750 GOSUB 4400
1760 PRINT "Huh, huh, onpa täällä kuum
    a.          Ethän pahastu, jos riisun tak
    kini?      Kiitos!"
1770 GOTO 970
1780 IF (NO=11 OR NO=19) AND (OB(11)=1 OR
    OL(11)=CP) THEN 1790 ELSE 1810
1790 PRINT "Siinä lukee: sapnilo änim älep
    mäk.      Olisikohan se jotain salakiel
    tä?"
1800 GOTO 970
1810 IF NO=15 AND CP=36 THEN 4980
1820 PRINT "Minulla ei ole ";NT$
1830 GOTO 970
1840 REM UEDA
1850 IF NO=25 AND CP=21 THEN 1860 ELSE 1940
1860 INPUT "Mikä niistä";Q
1870 IF Q=2 OR Q=3 THEN CP=16:GOTO 970
1880 IF Q=4 THEN CP=24:GOTO 970
1890 CP=46
1900 GOTO 970
1910 PRINT "Pieni muistikirja ilmestyi kuin
    tyhjas-tä! Veljet, tämäpä on mielenk
    iintoinen!"
1920 OL(11)=28
1930 GOTO 970
1940 IF NO=25 AND CP=47 THEN 1950 ELSE 1970
1950 CP=42
1960 GOTO 970
1970 IF NO=25 THEN 1980 ELSE 2080
1980 INPUT "Mikä niistä";Q
1990 IF Q=3 AND CP=42 THEN 2000 ELSE 2020
2000 CP=47
2010 GOTO 970
2020 IF (Q=3 OR Q=2) AND CP=28 THEN 1910

```

```

2030 IF Q=4 AND CP=28 THEN 2040 ELSE 2060
2040 CP=29
2050 GOTO 970
2060 CP=46
2070 GOTO 970
2080 PRINT "Mitään ei tapahtunut."
2090 GOTO 970
2100 IF NO=2 AND CP=13 THEN 2110 ELSE 2130
2110 VB=16
2120 GOTO 1200
2130 IF NO=5 AND CP=2 AND OB(5)=1 THEN 2140
ELSE 2230
2140 PRINT "Portti ei ole enää lukoss
a!          Odota hetki. Avaan sen..."
2150 PRINT
2160 GOSUB 4400
2170 PRINT "Saranat ovat vähän
ruosteessa...":PRINT
2180 GOSUB 4400
2190 PRINT "No niin, se on auki."
2200 M(2,1)=3
2210 R$(2)="kaupungin portin ulkopuolella."
2220 GOTO 970
2230 PRINT "En voi tehdä sitä!"
2240 GOTO 970
2250 IF NO=3 THEN 2260 ELSE 2270
2260 M(40,1)=24
2270 IF OB(NO)<>1 THEN 2330
2280 OB(NO)=0
2290 OL(NO)=CP
2300 PRINT "Olen pudottanut sen."
2310 IF RND(8)>.9 THEN PRINT "Huuts!
Jalkani jäi alle!"
2320 GOTO 970
2330 PRINT NT$;" ei ole minulla!"
2340 GOTO 970
2350 REM LIIKU, MENE JNE
2360 IF CP=32 AND NO=31 THEN 2370 ELSE 2400
2370 IF M(32,4)=99 THEN 2380 ELSE 2420
2380 CP=16
2390 M(32,4)=35
2400 IF NO=17 AND CP=50 THEN 2410 ELSE
2420
2410 NO=30

```

```

2420 IF CP=32 AND OB(6)=1 AND NO=31 THEN
2430 ELSE 2440
2430 M(35,4)=36
2440 REM KIELLETTY SUUNTA
2450 IF NO<28 OR NO>31 THEN 2470
2460 IF M(CP,NO-27)<>0 THEN 2490
2470 PRINT "En voi mennä siihen suuntaan!"
2480 GOTO 970
2490 IF M(CP,NO-27)<>88 THEN 2530
2500 PRINT "En voi mennä sinne vielä!"
2510 GOTO 970
2520 REM SALLITTU SUUNTA
2530 CP=M(CP,NO-27)
2540 GOTO 970
2550 REM TARKISTA ESINEET
2560 IF NO=27 THEN 2570 ELSE 2620
2570 FOR J=1 TO NM
2580 IF OB(J)<>1 THEN 2600
2590 PRINT OJ$(J)
2600 NEXT J
2610 GOTO 970
2620 IF NO=17 THEN PRINT "Huh! Yrität
nostaa sitä!":GOTO 970
2630 IF NO=23 THEN PRINT "En voi ottaa sitä,
se on ruuvattu sei- nään!":GOTO 970
2640 IF NO=26 THEN PRINT "Jos luulet minun
kantavan tuota homeis-ta rohjaketta y
mpäriinsä, saat etsiä itsellesi uuden
kaverin.":GOTO 970
2650 IF NO>13 THEN PRINT "Et voi ottaa
sitä!":GOTO 970
2660 FOR J=1 TO 15
2670 IF OB(J)=1 THEN CN=CN+1
2680 NEXT J
2690 IF CN>4 THEN PRINT "En voi kantaa
enempää!":CN=0:GOTO 970
2700 CN=0
2710 IF NO=3 THEN 2720 ELSE 2730
2720 M(40,1)=88
2730 IF OL(NO)=CP THEN OL(NO)=0:OB(NO)=1:
PRINT "Se on sinun." ELSE PRINT NT$;" e
i ole täällä!":GOTO 970
2740 GOTO 970
2750 REM TUTKI
2760 IF (NO=8 AND (OB(8)=1 OR OL(8)=CP)) OR

```

```

(NO=11 AND (OB(11)=1 OR OL(11)=CP))
THEN 2770 ELSE 2860
2770 IF CP<>26 THEN 2800
2780 PRINT "Hmm... Tuo valo tuntuu
    pahalta.          Mitähän tämäkin seikkailu si-
euttaa      silmäparoilleni..."
2790 PRINT
2800 IF CP<>28 THEN 2840
2810 PRINT "Hmm... Valo ei parane yhtään..."
2820 PRINT
2830 GOSUB 4400
2840 PRINT "Ahaa, siinä tuntuu olevan
    jotakin kir- joitusta."
2850 GOTO 970
2860 IF NO=9 AND (OB(9)=1 OR OL(9)=CP) THEN
2870 ELSE 2890
2870 PRINT "Siinä on useita pieniä valoja,
    jotka alkavat vilkkua - tämä taitaa
    olla vaa-rallista. Toivon totisesti, että
    tie- dät mitä teet!"
2880 GOTO 970
2890 IF NO=10 AND (OB(10)=1 OR OL(10)=CP)
THEN 2810
2900 IF NO=22 AND CP=26 THEN 2910 ELSE 2940
2910 PRINT "Pöydällä on pieni muovikuutio,
    joka näyttää tarpeeksi turvalliset
    ta."
2920 OL(8)=26
2930 GOTO 970
2940 IF NO=26 AND CP=13 THEN 2950 ELSE 2970
2950 PRINT "Se on ihan ruosteessa... Ahaa,
    yläosas-sa näkyy pieni rako ja alaosa
    ssa on sulkulaite."
2960 GOTO 970
2970 IF NO=18 AND CP=39 THEN 2980 ELSE 3030
2980 PRINT "Se on ihan pölyinen... Odota
    hetki, pyyhin vähän likaa pois."
2990 PRINT
3000 GOSUB 4400:GOSUB 4400
3010 PRINT "Löysin ainoastaan pienen reiän
    seinäs- tä. Se on suunnilleen vyötär
    ön korkeu-della."
3020 GOTO 970
3030 IF NO=16 AND CP=43 THEN 3040 ELSE 3080
3040 IF OB(9)=1 OR OL(9)<>0 THEN 3070

```

```

3050 PRINT "Katso!"
3060 OL(9)=43
3070 GOTO 970
3080 REM EI MITÄÄN ERIKOISTA
3090 PRINT "En näe mitään mielenkiintoista."
3100 GOTO 970
3110 REM OPEN
3120 IF CP=12 AND NO=23 THEN 3130 ELSE 3190
3130 PRINT "Katso!"
3140 IF OL(2)<>0 OR OB(2)=1 THEN 3160
3150 OL(2)=12
3160 IF OL(4)<>0 OR OB(4)=1 THEN 3180
3170 OL(4)=12
3180 GOTO 970
3190 IF CP=34 AND NO=23 THEN 3200 ELSE 3240
3200 PRINT "Katso!"
3210 IF OL(10)<>0 OR OB(10)=1 THEN 3230
3220 OL(10)=34
3230 GOTO 970
3240 IF CP=33 AND NO=20 THEN 3250 ELSE 3350
3250 PRINT "Varo! Se on ansa; nopeasti -"
3260 PRINT:PRINT "      TEE JOTAKIN !!":PRINT:
PRINT
3270 FOR J=1 TO 1000
3280 Q$=INKEY$
3290 IF Q$<>" " THEN J=1000
3300 NEXT J
3310 IF Q$<>" " THEN PRINT "Phuh, juuri
ajoissa!":GOTO 970
3320 PRINT "No nyt se räjähti..."
3330 GOSUB 4400
3340 GOTO 4890
3350 PRINT "En voi tehdä sitä!"
3360 REM EI MITÄÄN ERIKOISTA
3370 PRINT "Täällä ei ole mitään."
3380 GOTO 970
3390 DATA juotutmenluevedtyökäyotahanpanpu
dkatavakävliilaipai
3400 DATA laakietanlusavaamupulkuukypsäimui
nulunulmerhylpakseikirlipnespöykaan
apvipkonesipoheteitälän
3410 DATA 51,26,2
3420 R$(1)="käytävässä."
3430 DATA 12,11,13,14

```



```

3440 R$(2)="kaupungin portin ulkopuolell
a. Portti on painava ja tukevasti luk
ittu."
3450 DATA 88,0,0,0
3460 R$(3)="lyhyessä käytävässä. Mäntymis
en löyhkä täyttää ilman."
3470 DATA 4,2,0,0
3480 R$(4)="kosteassa kivitunnelissa. Ilma
ssa tiivistyvä vesi valuu alas seinä
pit- kin ja muodostaa lätköitä laattala
t- tialle."
3490 DATA 5,3,6,0
3500 R$(5)="liukastunut ja putoat syvään
kuop-paan... Terävät teräseipäät törö
tta- vät kuopan pohjalla..."
3510 DATA 0,0,0,0
3520 R$(6)="suuressa kaikuvassa metallitunn
e- lissa. Loistavat seinät luovat oud
on sinisen heijastuksen korkealla olevaan
kattoon."
3530 DATA 7,8,0,4
3540 R$(7)="pienessä synkässä huoneessa."
3550 R$(8)=R$(1)
3560 R$(9)="pitkanomaisessa hallissa.
Lattia notkuu hieman painosi johdosta."
3570 DATA 0,6,0,0,6,0,9,0,44,0,10,8
3580 R$(10)="romupuristimessa... Seinät
liik- kuvat sisäänpäin..."
3590 R$(11)="lyhyessä yhteystunnelissa."
3600 R$(12)="kapeassa ja korkeassa huoneess
a. Korkealla oleva kupolikatto pääst
ää lä- vitseen himmeänharmaata valoa."
3610 DATA 0,0,0,0,1,44,0,15,0,1,0,0
3620 R$(13)="suuressa tyhjässä halliss
a. Lattia on paksun, vuosisatoja van
han tomun peittämä."
3630 R$(14)="laser-kammiossa. Varo...
----- z a p ! -----"
3640 R$(15)=R$(1)
3650 DATA 0,10,0,1,0,0,0,0,39,0,11,0
3660 R$(16)="jonkinlaisessa voimakentäss
ä. Outo vihreä kaasu täyttää ilman."
3670 R$(17)="voimakentässä."
3680 FOR J=18 TO 23
3690 R$(J)=R$(17)

```

3700 NEXT J

3710 DATA 17,39,17,19,17,16,18,18,17,19,18,
17,18,38,16,17,38,18,21,22,17,18,19,2
0,40,17,20,23,25,17,22,18

3720 R\$(24)="kosteassa, haisevassa
huoneessa. Joku pieni otus istuu hiljaa yhde
ssä nurkassa ja katsoo sinua. Lattia
tuntuuolevan tehty kovaksipoljetusta lan-
nasta."

3730 R\$(25)="huoneessa, josta ei näy
olevan poispääsyä. Suuri lohikäärmeen nä
köinenotus lähestyy kieltään lipoen..."

3740 R\$(26)="vanhassa hylätyssä varastossa."

3750 DATA 26,40,41,25,0,0,0,0,0,24,0,0

3760 R\$(27)="hyvin valaistussa käytävässä,
jonka kunto ilmaisee sen olevan j
atku- vassa käytössä."

3770 R\$(28)="siirtihuoneessa. Huone näyttää
toimivan."

3780 R\$(29)="puhtaassa, hyvin valaistussa
käytävässä."

3790 DATA 45,0,28,41,0,0,0,27,31,27,30,32

3800 R\$(30)="ilmeisen tyhjässä teräksisessä
hautaholvissa."

3810 R\$(31)="laboratoriossa."

3820 R\$(32)=R\$(1)

3830 DATA 0,0,0,29,0,29,0,0,34,33,29,99

3840 R\$(33)="korkeakattoisessa kammiossa.
Lattia tuntuu ontolta jalkojen alla.

"

3850 R\$(34)="muukalaisten ravintolassa."

3860 R\$(35)="käytävän päässä."

3870 DATA 32,0,0,0,0,32,0,0,0,0,32,88

3880 R\$(36)="aarrekammiossa!"

3890 R\$(37)=R\$(17)

3900 R\$(38)=R\$(17)

3910 DATA 0,0,35,0,18,19,16,38,19,20,37,17

3920 R\$(39)="sileän seinän edessä."

3930 R\$(40)="voimakentän reunamilla."

3940 R\$(41)=R\$(1)

3950 DATA 88,15,0,0,88,22,18,16,42,43,27,24

3960 R\$(42)="pienessä siirtokammiossa."

3970 R\$(43)="tavallisessa huoneessa."

3980 R\$(44)="kapeassa käytävässä - kimmelt

```

ava oranssinen voimakenttä estää kulk
usi."
3990 DATA 0,41,0,0,41,5,0,0,88,9,0,0
4000 R$(45)="kapeassa tunnelissa. Tie on
tukit-tu tiiviisti suljetulla, painaval
la ovella."
4010 R$(46)="syvällä avaruudessa tuhansie
n tähtien ympäröimänä..."
4020 DATA 88,27,0,0,0,0,0,0
4030 R$(47)="pienessä muovikupolissa."
4040 R$(48)="tuuhean aarniometsän reunassa."
4050 R$(49)="keskellä metsää."
4060 R$(50)="taivaansinisen järven rannalla."
4070 DATA 0,0,48,0,0,0,49,47,49,50,49,48,
49,0,51,0
4080 R$(51)="pakoaluksen sisällä."
4090 DATA 0,0,0,50
4100 DATA 7,0,0,0,51,31,0,0,0,0,26,21,42,
0,12,34,33,13,28,43,47,50,36,16,0
4110 DATA Laatikko - jossa musta nappi,Pieni
metallikiekko,Ohut musta metallitan
ko,Lusikka,Avain,Amuletti,Pullo sinistä
nestettä,Kuutio,Kypärä
4120 DATA Säilykepurkki,Muistikirja,Pöytä,4
vipua (1 - 4),4 vipua (1 - 4),Koroke
,Kaappi,Kaappi,Teräslipas
4130 DATA Kone,4 vipua (1 - 4),Hylly,Erill
inen vipu,Pakoalus,Merkki,Töherretty s
einäkirjoitus,Lusikka
4140 DATA "P ","E ","I ","L "
4150 OB(5)=1
4160 GOTO 220
4170 IF CP=17 OR CP=19 OR CP=18 OR CP=37
OR CP=38 OR CP=20 OR CP=22 OR CP=23 THE
N 4180 ELSE 4200
4180 PRINT "Ulosmenotie on olemassa. Koeta
merkitä sijaintisi pudottamalla esine
itä."
4190 GOTO 970
4200 IF CP<>16 THEN 4230
4210 PRINT "Yritä lukea."
4220 GOTO 970
4230 IF CP<>13 THEN 4260
4240 PRINT "On aika ottaa pienet limonaadit!"
4250 GOTO 970

```

```

4640 PRINT#1,CP
4650 CLOSE
4660 GOTO 970
4670 REM PELIN LATAUS KASETILTA
4680 CLS
4690 LOCATE 0,6
4700 LINE INPUT "PAINA ENTER, KUN KASETTI
ON VALMIINA";W$
4710 LOCATE 0,9
4720 PRINT "LATAUS..."
4730 OPEN "KAS:KIEL" FOR INPUT AS 1
4740 FOR J=1 TO NN
4750 INPUT#1,OB(J)
4760 INPUT#1,OL(J)
4770 NEXT J
4780 INPUT#1,A
4790 INPUT#1,R$(44)
4800 INPUT#1,B
4810 INPUT#1,R$(39)
4820 INPUT#1,C
4830 INPUT#1,D
4840 INPUT#1,E
4850 INPUT#1,CP
4860 M(44,1)=A:M(39,1)=B:M(40,1)=C:M(32,4)
=D:M(35,4)=E
4870 CLOSE
4880 GOTO 970
4890 FOR DE=1 TO 400:NEXT DE
4900 CLS
4910 LOCATE 7,8
4920 PRINT "Ja niin tuli AKKILOPPU!"
4930 LOCATE 0,20
4940 INPUT "Uusi yritys";Q$
4950 IF LEFT$(Q$,1)="k" THEN RUN
4960 PRINT:INPUT "Oletko aivan varma päätö
ksestäsi";Q$
4970 IF LEFT$(Q$,1)="k" THEN 5020 ELSE RUN
4980 CLS
4990 LOCATE 5,8
5000 PRINT "<<      HARRE ON SINUN!      >>"
5010 LOCATE 0,20
5020 END


```

```

4640 PRINT#1,CP
4650 CLOSE
4660 GOTO 970
4670 REM PELIN LATAUS KASETILTA
4680 CLS
4690 LOCATE 0,6
4700 LINE INPUT "PAINA ENTER, KUN KASETTI
  ON VALMIINA";W$
4710 LOCATE 0,9
4720 PRINT "LATAUS..."
4730 OPEN "KAS:KIEL" FOR INPUT AS 1
4740 FOR J=1 TO NN
4750 INPUT#1,OB(J)
4760 INPUT#1,OL(J)
4770 NEXT J
4780 INPUT#1,A
4790 INPUT#1,R$(44)
4800 INPUT#1,B
4810 INPUT#1,R$(39)
4820 INPUT#1,C
4830 INPUT#1,D
4840 INPUT#1,E
4850 INPUT#1,CP
4860 M(44,1)=A:M(39,1)=B:M(40,1)=C:M(32,4)
  =D:M(35,4)=E
4870 CLOSE
4880 GOTO 970
4890 FOR DE=1 TO 400:NEXT DE
4900 CLS
4910 LOCATE 7,8
4920 PRINT "Ja niin tuli AKKILOPPU!"
4930 LOCATE 0,20
4940 INPUT "Uusi yritys";Q$
4950 IF LEFT$(Q$,1)="k" THEN RUN
4960 PRINT:INPUT "Oletko aivan varma päätö
ksestäsi";Q$
4970 IF LEFT$(Q$,1)="k" THEN 5020 ELSE RUN
4980 CLS
4990 LOCATE 5,8
5000 PRINT "<<      HARRE ON SINUN!      >>"
5010 LOCATE 0,20
5020 END

```


KOLMIULOTTEINEN SOKKELO



Olet ansassa kiemurtelevien käytävien muodostaman sokkelon eteläosassa, josta sinun täytyy etsiä jossakin pohjoisessa oleva ulospääsy. Kulkiessasi näet sokkelon sisäosan perspektiivikuvana ja joko edessäsi tai sivuillasi on avonaisia tai suljettuja ovia.

Kuinka peliä pelataan

Halutessasi kääntyä johonkin ilmansuuntaan paina näppäimiä P, E, I tai L. Painamalla näppäintä K pääset eteenpäin.

Vaikka ilmansuuntanäppäimet muuttavatkin näkymääsi, sinä pysyt paikallasi.

Pelin alussa annettavalla vaikeusasteella määritellään sokkelon koko. Ole varovainen, sillä vaikeimmat pelit on tarkoitettu vain niille, joilla on hyvä suuntavaisto ja loistava muisti.

Tietoa ohjelmasta

Rivin n:o

170	Aikarajan asetus
180	Antaa RND-funktiolle siemenluvun.
210	Aikarajan alkuarvo
250-260	Vaikeusaste
300-310	Piirtää yhtenäisen peräseinän.
330-340	Piirtää peräseinän ja oven.
360-370	Piirtää yhtenäisen seinän oikealle.
390-400	Piirtää yhtenäisen seinän vasemmalle.
420-430	Piirtää seinän ja oven oikealle.
450-460	Piirtää seinän ja oven vasemmalle.
480-700	Järjestää sokkelon valmiiksi.
740	Kytkee keskeytysrutiinin.
780-990	Tulostaa kunkin suunnan näkymän.
1010-1030	Odottaa ohjetta näppäimistöltä.
1040-1070	Muuttaa suunnat numeroiksi.
1100-1130	Liike eteenpäin
1180-1250	Vapautuminen sokkelosta
1320-1460	Keskeytyksen kutsuma aikarajaa pienentävä aliohjelma

Ohjelma: SOKK-JS

```

100 REM *****
110 REM *
120 REM * SOKKELO *
130 REM *
140 REM *****
150 REM
160 CLS
170 ON INTERVAL=200 GOSUB 1320
180 X=RND(-TIME)
190 COLOR 15,1
200 LOCATE 13,4:PRINT "SOKKELO "
```



```

210 IV=100
220 OPEN "GRP:" AS 1
230 DIM M(22,42)
240 LOCATE 2,12
250 INPUT "VAIKEUSASTE (5 - 40)";SK
260 IF SK<5 OR SK>40 THEN CLS:GOTO 240
270 LOCATE 2,16:PRINT "ODOTA HETKI..."
280 GOTO 480
290 REM YHTENÄINEN PERÄSEINÄ
300 DRAW "BM 90,60;C12R80D80L80U80"
310 RETURN
320 REM OVELLINEN PERÄSEINÄ
330 DRAW "BM 90,60;C12R80D80L25U50L30D50
L25U80"
340 RETURN
350 REM YHTENÄINEN SIVUSEINÄ
360 DRAW "BM 170,60;C12E50D180H50U80"
370 RETURN
380 REM YHTENÄINEN SIVUSEINÄ
390 DRAW "BM 90,60;C12H50D180E50U80"
400 RETURN
410 REM OVELLINEN SIVUSEINÄ
420 DRAW "BM170,60;C12E50D180H20U90G10D70
H20U80"
430 RETURN
440 REM OVELLINEN SIVUSEINÄ
450 DRAW "BM90,60;C12H50D180E20U90F10D70E
20U80"
460 RETURN
470 REM SOKKELON TEKÖ
480 C=INT(RND(1)*26)+2
490 R=1
500 R1=INT(RND(1)*4)+1
510 D=INT(RND(1)*3)+1
520 IF D=0 THEN 510
530 IF D=D1 THEN 510
540 D1=D
550 FOR J=1 TO R1
560 IF D=1 THEN C=C-1:IF C<2 THEN C=2
570 IF D=2 THEN R=R+1:IF R=SK+1 THEN 620
580 IF D=3 THEN C=C+1:IF C>19 THEN C=19
590 M(C,R)=1
600 NEXT J
610 GOTO 500

```

```

620 FOR J=1 TO 20
630 M(J,1)=3:M(J,SK)=3:M(J,SK+1)=0
640 FOR K=1 TO SK
650 M(1,K)=0:M(20,K)=0:M(19,K)=0
660 IF RND(1)<.8 THEN 690
670 IF M(J,K)<>0 THEN 690
680 M(J,K)=2
690 NEXT K
700 NEXT J
710 COLOR 12,1,1
720 SCREEN 2
730 CLS
740 INTERVAL ON
750 R=SK:C=15:DR=1
760 COLOR 14:ON DR GOTO 780,840,900,960
770 REM SUUNTA POHJOISEEN
780 IF M(C,R-1)>0 THEN GOSUB 330 ELSE
GOSUB 300
790 IF M(C+1,R)>0 THEN GOSUB 420 ELSE
GOSUB 360
800 IF M(C-1,R)>0 THEN GOSUB 450 ELSE
GOSUB 390
810 PRESET(74,180):PRINT#1,"SUUNTA
POHJOISEEN"
820 GOTO 1010
830 REM SUUNTA ETELAÄN
840 IF M(C,R+1)>0 THEN GOSUB 330 ELSE
GOSUB 300
850 IF M(C-1,R)>0 THEN GOSUB 420 ELSE
GOSUB 360
860 IF M(C+1,R)>0 THEN GOSUB 450 ELSE
GOSUB 390
870 PRESET(74,180):PRINT#1,"SUUNTA ETELAÄN"
880 GOTO 1010
890 REM SUUNTA ITÄÄN
900 IF M(C+1,R)>0 THEN GOSUB 330 ELSE
GOSUB 300
910 IF M(C,R+1)>0 THEN GOSUB 420 ELSE
GOSUB 360
920 IF M(C,R-1)>0 THEN GOSUB 450 ELSE
GOSUB 390
930 PRESET(74,180):PRINT#1,"SUUNTA ITÄÄN"
940 GOTO 1010
950 REM SUUNTA LÄNTEEN
960 IF M(C-1,R)>0 THEN GOSUB 330 ELSE

```


```

GOSUB 300
970 IF M(C,R-1)>0 THEN GOSUB 420 ELSE
GOSUB 350
980 IF M(C,R+1)>0 THEN GOSUB 450 ELSE
GOSUB 390
990 PRESET(74,180):PRINT#1,"SUUNTA LANTEEN"
1000 REM ODOTTAA TIETOA NÄPPÄIMISTÖLTÄ
1010 A$=INKEY$
1020 IF A$="" THEN 1010
1030 BEEP
1040 IF A$="P" THEN DR=1:GOTO 1140
1050 IF A$="E" THEN DR=2:GOTO 1140
1060 IF A$="I" THEN DR=3:GOTO 1140
1070 IF A$="L" THEN DR=4:GOTO 1140
1080 IF A$<>"K" THEN 1010
1090 REM LIIKE ETEENPAIN
1100 IF DR=1 THEN R=R-1:IF M(C,R)=0 THEN
R=R+1:GOTO 1010
1110 IF DR=2 THEN R=R+1:IF M(C,R)=0 THEN
R=R-1:GOTO 1010
1120 IF DR=3 THEN C=C+1:IF M(C,R)=0 THEN
C=C-1:GOTO 1010
1130 IF DR=4 THEN C=C-1:IF M(C,R)=0 THEN
C=C+1:GOTO 1010
1140 CLS
1150 COLOR 12,1,1
1160 SCREEN 2
1170 IF R>=2 THEN 1260
1180 INTERVAL OFF
1190 SCREEN 0
1200 REM ONNISTUIT
1210 CLS:COLOR 15,4,5
1220 PLAY "T24004CDEFAB05CDEFAB"
1230 LOCATE 12,8:PRINT "ONNEKSI OLKOON!"
1240 LOCATE 4,10:PRINT" TARVITSIT AINOASTAAN";CN
; "SIIRTOA."
1250 GOTO 1280
1260 CN=CN+1
1270 GOTO 760
1280 LOCATE 1,22
1290 CLOSE
1300 END
1310 REM VÄHENTÄÄ JÄLJELLÄ OLEVAA AIKAA
1320 IV=IV-1
1330 PRESET(168,10):COLOR 1:PRINT#1, CHR$

```

```
(219);CHR$(219)
1340 PRESET(72,10):COLOR 14
1350 PRINT#1,"PELIAIKAA =";IV
1360 IF IV>0 THEN 1460
1370 REM AIKA LOPPUI
1380 PLAY "T24005GFEDCBA04GFEDCBA"
1390 FOR DE=1 TO 180:NEXT DE
1400 SCREEN 0
1410 CLS
1420 COLOR 15,4,5
1430 LOCATE 12,9:PRINT "AIKA LOPPUI!"
1440 LOCATE 1,22
1450 GOTO 1290
1460 RETURN
```

SÄHKÖTYS



Tämän ohjelman avulla voit verestää sähkötystaitojasi, vaikkapa purjehduslomaa varten.

Voit kuunnella satunnaisia kirjaimia, arvata sanoja tai lähettää sanomia.

Kuinka ohjelmaa käytetään

1. Satunnaiset kirjaimet

Tällä toiminnolla tietokone lähettää satunnaisia kirjaimia valitsemallasi nopeudella.

2. Satunnaiset sanat

DATA-lauseista otetaan sanoja satunnaisesti ja ne tulostetaan kuultavana koodina. Tämän jälkeen sinun on kirjoitettava ko. sana tietokoneelle. Vaikeusaste vaihtelee sanojen ja nopeuden mukaan.

3. Sanoman lähettäminen

Tämän toiminnon avulla voit lähettää haluamasi sanoman.

Ohjelma pystyy lähettämään suunnilleen 60 merkkiä minuutissa. Tämä maksiminopeus riittää radioamatöörien sähkötyskokeen suorittamiseen. Jos olet aloittelija, älä yritä oppia kirjainten pisteitä ja viivoja ulkoa. Kuuntele vain kunkin kirjaimen ääntä ja varmasti hämmästyit, kuinka helposti ne voi oppia.

Tietoa ohjelmasta

Rivin n:o

70

Morsemerkit ja sanoja sisältävien taulukoiden tilavaraukset

100–120

Lukee morsemerkit.

140–200

Kirjoittaa valikon.

210–230

Odottaa näppäimen painallusta.

240

Tarkistaa vastauksen.

250–260

Pyytää nopeuden arvoa.

290

Osa ohjelmasta erikoistarkoituksiin

310

Satunnaiset sanat

330–350

Lukee sanat DATA-lausekkeista.

370

Valitsee sanan satunnaisesti.

380–460

Tulostaa sanan morsekoodina.

420–440

Skandinaaviset merkit

470

Odottaa vastausta.

490

Tarkistaa, oliko vastaus oikein.

500–520

Vastaus oli oikein.

530–540

Vastaus oli väärin.

550–600

Kirjoittaa sanan ja tyhjentää näytön.

660

Kysyy sanomaa.

680–690

Odottaa sanomaa.

700–810

Tulostaa sanoman morsekoodina.

850

Satunnaiskirjaimia

880

Valitsee kirjaimet satunnaisesti.

890–910

Skandinaaviset kirjaimet

920

Tulostaa kirjaimen.

1080-1180 Aliohjelma morsetusääntä varten
 1230-1260 DATA-tiedot morsemerkkejä varten, 1 vastaa
 pistettä ja 3 viivaa
 1280-1310 Sanojen DATA-tiedot

Ohjelma: MORS-JS

```

10 REM *****
20 REM *      *
30 REM * SÄHKÖTYS *
40 REM *****
50 REM VV$=""
60 REM VV$=VV$+"
70 DIM M$(29),U$(29)
80 L=RND(-TIME)
90 REM LUKEE MERKKITIEDOT
100 FOR J=1 TO 29
110 READ M$(J)
120 NEXT J
130 REM VALIKON TULOSTAMINEN
140 CLS
150 COLOR 12,1,1
160 LOCATE 11,3:PRINT "SÄHKÖTYS"
170 LOCATE 1,8:COLOR 13:PRINT "1. LÄHETÄ
  SANOMA"
180 LOCATE 1,10:PRINT "2. SATUNNAISET
  KIRJAIMET"
190 LOCATE 1,12:PRINT "3. SANAT"
200 LOCATE 1,17:COLOR 5:PRINT "PAINA 1, 2
  TAI 3"
210 A$=INKEY$
220 IF A$="" THEN 210
230 BEEP
240 IF A$<"1" OR A$>"3" THEN 210
250 LOCATE 1,20:COLOR 13:INPUT "NOPEUS
(1 - 100)";SP
260 IF SP>100 OR SP<1 THEN 250
270 CLS
280 LOCATE 12,1:COLOR 11:PRINT
  "NOPEUS = ";SP
290 ON VAL(A$) GOTO 660,860,310
300 REM SANAT
310 RESTORE 1190

```

```

320 REM LUKEE SANAT
330 FOR J=1 TO 20
340 READ W$(J)
350 NEXT J
360 FOR N=1 TO 10
370 R=INT(RND(1)*20)+1
380 FOR J=1 TO LEN(W$(R))
390 L$=MID$(W$(R),J,1)
400 IF L$=" " THEN 460
410 L=ASC(L$)-64
420 IFL=79THENL=27
430 IFL=78THENL=28
440 IFL=89THENL=29
450 GOSUB 990
460 NEXT J
470 LOCATE 1,20:INPUT "MIKÄ SANA ON
  KYSEESSÄ":Q$
480 LOCATE 1,20:PRINT VV$
490 IF Q$<>W$(R) THEN 530
500 PLAY "T24004ABCDEFGG05CDEF"
510 LOCATE 12,8:PRINT "OIKEIN!"
520 GOTO 550
530 PLAY "T220BAGFEDC"
540 LOCATE 13,8:PRINT "VÄÄRIN!"
550 LOCATE 10,10:PRINT "SANA OLI"
560 LOCATE 10,12:PRINT W$(R)
570 FOR DE=1 TO 2000:NEXT DE
580 LOCATE 7,8:PRINT VV$
590 LOCATE 10,10:PRINT VV$
600 LOCATE 10,12:PRINT VV$
610 NEXT N
620 FOR DE=1 TO 2000:NEXT DE
630 CLS
640 GOTO 160
650 REM SANOMAN LÄHETYS
660 LOCATE 4,10:COLOR 7:PRINT "KIRJOITA
  SANOMASI"
670 PRINT:PRINT
680 INPUT I$
690 PRINT:PRINT
700 FOR J=1 TO LEN(I$)
710 L$=MID$(I$,J,1)
720 IFL$="A"THENL=27:PRINT L$::GOTO800
730 IFL$="Ä"THENL=28:PRINT L$::GOTO800

```



```

740 IFL$="ö" THEN L=29:PRINT L$;:GOTO 800
750 PRINT L$;
760 IF L$=" " THEN GOSUB 1100:GOTO 810
770 IF L$="." OR L$="!" OR L$="?" THEN 820
780 IFL$>"Z" OR L$<"A" THEN PRINT:PRINT
    "KÄYTÄ VAIN ISOJA KIRJAIMIA!":GOTO 670
790 L=ASC(L$)-64
800 GOSUB 990
810 NEXT J
820 FOR DE=1 TO 2000:NEXT DE
830 CLS
840 GOTO 160
850 REM SATUNNAISET KIRJAIMET
860 LOCATE 1,4
870 FOR J=1 TO 50
880 L=INT(RND(1)*29)+1
890 IFL=27 THEN PRINT CHR$(79+64);:GOTO 930
900 IFL=28 THEN PRINT CHR$(78+64);:GOTO 930
910 IFL=29 THEN PRINT CHR$(89+64);:GOTO 930
920 PRINT CHR$(L+64);
930 GOSUB 990
940 NEXT J
950 FOR DE=1 TO 2000:NEXT DE
960 CLS
970 GOTO 160
980 REM MERKKIEN ÄÄNI
990 SOUND 0,90:SOUND 1,0:SOUND 8,0:SOUND
    7,254
1000 FOR K=1 TO LEN(M$(L))
1010 IF MID$(M$(L),K,1)="0" THEN 1060
1020 SOUND 8,12
1030 FOR DE=1 TO 30*(VAL(MID$(M$(L),K,1)))+6
1040 NEXT DE
1050 SOUND 8,0
1060 FOR DE=1 TO 90-SP/2:NEXT DE
1070 NEXT K
1080 FOR DE=1 TO 1100-SP*10:NEXT DE
1090 RETURN
1100 FOR DE=1 TO 200
1110 NEXT DE
1120 RETURN
1130 REM MORSEMERKKIEN DATA-TIEDOT
1140 DATA 1300,3111,3131,3110,1000,1131,
    3310
1150 DATA 1111,1100,1333,3130,1311,3300,

```

3100
1160 DATA 3330,1331,3313,1310,1110,3000,
1130
1170 DATA 1113,1330,3113,3133,3311,13313,
1313,3331
1180 REM SANOJEN DATA-TIEDOT
1190 DATA MÄNTY,KUUSI,KOIVU,LEPPÄ,HÄRPÄ
1200 DATA FILOSOFI,OPETTAJA,LÄÄKÄRI,PORMEST
ARI,INSINÖÖRI
1210 DATA VARIS,HAI,KISSA,KALA,MAJAVA
1220 DATA VOLVO,TOYOTA,MAZDA,OPEL,SAB

FAARAON KIROUS



Tämä on peli, josta jopa vanhemmat ystäväsi tai koko perheesi pitävät, sillä peli vaatii logiikkaa, muistia ja päättelykykyä. Nyt voit unohtaa taistelupelien pelaamisen.

Tässäkin pelissä etsit tietenkin aarretta, mutta sinun on päätettävä, kannattaako keräillä kiviurnia, rautatankoja, savi-ruukkuja tai tikareita matkallasi pyramidin keskusta.

Sinun on päätettävä, menetkö itään, länteen, pohjoiseen vai etelään ja avaatko ovia, astutko sisään muinaisiin huoneisiin vai ryntäätkö epäröimättä sokkeloihin. Lattiat voivat muren-

tua, seinät sortua ja sumut voivat ympäröidä sinut, koska haudat on suojattu sinun tapaisia rosvoja vastaan.

Kuinka peliä pelataan

Syöttötiedot annetaan suuraakkosin ja tiedoiksi riittää tavallinen verbi ja substantiivi, esimerkiksi MENE ETELÄÄN, TUTKI ARKKUA, OTA AVAIN jne. Ilmansuunnat voidaan antaa niiden etukirjaimina P, E, I ja L.

Seikkailupelin kirjoittaminen koneeseen vie niin paljon aikaa, että se kannattaa kirjoittaa virheiden välttämiseksi useassa vaiheessa ja välillä tallentaa nauhalle. Ole hyvin huolellinen kirjoittaessasi DATA-lausekkeita, sillä pienikin virhe siellä sekoittaa helposti ohjelman loogisen rakenteen.

Pelitilanne voidaan myös tallentaa, joten peliä ei tarvitse pelata loppuun yhdellä istumalla. Tallennus tapahtuu TALLENA-käskyllä ja vastaavasti vanha pelitilanne ladataan käskyllä LATAA.

Tietoa ohjelmasta

Rivin n:o	
210	Tilavaraukset taulukoille, joissa ovat esineiden ja huoneiden kuvaukset.
240-380	Lukee suunnat, esineiden sijainnit ja kuvaukset DATA-lausekkeista.
420-660	Kirjoittaa ohjeet.
700-710	Tyhjentää näytön yläosan.
730	Kirjoittaa paikan kuvauksen.
750-830	Esineet, jotka ovat tässä paikassa.
850-900	Kirjoittaa mahdolliset suunnat.
920	Piirtää viivan kuvauksen alle.
1030-1050	Odottaa seuraavaa ohjetta.
1060-1120	Erikoistapauksia
1150	Etsii välilyönnin ohjeesta.
1170	Verbi sijoitetaan muuttujaan VB\$.
1180-1240	Erottaa substantiivin verbistä ja antaa kummallekin numeroarvon.
1260	Hyppää oikeaan kohtaan verbiä varten.
1270-1340	Verbi Lyö
1350-1370	Lue
1380-1460	Liikuta
1480-1540	Pudota

1560-1590	Mene
1610-1710	Tarkista mahdollinen suunta.
1730-1940	Verbi Ota
1960-2680	Tutki
2700-3140	Avaa
3160-3250	Kaiva
3260-3280	Verbien ja substantiivien DATA-tiedot
3290-3930	Huonekuvaukset ja suuntien DATA-tiedot
3950-4120	Auta
4160-4320	Tallenna pelitilanne nauhalle.
4340-4500	Lataa vanha pelitilanne nauhalta.
4520-4600	Epäonnistuit.
4610-4650	Onnistuit.
4670-4690	Viiverutiini

Ohjelma: FAAR-JS

```

100 REM *****
110 REM *
120 REM * FAARAON KIROUS *
130 REM *
140 REM *****
150 REM
160 CLS
170 CLEAR 2000
180 COLOR 1,15
190 LOCATE 10,10
200 PRINT "FAARAON KIROUS"
210 DIM M(44,4),OB(40),OJ$(40),R$(44),DR$(4),
    OL(40)
220 GOTO 3260
230 REM LUKEE DATA-TIEDOT
240 READ U$,N$,Z$,RN,NN,CP
250 N$=N$+Z$
260 FOR J=1 TO RN
270 FOR K=1 TO 4
280 READ M(J,K)
290 NEXT K,J
300 FOR J=1 TO NN
310 READ OL(J)
320 NEXT J
330 FOR J=1 TO NN
340 READ OJ$(J)
350 NEXT J

```

```

360 FOR J=1 TO 4
370 READ DR$(J)
380 NEXT J
390 CLS
400 LOCATE 0,4
410 REM KIRJOITTAA OHJEET
420 PRINT "EGYPTISSÄ ON ERÄS KAUAN UNOHDUKS
ISSA"
430 PRINT "OLLUT LAAKSO, JOSSA HUHUN MUKAAN
ON"
440 PRINT "KÄTKETTYNÄ VALTAVA AARRE."
450 PRINT:PRINT
460 PRINT "MONET OVAT YRITTÄNEET LÖYTÄÄ
TÄMÄN"
470 PRINT "FAARAON KÄTKÖN, MUTTA KUKAAN EI
OLE TULLUT TAKAISIN."
480 PRINT:PRINT
490 PRINT "JOS OLET HALUKAS, VOISIMME YRI
TTÄÄ"
500 PRINT "LÖYTÄÄ AARTEEN YHDESSÄ."
510 PRINT:PRINT:PRINT
520 PRINT "PAINA JOTAIN NÄPPÄINTÄ..."
530 NU$=INKEY$
540 IF NU$="" THEN 530
550 CLS
560 LOCATE 0,4
570 PRINT "ANNA MINULLE KAKSI SANAA OHJEE
KSI;"
580 PRINT "ESIM. MENE POHJOISEEN, TUTKI
OVI, OTA ESINELUETTELO, PUDDOTA TANKO."
590 PRINT:PRINT:PRINT
600 PRINT "ON VIELÄ MONIA MUITAKIN
KÄSKYJÄ, JOITA"
610 PRINT "VOIT KÄYTTÄÄ, MUTTA SAAT KEKSIA
NE ITSE."
620 PRINT:PRINT
630 PRINT "EPÄTOIVON VALLATESSA MIELESI,
VOIT KIR-JOITTAA KÄSKYN 'AUTA'."
640 PRINT:PRINT
650 INPUT "OLETKO HALUKAS LÄHTEMÄÄN SEIKK
AILUUN, JOKA TUO EHKÄ VAAROJA TULLESSA
AN";NU$
660 IF LEFT$(NU$,1)<>"K" THEN END
670 CLS

```

```

680 GOTO 1000
690 REM KIRJOITTAA HUONEIDEN KUVAUKSE
T,      NÄKYVILLÄ OLEVAT ESINEET SEKÄ
      MAHDOLLISET SUUNNAT
700 LOCATE 0,0
710 PRINT SPACE$(255);SPACE$(255)
720 LOCATE 0,2
730 PRINT "OLET ";R$(CP)
740 REM TARKISTAA ONKO HUONEESSA
ESI-      NEITÄ
750 FOR J=1 TO NN
760 IF OL(J)=CP THEN 790
770 NEXT J
780 GOTO 840
790 PRINT:PRINT "NÄKYVILLÄ: ";
800 FOR J=1 TO NN
810 IF OL(J)<>CP THEN 830
820 PRINT OJ$(J)
830 NEXT J
840 PRINT:PRINT "SUUNNAT: ";
850 FOR J=1 TO 4
860 IF M(CP,J)=0 OR M(CP,J)=88 THEN 880
870 PRINT DR$(J);:DF=1
880 NEXT J
890 IF DF=0 THEN PRINT "??"
900 DF=0
910 PRINT:PRINT
920 PRINT STRING$(37,219)
930 REM VIRHEELLINEN LIIKE!
940 IF CP=13 OR CP=35 OR CP=36 OR CP=25
OR CP=42 THEN 960
950 GOTO 990
960 GOSUB 4670
970 GOSUB 4670
980 GOTO 4520
990 RETURN
1000 PRINT
1010 GOSUB 700
1020 REM ODOTTAA SEURAAVAA OHJETTA
1030 LOCATE 0,23:PRINT
1040 INPUT "ANNA OHJEET";IP$
1050 PRINT
1060 IF IP$="AUTA" THEN 3950
1070 IF IP$="P" THEN IP$="MEN POHJOISEEN"
1080 IF IP$="E" THEN IP$="MEN ETELÄÄN"

```

```

1090 IF IP$="I" THEN IP$="MEN ITÄÄN"
1100 IF IP$="L" THEN IP$="MEN LÄNTEEN"
1110 IF IP$="TALLENNA" THEN 4150
1120 IF IP$="LATAA" THEN 4330
1130 IF LEN(IP$)<6 THEN PRINT "EN
    YMMÄRRÄ!":GOTO 1000
1140 REM ETSII VALILYÖNNIN SYÖTTÖTI
E-          DOSTA
1150 X=INSTR(1,IP$," ")
1160 REM EROTTAA VERBIN
1170 VB$=LEFT$(IP$,3)
1180 NT$=MID$(IP$,X+1,LEN(IP$)-X)
1190 REM EROTTAA SUBSTANTIIVIN
1200 NO$=MID$(IP$,X+1,3)
1210 VB=(INSTR(1,VB$,VB$)+2)/3
1220 IF VB<1 THEN PRINT "EN YMMÄRRÄ VERB
    IÄ.":GOTO 1000
1230 NO=(INSTR(1,N$,NO$)+2)/3
1240 IF NO<1 OR (INT(NO)<>NO) THEN PRIN
    T "EN YMMÄRRÄ SANAA VERBIN JÄLKEEN.":GOTO
    1000
1250 REM VALITSEE VERBIA VASTAAVAN O
H-          JELMARIIVIN
1260 ON VB GOTO 1550,1380,1950,2690,3150,
    1720,1720,1480,1480,1350,1380,1270
1270 REM LYÖ
1280 IF NO=24 AND OB(5)=1 AND CP=20 THEN
    1300
1290 GOTO 1330
1300 PRINT "SALAHOVI ON RUENNUT!"
1310 M(20,4)=30
1320 GOTO 1000
1330 PRINT "MITÄÄN EI TAPAHTUNUT."
1340 GOTO 1000
1350 REM LUE
1360 PRINT "KOETA TUTKIA."
1370 GOTO 1000
1380 REM LIIKU
1390 IF NO=26 AND CP=4 THEN 1410
1400 GOTO 1440
1410 PRINT "SALAKÄYTÄVÄ!"
1420 M(4,4)=39
1430 GOTO 1000
1440 REM EI MITÄÄN ERIKOISTA

```



```
1450 PRINT "MITÄÄN EI TAPAHTUNUT."  
1460 GOTO 1000  
1470 REM DROP  
1480 IF OB(NO)<>1 THEN 1530  
1490 OB(NO)=0  
1500 OL(NO)=CP  
1510 PRINT "OLET PUHDOTTANUT SEN."  
1520 GOTO 1000  
1530 PRINT NT$;" EI OLE SINULLA."  
1540 GOTO 1000  
1550 REM MEME  
1560 IF NO=20 AND CP=30 AND OB(1)=1 T  
HEN 1580  
1570 GOTO 1600  
1580 CP=11  
1590 GOTO 1000  
1600 REM SUUNNAT  
1610 IF NO<28 OR NO>31 THEN 1630  
1620 GOTO 1650  
1630 PRINT "ET VOI MENNÄ SIIHEN SUUNTA  
AN!"  
1640 GOTO 1000  
1650 IF M(CP,NO-27)=0 THEN 1630  
1660 IF M(CP,NO-27)=88 THEN 1680  
1670 GOTO 1700  
1680 PRINT "ET VOI MENNÄ SINNE - VIEL  
Ä!"  
1690 GOTO 1000  
1700 CP=M(CP,NO-27)  
1710 GOTO 1000  
1720 REM OTA  
1730 IF NO<>32 THEN 1790  
1740 FOR J=1 TO NN  
1750 IF OB(J)<>1 THEN 1770  
1760 PRINT OJ$(J)  
1770 NEXT J  
1780 GOTO 1000  
1790 IF NO>13 THEN PRINT "ET VOI OTTA  
A SITÄ!":GOTO 1000  
1800 FOR J=1 TO 15  
1810 IF OB(J)=1 THEN CN=CN+1  
1820 NEXT J  
1830 IF CN<4 THEN 1870  
1840 PRINT "ET VOI KANTAA ENEMPÄÄ!"
```

```

1850 CN=0
1860 GOTO 1000
1870 CN=0
1880 IF OL(NO)<>CP THEN 1930
1890 OL(NO)=0
1900 OB(NO)=1
1910 PRINT "SE ON SINUN."
1920 GOTO 1000
1930 PRINT NT$;" EI OLE TÄÄLLÄ!"
1940 GOTO 1000
1950 REM TUTKI
1960 IF NO=1 AND (OB(1)=1 OR OL(1)=CP)
THEN 1980
1970 GOTO 2120
1980 PRINT "SE ON TÄYNNÄ HIEROGLYFEJ
A!"
1990 PRINT
2000 GOSUB 4670
2010 PRINT "ODOTA...":PRINT
2020 GOSUB 4670
2030 PRINT "SISÄPUOLELLA ON PAPERISU
IKALE..."
2040 PRINT
2050 GOSUB 4670
2060 PRINT "SIINÄ LUKEE: TAMÄ MUKANAA
N VOI MENNÄ"
2070 PRINT "OMAA PEILIKUVAANSA."
2080 PRINT
2090 PRINT "SUOMENTANUT: OY MUINAISU
OMENNOS AB"
2100 PRINT "PAINOVUOSI 2040 E.KR."
2110 GOTO 1000
2120 IF NO=3 AND (OB(3)=1 OR OL(3)=
CP) THEN 2140
2130 GOTO 2250
2140 PRINT "HMMM..."
2150 PRINT
2160 GOSUB 4670
2170 PRINT "SITÄ ON VAIKEA LUKEA..."
2180 PRINT
2190 GOSUB 4670
2200 PRINT "LUULEN, ETTÄ SIINÄ LUKEE:
LYÖ MIESTÄ RAUTATANGOLLA JA LEPOPAIKKASI
ON **MEN-NYTTÄ**."

```

```
2210 PRINT
2220 GOSUB 4670
2230 PRINT "ENEMPÄÄ EN SAA SELVILLE."
2240 GOTO 1000
2250 IF NO=26 AND CP=4 THEN 2270
2260 GOTO 2290
2270 PRINT "SE EI NÄYTÄ PAINAVALTALA!"
2280 GOTO 1000
2290 IF NO=7 AND (OB(7)=1 OR OL(7)=
CP) THEN 2310
2300 GOTO 2330
2310 PRINT "SIINÄ ON REIKÄ!"
2320 GOTO 1000
2330 IF NO=9 AND (OB(9)=1 OR OL(9)=
CP) THEN 2350
2340 GOTO 2380
2350 PRINT "SIELLÄ ON VANHA PERGAMEN
TTI."
2360 OL(3)=CP
2370 GOTO 1000
2380 IF NO=16 AND CP=5 THEN 2400
2390 GOTO 2420
2400 PRINT "SIINÄ ON AVAIMENREIKÄ."
2410 GOTO 1000
2420 IF NO=20 AND CP=30 THEN 2440
2430 GOTO 2460
2440 PRINT "SIELLÄ ON RUMAN NÄKÖINEN OLIO!"
2450 GOTO 1000
2460 IF NO=10 AND (OB(10)=1 OR OL(10)=CP)
THEN 2480
2470 GOTO 2570
2480 PRINT "SISÄLLÄ ON MYRKYLLINEN HÄMÄHÄKK
I."
2490 PRINT
2500 GOSUB 4670
2510 PRINT "OOPS!"
2520 PRINT
2530 GOSUB 4670
2540 PRINT "LIIAN MYÖHÄÄN!"
2550 GOSUB 4670
2560 GOTO 4520
2570 IF NO=25 AND CP=10 THEN 4610
2580 IF NO=26 AND CP=15 THEN 2600
2590 GOTO 2620
```

```

2600 PRINT "ET VOI LIIKUTTAA TÄTÄ!"
2610 GOTO 1000
2620 IF NO=11 AND (OB(11)=1 OR OL(11)=CP)
THEN 2640
2630 GOTO 2660
2640 PRINT "SEN SISÄLLÄ ON PALSAMOITU TORA
KKA."
2650 GOTO 1000
2660 REM EI MITÄÄN ERIKOISTA
2670 PRINT "EI MITÄÄN ERIKOISTA."
2680 GOTO 1000
2690 REM AVAA
2700 IF NO=17 AND CP=5 THEN 2720
2710 GOTO 2750
2720 PRINT "KATSO!"
2730 OL(23)=5
2740 GOTO 1000
2750 IF NO=17 AND CP=7 THEN 2770
2760 GOTO 2810
2770 PRINT "KATSO!"
2780 IF OL(1)<>0 OR OB(1)=1 THEN 1000
2790 OL(1)=7
2800 GOTO 1000
2810 IF NO=17 AND CP=9 THEN 2830
2820 GOTO 2890
2830 PRINT "KATSO!"
2840 IF OL(2)<>0 OR OB(2)=1 THEN 2860
2850 OL(2)=9
2860 IF OL(6)<>0 OR OB(6)=1 THEN 1000
2870 OL(6)=9
2880 GOTO 1000
2890 IF NO=6 AND (OB(6)=1 OR OL(6)=
CP) THEN 2910
2900 GOTO 3080
2910 PRINT "VARO!"
2920 PRINT:GOSUB 4670
2930 PRINT "SISÄLLÄ ON MYRKYLLINEN KÄÄ
RME!"
2940 PRINT
2950 GOSUB 4670
2960 PRINT "NOPEASTI! MIKÄ RUTTAA KÄÄ
RMEENPUUREMAAN?"
2970 PRINT
2980 FOR J=1 TO 1000

```

```

2990 NU$=INKEY$
3000 IF NU$<>" " THEN J=1000
3010 NEXT J
3020 IF NU$="" THEN 3050
3030 PRINT "PHUH! JUURI AJOISSA!"
3040 GOTO 1000
3050 PRINT "LIIAN MYÖHÄÄN..."
3060 GOSUB 4670
3070 GOTO 4520
3080 IF MO=16 AND CP=5 AND OB(2)=1 THEN 3100
3090 GOTO 3130
3100 PRINT "SE ON RUKI."
3110 M(5,3)=16
3120 GOTO 1000
3130 PRINT "ET VOI TEHDÄ SITÄ."
3140 GOTO 1000
3150 REM KAIVA
3160 IF CP<>23 THEN 3230
3170 PRINT "OLET LÖYTÄNYT SALAISEN SIS
AANKÄYNNIN!"
3180 PRINT
3190 GOSUB 4670
3200 PRINT "SE NÄYTTÄÄ PELOTTAVALT
A... MIELESTÄNI SINUN EI PITÄISI
MENNÄ SINNE!"
"
3210 M(23,1)=24
3220 GOTO 1000
3230 REM EI MITÄÄN ERIKOISTA
3240 PRINT "SIELLÄ EI OLE MITÄÄN."
3250 GOTO 1000
3260 DATA MENLIITUTAVAKRIOTAHANPU
DJATLUEVEDLYS
3270 DATA VIIIVAPERTYHTANLAASANTI
KUURRUUPULTYH
3280 DATA TYHHIEPYRÖVIARKKAMLUOPEITY
HTYHMUUPATMERKIUTYHPOHEJEITALANESI
3290 DATA 44,27,3
3300 R$(1)="KUOLLEIDEN LAAKSON PÄÄSS
A. PYSTY- SUORAT KALLIOSEINÄT KOHOAVAT
YMPÄR
IL- LASI."
3310 R$(2)="KUOLLEIDEN LAAKSOSSA."
3320 R$(3)=R$(2)
3330 R$(4)="SYNKASSÄ LUOLASSA. VESI

```

VALUU HI- TAASTI ALAS YHTÄ SEINÄÄ
 PITKIN."
 3340 R\$(5)="PIENESSÄ HAUTAKAMMIOSSA.
 LATTIA ONKOVAKSIJOLJETTUA MAATA."
 3350 DATA 2,0,0,0,3,1,9,0,6,2,5,4,0,
 0,3,88,0,0,88,3
 3360 R\$(6)=R\$(2)
 3370 R\$(7)="MATALASSA, KALLIOON LOUH
 ITUSSA HAUTAKAMMIOSSA."
 3380 R\$(8)=R\$(2)
 3390 R\$(9)="SUURESSA, PIMEASSÄ HAUTAK
 AMMIOSSA, JOSSA ON KIVILAATTALATTIA."
 3400 R\$(10)="MATKASI PÄÄSSÄ."
 3410 DATA 8,3,7,0,0,0,0,6,0,6,0,0,0,
 0,0,2,12,0,0,0
 3420 R\$(11)="PEILIN TOISELLA PUOLELL
 A."
 3430 R\$(12)="HAUTAKAMMION SIVUHUONEE
 SSA."
 3440 R\$(13)="PUTOAMASSA SYVAÄN, LIEJU
 ISEEN KUOPPAAN, JOKA ON TEHTY HAUDANRYÖ
 STAJIAVARTEN."
 3450 R\$(14)="KAHDEN TUNNELIN RISTEYKS
 ESSÄ."
 3460 R\$(15)="PIENESSÄ ETEISKAMMIOSSA.
 OLE VARO-VAINEN. LATTIA ON LIUKAS."
 3470 DATA 37,31,30,0,0,10,14,0,0,0,0,
 0,13,38,15,12,0,0,0,14
 3480 R\$(16)="PTAHIN PYRAMIDIN LÄNSIPU
 OLELLA."
 3490 R\$(17)="AMONIN PYRAMIDIN LÄNSIPU
 OLELLA."
 3500 R\$(18)="AMONIN PYRAMIDIN ETELÄPU
 OLELLA."
 3510 R\$(19)="AMONIN PYRAMIDIN POHJOISP
 UOLELLA."
 3520 R\$(20)="AMONIN PYRAMIDIN ITÄPUOL
 ELLA."
 3530 DATA 21,23,0,5,19,18,0,0,0,0,20,
 17,23,0,20,17,19,18,0,88
 3540 R\$(21)="PTAHIN PYRAMIDIN POHJOISP
 UOLELLA."
 3550 R\$(22)="PTAHIN PYRAMIDIN ITÄPUOL
 ELLA."
 3560 R\$(23)="PTAHIN PYRAMIDIN ETELÄPU

OLELLA."

3570 R\$(24)="SOIKEASSA ETEISESSÄ."

3580 R\$(25)="PUTOAMASSA SYVÄÄN, KÄÄRM
EITÄ KIHII-SEVÄÄN KUOPPAAN..."

3590 DATA 0,0,22,16,21,23,0,0,88,19,
22,16,27,23,26,25,0,0,0,0

3600 R\$(26)="PUISESSA VAINAJIEN KAMM
IOSSA."

3610 R\$(27)="KYLMASSA HAUTAKAMMIOSSA."

3620 R\$(28)="JONKINLAISESSA TAIKAHUON
EESSA. ESINEET EIVÄT OLE SITÄ MILT

A NE N

AYTTÄ-VÄT...

3630 R\$(29)="PIENESSÄ ETEISESSÄ.

KIVILATTIA TUNTUU KYLMÄLTÄ JÄLKOJESI
ALLA."

3640 R\$(30)="SUURESSA, NELIÖMAISESSA
SISÄÄN- TULOHALLISSA."

3650 DATA 0,0,0,24,0,24,28,29,28,28,
28,27,0,0,27,0,0,0,20,88

3660 R\$(31)=R\$(14)

3670 R\$(32)="ETEISKAMMIOSSA. PUISET
SEINÄT MU- RENTUVAT KOSKETUKSESTASI."

3680 R\$(33)="MATALAKATTOISESSA VARAST
OSSA."

3690 R\$(34)="ANUBISIN KAMMIOSSA.

ILMA TUNTUU KYLMÄLTÄ."

3700 R\$(35)=R\$(13)

3710 DATA 11,34,0,32,0,33,31,0,32,0,0,
0,31,36,35,0,0,0,0,0

3720 R\$(36)=R\$(13)

3730 R\$(37)="UPAUTIN KAMMIOSSA.

ILMA ON HIUKANUTUINEN."

3740 R\$(38)="TUNNELIN MUTKASSA."

3750 R\$(39)="SOKKELOMAISESSA

LUOLARAKENNEL- MASSA."

3760 R\$(40)=R\$(39)

3770 DATA 0,0,0,0,0,11,0,38,14,0,37,
0,0,40,4,0,39,41,0,0

3780 R\$(41)=R\$(39)

3790 R\$(42)="PUTOAMASSA KALLIOLTA

SUORAAN KUOL-LEIDEN LAAKSOON..."

3800 R\$(43)=R\$(39)

3810 R\$(44)=R\$(39)

```

3820 DATA 40,0,42,43,0,0,0,0,44,0,41,
0,0,43,0,0
3830 DATA 0,0,0,0,0,27,0,33,26,44,29,
33,0,0,23,0,5,5,0
3840 DATA 0,30,7,9,0,20,10,15,4
3850 DATA VIITTA,SUURI RAUTA-AVAIN,
HAALISTUNUT PERGAMENTTI,TYHJÄ
3860 DATA RAUTATANKO,LAATIKKO,
NAHKAINEN SANKO
3870 DATA TIKARI,KIVIUURNA,SAVIRUUKKU
3880 DATA PULLO PALSAMOINTINESTETTÄ,
TYHJÄ,TYHJÄ,HIEKKAA
3890 DATA PYRAMIDI,SUURI PUUOVI,
KIIVINEN ARKKU
3900 DATA HAUTAKAMMIO,LUOLA,
ISOKOKOINEN PEILI,KIIVINEN ARKKU
3910 DATA KIIVINEN ARKKU,MUUMIO,PATSAS,
MERKKI,SUURI KIVI,SUURI KIVI
3920 DATA "POHJOINEN ","ETELÄ ",
"ITÄ ","LÄNSI "
3930 GOTO 230
3940 REM AUTA
3950 IF CP<>4 THEN 3980
3960 PRINT "SEIKKAILUT SAATTAVAT OLLA
HYVIN      LIIKUTTAVIA KOKEMUKSIA!"
3970 GOTO 1000
3980 IF CP<>5 AND CP<>7 AND CP<>9
THEN 4010
3990 PRINT "SESAM AUKENE!"
4000 GOTO 1000
4010 IF CP<>23 THEN 4040
4020 PRINT "HIEKKA NÄYTTÄÄ PEHMEÄLTÄ."
4030 GOTO 1000
4040 IF CP<>20 THEN 4070
4050 PRINT "MITÄ LUKI PERGAMENTISSA?"
4060 GOTO 1000
4070 IF CP<>30 THEN 4100
4080 PRINT "ONKO SINULLA VIITTA
MUKANASI?"
4090 GOTO 1000
4100 IF CP<>16 AND CP<>21 AND CP<>22
AND CP<>19 AND CP<>17 AND CP<>18 THEN 4130
4110 PRINT "EHKÄPÄ SISÄÄNKÄYNTI ON
TOISELLA      PUOLELLA?"


```



```
4120 GOTO 1000
4130 PRINT "ÄLÄ UNOHDA TUTKIA
      ESINEITÄ."
4140 GOTO 1000
4150 REM PELIN TALLENNUS NAUHALLE
4160 CLS
4170 LOCATE 0,6
4180 LINE INPUT "PAINA ENTER, KUN
      KASSETTI ON VALMIINA";W$
4190 LOCATE 0,9
4200 PRINT "TALLENNUS..."
4210 OPEN "KAS:FAAR" FOR OUTPUT AS 1
4220 FOR J=1 TO NN
4230 PRINT#1,OB(J)
4240 PRINT#1,OL(J)
4250 NEXT J
4260 PRINT#1,M(20,4)
4270 PRINT#1,M(4,4)
4280 PRINT#1,M(5,3)
4290 PRINT#1,M(23,1)
4300 PRINT#1,CP
4310 CLOSE
4320 GOTO 1000
4330 REM PELIN LATAUS NAUHALTA
4340 CLS
4350 LOCATE 0,6
4360 LINE INPUT "PAINA ENTER, KUN
      KASSETTI ON VALMIINA";W$
4370 LOCATE 0,9
4380 PRINT "LATAUS..."
4390 OPEN "KAS:FAAR" FOR INPUT AS 1
4400 FOR J=1 TO NN
4410 INPUT#1,OB(J)
4420 INPUT#1,OL(J)
4430 NEXT J
4440 INPUT#1,M(20,4)
4450 INPUT#1,M(4,4)
4460 INPUT#1,M(5,3)
4470 INPUT#1,M(23,1)
4480 INPUT#1,CP
4490 CLOSE
4500 GOTO 1000
4510 REM PELIN LOPPU
4520 CLS
```

```
4530 LOCATE 5,9
4540 PRINT "L O P P U      T U L I !"
4550 LOCATE 0,20
4560 INPUT "UUSI YRITYS";Q$
4570 IF LEFT$(Q$,1)="K" THEN RUN
4580 PRINT:INPUT "EIKÖ EDES YHTÄ YRITYSTÄ";Q$
4590 IF LEFT$(Q$,1)<>"E" THEN RUN
4600 GOTO 4700
4610 CLS
4620 LOCATE 3,8
4630 PRINT "<<<  AARRE ON SINUN!!  >>>"
4640 LOCATE 0,20
4650 GOTO 4700
4660 REM VIIIVE
4670 FOR J=1 TO 2000
4680 NEXT J
4690 RETURN
4700 END
```

PIILOSANA



MSX-tietokoneesi tekee antamistasi sanoista piilosanoja muuttamalla kirjainten järjestystä.

Kuinka sanoja ratkotaan

Kaikki on tietysti hyvin helppoa. Sinun tarvitsee vain keksiä mikä sana kulloinkin on kyseessä ja kirjoittaa se tietokoneeseen. Kirjoita LUOVUN, jos et tiedä jotain piilosanaa ja SEIS, kun haluat lopettaa kokonaan.

Ohjelmointineuvoja

Sanat piilosanoja varten ovat DATA-lausekkeissa riveillä 1040-1140. Sanoja voidaan helposti vaihtaa halutun vaikeusasteen ja kokemuspiirin mukaan. Varmista, että vain yksi sana on kunkin muodostettavan piilosanan oikea vastaus, muuten saat ehkä kuulla kunniasi muiden taholta.

DATA-tietoja on helppo vaihtaa; riittää, kun vain poistat sanoja DATA-lausekkeista ja lisäät tilalle uusia sanoja. Muista kuitenkin, että jos lisäät sanojen kokonaismäärää, sinun täytyy suurentaa rivillä 190 olevaa WD\$:n tilavarausta vastavasti.

Tietoa ohjelmasta

Rivin n:o	
190	Sanataulukon tilavaraus
230-270	Lukee sanat taulukkoon WD\$.
290	Valitsee sanan satunnaisesti.
320-360	Sotkee kirjainten järjestyksen.
390-430	Kirjoittaa piilosanan.
490-610	Odottaa vastausta.
650	Tarkistaa, onko vastaus oikein.
680-710	Väärä vastaus
720	Oikea vastaus
760-870	Piirtää hymyilevät kasvot.
890-940	Antaa muuttujille alkuarvot.
980-1000	Toteaa luovutuksen.
1010-1030	Viiverutiini
1040-1140	Sanojen DATA-tiedot
1170-1220	Ilmoittaa tuloksen ja lopettaa pelin.

Ohjelma: PIIL-JS

```

100 REM *****
110 REM *
120 REM * PIILSANA *
130 REM *
140 REM *****
150 CLS
160 R=RND(-TIME)
170 SCREEN 2
180 OPEN "GRP:" AS 1

```

```

190 DIM WD$(100),M$(12)
200 COLOR 13,1,1
210 J=1
220 REM LUKEE DATA-TIEDOT
230 READ WD$(J)
240 IF WD$(J)="LOPPU" THEN 270
250 J=J+1
260 GOTO 230
270 J=J-1
280 REM SANAN VALINTA
290 R=INT(RND(1)*J)+1
300 A$=WD$(R)
310 TR=TR+1
320 FOR I=1 TO LEN(A$)
330 R=INT(RND(1)*12)+1
340 IF M$(R)<>" " THEN 330
350 M$(R)=MID$(A$,I,1)
360 NEXT I
370 CLS
380 REM SANAN KIRJOITUS
390 PRESET(80,48)
400 FOR I=1 TO 12
410 IF M$(I)=" " THEN 430
420 PRINT#1,M$(I);
430 NEXT I
440 PRESET(40,96)
450 COLOR 10,1
460 PRINT#1,"MIKÄ SANA ON KYSEESSÄ?"
470 COLOR 3,1
480 REM SANAN SYÖTTÖ TIETOKONEELLE
490 G$=""
500 T$=INKEY$:IF T$="" THEN 500
510 IF T$=CHR$(13) THEN 620
520 IF T$<>CHR$(8) AND T$<>CHR$(29) THEN 590
530 IF LEN(G$)=0 THEN 500
540 G$=LEFT$(G$,LEN(G$)-1)
550 COLOR 1,1
560 PRESET(80,120):PRINT#1,STRING$(12,219)
570 COLOR 3,1
580 GOTO 600
590 G$=G$+T$
600 PRESET(80,120):PRINT#1,G$
610 GOTO 500
620 PRESET(80,144)

```


```

630 COLOR 6,1
640 REM TARKISTAA VASTAUKSEN
650 IF G$=A$ THEN 720
660 IF G$="LUOVUN" THEN 980
670 IF G$="SEIS" THEN 1170
680 PLAY "O2;BAFEDCB"
690 PRINT#1,"YRITÄ UUDELLEEN!"
700 GOSUB 1010
710 GOTO 370
720 PRINT#1,"OIKEIN!"
730 PLAY "T230;O5;ABCDEF;O6;ABCDEF"
740 COLOR 11,1,1
750 REM REUNUKSEN PIIRTO
760 FOR I=16 TO 239 STEP 8
770 PRESET(I,0)
780 PRINT#1,CHR$(1);CHR$(65);
790 PRESET(I,183)
800 PRINT#1,CHR$(1);CHR$(65);
810 NEXT I
820 FOR I=8 TO 183 STEP 8
830 PRESET(231,I)
840 PRINT#1,CHR$(1);CHR$(65)
850 PRESET(16,I)
860 PRINT#1,CHR$(1);CHR$(65)
870 NEXT I
880 REM MUUTTUJIEN NOLLAUS
890 NC=NC+1
900 A$=""
910 G$=""
920 FOR I=1 TO 12
930 M$(I)=""
940 NEXT I
950 FOR DE=1 TO 2000:NEXT DE
960 GOTO 290
970 REM LUOPUMINEN
980 PRINT#1,A$
990 GOSUB 1010
1000 GOTO 900
1010 FOR DE=1 TO 2000
1020 NEXT DE
1030 RETURN
1040 DATA KALA,KAKI,HIIRI,KISSA,TIIKERI,
KENGURU,ANTILOOPPI,LEIJONA
1050 DATA TIETOKONE,OHJELMA,SANA,BITTI,TAVU,
RIVI,MERKKI,NUMERO,MUISTI

```

1060 DATA IDEA, AJATUS, AATOS, MIETE, MIELIPIDE,
VALÄHDYS, INTUITIO
1070 DATA TAMMI, PÄJU, LEPPÄ, MÄNTY, KUUSI,
KOIVU, HAAPA, VAAHTERA, POPPELI
1080 DATA TILLI, PAPU, HERNE, NAURIS, LANTTU,
SIPULI, KORI, KIVI, PALLO, OVI, IKKUNA
1090 DATA KYNÄ, PAPERI, KUMI, LYIJY, HARPPI,
VIIVOTIN, MUSTE, HELIKOPTERI
1100 DATA TELEVISIO, PUHELIN, VIDEO, KASETTI,
ANTENNI, JOHTO, KUVA, MAINOS, PELI, FILMI
1110 DATA ANKKA, HANHI, VUOHI, LAMMAS, SONNI,
HÄKÄ, LEHMA, SIKA, HEVONEN
1120 DATA KIELI, MATKAL, KAUPUNKI, MAANTIE,
KARTTA, BENSIINI, AUTO, LENTOKONE
1130 DATA KORTTI, LIPPU, METRO, KOTI, KANA, KOLO,
VÄRI, JÄRVI, MERI, JOKI, KUOPPA
1140 DATA VESI, LUMI, JÄÄ, TULI, ILMA, SUMU, RAE,
HÄKÄ, KAASU
1150 DATA LOPPU
1160 REM PELIN LOPPU
1170 SCREEN 0:COLOR 15,4,4
1180 CLOSE
1190 LOCATE 0,7
1200 PRINT " ARVASIT"NC"KERTAA OIKEIN":PRI
NT:PRINT" SANOJA OLI"TR-1"KPL"
1210 LOCATE 0,21
1220 END

PILOTTI



Pilotti on lentosimulaattoripeli yhdelle pelaajalle. Pelissä on kaksi lentokenttää, nousua ja laskeutumista varten. Näytössä näkyy myös maastokartta, jossa kummatkin lentokentät, läntinen ja itäinen, näkyvät pieninä. Kiitorata tulee automaattisesti näkyviin lähestyessäsi kenttää silloin, kun olet alle 2 000 jalan korkeudessa. Maastokartta tulee samoin automaattisesti näyttöön, kun lähdet kiitorata-alueelta. Lentokoneen sijainti näkyy näytössä vilkkuvana lentokoneen kuvana. Lentosuunta otetaan SUU-lukemana. Kuvaruudun alaosassa on kojelauta, josta saat lentoon tarvittavat tiedot.

Mittarit

SUU	Lentosuunta asteina. Pohjoinen on 0, itä 90, etelä 180 ja länsi on 270.
PER	Sivuperäsimen asento, negatiivinen vasemmalle, positiivinen oikealle
BEN	Jäljellä olevan polttoaineen määrä
NOP	Ilmanopeus eteenpäin
TEH	Moottorin teho
KOR	Korkeus
NON tai NOL	Nousunopeus tai laskunopeus. Laskeutumista ilmoittaa myös punainen valo.
SII	Laskusiivekkeiden asento. 0 on täysin kiinni, 30 täysin auki
LASK YLÖS tai LASK ALAS	Laskutelineiden asento

Kojelaudan oikealla puolella oleva neliö on keinohorisontti. Kun punainen viiva on alhaalla, osoittaa koneen nokka ylöspäin ja kun viiva on ylhäällä, osoittaa nokka alaspäin. Kallistus on ilmaistu samalla tavalla; kun punainen viiva kulkee vasemmasta alanurkasta oikeaan ylänurkkaan, olet kallistunut oikealle.

Kuinka peliä pelataan

ESDX	Näillä näppäimillä matkitaan ohjaussauvan liikkeitä. Näppäimellä E saat nokan alas ja näppäimellä X ylös. Samalla tavalla käytetään näppäimiä S ja D koneen kallistamiseen. Kallistus aiheuttaa sivusuuntaisen voiman, joka kääntää konetta kallistuksen suuntaan. Pidä näppäintä aina alhaalla, kunnes huomaat koneen reagoivan.
O P	Moottorin tehoa vähennetään O:lla ja lisätään P:llä.

- W R** Sivuperäsintä käännetään R:llä oikeaan ja W:llä vasempaan. Sivuperäsिमellä voidaan ohjata konetta maassa. Se auttaa myös kääntymistä ilmassa.
- A F** Laskusiivekkeet suljetaan A:lla ja avataan F:llä. Laskusiivekkeiden ollessa täysin suljettuina (0), on sakkausnopeus, jolla menetät koneen hallinnan, noin 80 solmua. Vastaavasti, kun laskusiivekkeet ovat täysin auki (30), kone sakkaa vasta 60 solmun nopeudessa. Älä aukaise laskusiivekkeitä suurilla nopeuksilla, sillä silloin ilmanvastus kasvaa ja siivekkeet saattavat vahingoittua.
- G** Laskutelineitä käytetään näppäimellä G. Pidä laskutelineet yläasennossa muulloin, paitsi nousussa ja laskussa. Jos laskutelineet ovat alhaalla suurilla nopeuksilla, on seurauksena ilmanvastuksen kasvu ja ehkä koneen vaurioituminenkin.

Muista aina pitää näppäintä alhaalla niin kauan, kunnes huomaa sen vaikuttavan.

Maaston korkeus

Vihreän alueen korkeus on alle 100 jalkaa (30 m).
Punaisen alueen korkeus on alle 1 000 jalkaa (300 m).
Musta alue taas on alle 4 000 jalkaa (1 200m).

Tarkkaile korkeuttasi ylittäessäsi näitä alueita, muuten voit yllättäen törmätä maahan.

Vaihtoehdot

Peli voidaan aloittaa kolmella eri tavalla: nousulla, laskulla tai lento-osalla. Ensimmäisillä kerroilla kannattaa valita vain lento-osa, koska osatessasi käsitellä konetta ilmassa voit sitten helposti siirtyä harjoittelemaan laskeutumista, joka on lentämisen vaikein vaihe. Syöksyt varmasti maahan monta kertaa, ennen kuin opit laskeutumaan oikein.

Pelissä on myös mahdollisuus ottaa tuulen vaikutus mukaan. Harjoittele aluksi tyynellä ilmalla, sillä vaatii todella taitoa yritt-

tää navigoida ja laskeutua todellisella myrskytuulella.

Laskeutuminen

Molempia lentokenttiä on parasta lähestyä idästä, jolloin suuntalukema on 270. Korkeuden pitää olla alle 2000 jalkaa. Kun lähestyt kenttäaluetta, tulee kiitorata automaattisesti kuvatuudun yläosaan maastokartan paikalle. Vähennä ilmanopeutta maakosketusta varten ja pienennä moottoritehoa välittömästi kosketuksen jälkeen, jotta ehtisit pysähtyä ennen kiitoradan loppupäätä. Laskusiivekkeet voivat olla täysin auki laskeutumisen loppuvaiheessa, jotta sakkausnopeus pienenee.

Lentämisen aerodynamiikka on erittäin monimutkaista ja niinpä yhden tekijän muuttaminen vaikuttaa myös muihin tekijöihin. Esimerkiksi kallistaminen vaikuttaa myös siipien nostovoimaan aiheuttaen koneen vajoamisen alaspäin.

Varo ääri nopeuksia. Sekä hyvin nopealla että hyvin hitaalla nopeudella saatat menettää koneen hallinnan. Samoin ohjainsauvaa (E S D X) ei saa työntää ääriasentoihin ennen kuin pystyy mielestään varmasti selviytymään silmukoista ja vaakakierteistä.

Älä kuvittele, että lentäminen on helppoa. Siihen tarvitaan paljon harjoittelua ja erityisesti laskeutumisvaihe vaatii paljon. Jos osaat nousta ilmaan ja lentää toiselle kentälle sekä tulla jälleen takaisin, voit pitää itseäsi hyvänä lentäjänä. Polttoainetäydennystä saat aina laskeuduttuasi.

Yhteenveto hallintalaitteista

E	Nokka alas
X	Nokka ylös
S	Kallistuminen (ja kääntyminen) vasemmalle
D	Kallistuminen (ja kääntyminen) oikealle
O	Tehon vähentäminen
P	Tehon lisääminen
R	Sivuperäsin oikealle
W	Sivuperäsin vasemmalle
F	Laskusiivekkeet auki
A	Laskusiivekkeet kiinni
G	Laskutelineet

Tietoja ohjelmasta

Rivin n:o	
200-270	Kirjoittaa valikon.
280-300	Odottaa toimintavaihtoehtoa.
310-340	Odottaa tuulitietoa.
370	Vaihtoehto-aliohjelmat
390-430	Pääohjelmasilmut
450-740	Mittarinlukemien näyttörutiinit
760-790	Tarkistaa polttoaineen määrän.
800-850	Tarkistaa nopeuden.
860-990	Tarkistaa korkeuden.
910-980	Tarkistaa törmäyksen.
1000-1140	Tarkistaa ja ilmoittaa nousunopeuden.
1170-1310	Sakkausvaikutukset
1360-1450	Nopeus nousua varten liian suuri
1480-1510	Odottaa suraavaa komentoa.
1530-1820	Ohjaussauvan toiminnot
1840-1950	Silmukka ja vaakakierre
1960-2030	Tarkistaa tehon.
2040-2100	Tarkistaa laskusiivekkeet.
2110-2170	Tarkistaa sivuperäsimen.
2190-2270	Vaihtaa laskutelineiden asentoa.
2300-2380	Laskee ja näyttää uuden suunnan.
2390-2420	Laskee tuulen vaikutuksen.
2440-2450	Laskee uuden sijainnin.
2830-3030	Muuttaa karttaa näytössä.
3060-3280	Valmistelee taustan.
3300-3440	Tulostaa kojelaudan alkuarvoissaan.
3450-3680	Alkukuva
3700-3810	Tulostaa maastokartan näytön yläosaan.
3820-3930	DATA-tiedot maastokarttaa varten
3960-4010	Alkuarvot tuulelle
4090-4140	Törmäyksen ääni
4200-4240	Alkuarvot muuttujille uuteen peliin
4270-4310	Tarkistaa laskutelineet laskussa.
4330-4380	Tarkistaa kallistuksen laskussa.
4390-4460	Onnistunut laskeutuminen
4470-4550	Valmis nousuun
4570-5160	Alkuarvot lennon alkamiseksi

Ohjelma: PILO-SJ

```

100 REM *****
110 REM *
120 REM *   PILOTTI   *
130 REM *
140 REM *****
150 REM
160 X=RND(-TIME)
170 OPEN "GRP:" AS 1
180 REM OTSIKKORUUTU JA ERI VAIHTOEHDOT
190 CLS
200 SCREEN 0
210 COLOR 1,15,1
220 LOCATE 3,5:PRINT "PILOTTI"
230 LOCATE 3,8:PRINT "1. Nousu"
240 LOCATE 3,10:PRINT "2. Laskeutuminen"
250 LOCATE 3,12:PRINT "3. Lento-osuus"
260 LOCATE 3,18
270 PRINT "Valinta?"
280 Q$=INKEY$:IF Q$="" THEN 280
290 IF Q$>"3" OR Q$<"1" THEN 280
300 Q=VAL(Q$)
310 LOCATE 3,20
320 PRINT "Tuulen vaikutus?"
330 A$=INKEY$:IF A$="" THEN 330
340 IF A$="K" THEN WN=1
350 SCREEN 2
360 CLS
370 ON Q GOSUB 4570,4690,4840
380 REM PAAOHJELMASILMUKKA
390 GOSUB 1480
400 GOSUB 760
410 GOSUB 1480
420 GOSUB 2300
430 GOTO 390
440 REM NÄYTTÖRUTIINIT
450 IF LEN(P$)>1 THEN 480
460 P$=" "+P$
470 GOTO 620
480 IF LEN(P$)>2 THEN 510
490 P$=" "+P$
500 GOTO 620
510 IF LEN(P$)>3 THEN 540
520 P$=" "+P$
530 GOTO 620

```

```

540 IF LEN(P$)>4 THEN 620
550 P$=" "+P$
560 GOTO 620
570 IF LEN(P$)>1 THEN 600
580 P$=" "+P$
590 GOTO 620
600 IF LEN(P$)>2 THEN 620
610 P$=" "+P$
620 IF LEFT$(P$,1)=CHR$(219) OR P$=CHR$(43)
THEN 680
630 IF MID$(P$,2,1)>="0" AND MID$(P$,2,1)<=
"9" THEN P$=RIGHT$(P$,LEN(P$)-1)
640 C=C-1
650 GOSUB 710
660 C=C+1
670 COLOR 11
680 PRESET((C-1)*8,(R-1)*8)
690 PRINT#1,P$
700 RETURN
710 PRESET((C-1)*8,(R-1)*8)
720 COLOR 1
730 PRINT#1,STRING$(LEN(P$)+2,219)
740 RETURN
750 REM POLTTORINEEN MÄÄRÄN,
NOPEUDEN JA KORKEUDEN TARKISTUS
760 FU=FU-PW*.013
770 IF FU<=0 THEN FU=0:PW=0
780 P$=STR$(INT(FU))
790 R=15:C=4:GOSUB 570
800 VE=(PW*1.14-B2*44.978-VE*((FL>0)*
(VE>100))*FL/150-.1*VE*((GR=0)*(VE>100))+VE
/2*(VE<10)+VE*4)/5
810 IF VE<20 THEN 830
820 VE=VE+RND(1)*4-2
830 IF VE<0 THEN VE=0
840 P$=STR$(INT(VE))
850 R=18:C=4:GOSUB 570
860 AL=AL+R0/10
870 IF AL<0 THEN AL=0
880 P$=STR$(INT(AL))
890 R=16:C=16:GOSUB 450
900 REM TÖRMÄYKSEN TARKISTUS
910 IF AL<1 AND TS<>6 AND TS<>10 THEN 4090
920 IF AL<1 AND GR=1 THEN 4270
930 IF AL<1 AND ABS(B3-B1)>3 THEN 4270

```

```

940 IF AL<1 AND RO<-320 THEN 4090
950 IF AL<1 AND VE<1 AND (TS=6 OR TS=10)
AND RO<0 AND RO>-320 THEN 4270
960 IF AL<100 AND PS=12 AND WM=2 THEN 4090
970 IF AL<1000 AND PS=6 AND WM=2 THEN 4090
980 IF AL<4000 AND PS=1 AND WM=2 THEN 4090
990 REM NOUSUNOPEUS
1000 RO=((((B2*62.4-ABS(B3-B1)*22.28)*31.33)
-(220*VE/150)*(AL>5)*SGN(RO)-VE*B2*2*
(B2<-1)+RO*2)/3
1010 GOSUB 1030
1020 GOTO 1170
1030 IF AL<5 THEN 1060
1040 RO=RO+RND(1)*8-4
1050 REM NOUSUNOPEUDEN TULOSTUS
1060 IF RO>=0 THEN 1110
1070 P$=STR$(ABS(INT(RO)))
1080 PRESET(17*8,17*8):COLOR 1:PRINT#1,
STRING$(3,219)
1090 PRESET(15*8,17*8):COLOR 11:PRINT#1,
"NOL":COLOR 6:PRINT#1,CHR$(219)
1100 GOTO 1140
1110 PRESET(17*8,17*8):COLOR 1:PRINT#1,
STRING$(3,219)
1120 PRESET(15*8,17*8):COLOR 11:PRINT#1,
"NON":COLOR 1:PRINT#1,CHR$(219)
1130 P$=STR$(INT(RO))
1140 R=19:C=16:GOSUB 450
1150 RETURN
1160 REM SAKKAUS
1170 IF VE>(80-FL*.66) OR AL<5 THEN 1320
1180 RO=RO-ABS(RO*RND(1)/2)-RND(1)*320*
(80-VE)
1190 GOSUB 1030
1200 PLAY "ACACACAC"
1210 IF RND(1)>.5 THEN 1240
1220 K=69
1230 GOTO 1560
1240 IF RND(1)>.5 THEN 1270
1250 K=83
1260 GOTO 1560
1270 IF RND(1)>.5 THEN 1300
1280 K=68
1290 GOTO 1560

```



```

1300 K=88
1310 GOTO 1560
1320 IF AL<20000 OR R0<0 THEN 1350
1330 IF AL=0 THEN R0=0:GOTO 1350
1340 R0=R0*15000/AL
1350 IF VE>350 THEN 1200
1360 IF VE<140 OR AL>0 THEN 1460
1370 GOSUB 4020
1380 REM NOUSUNOPEUS LIIAN SUURI
1390 P$="NOPEUS"
1400 R=15:C=9:GOSUB 620
1410 P$="LIIAN"
1420 R=17:C=9:GOSUB 620
1430 P$="SUURI"
1440 R=19:C=9:GOSUB 620
1450 GOTO 4090
1460 RETURN
1470 REM ODOTTAA OHJETTA NÄPPÄIMISTÖLTÄ
1480 K$=INKEY$
1490 IF K$="" THEN 2280
1500 K=ASC(K$)
1510 PLAY "06;C"
1520 REM OHJAINSAUVAN TOIMINNOT
1530 IF K<>69 AND K<>88 AND K<>83 AND
K<>68 THEN 1960
1540 IF AL<5 AND VE<(80-FL*.666) THEN 1960
1550 IF K<>88 AND AL<4 THEN 1960
1560 IF B1>6 OR B1<-5 THEN 1580
1570 R=17+B1:C=24:P$=STRING$(2,219):
COLOR 15:GOSUB 620
1580 IF B2>6 OR B2<-5 THEN 1600
1590 R=17+B2:C=26:P$=STRING$(2,219):
COLOR 15:GOSUB 620
1600 IF B3>6 OR B3<-5 THEN 1620
1610 R=17+B3:C=28:P$=STRING$(2,219):
COLOR 15:GOSUB 620
1620 IF K<>69 THEN 1670
1630 B1=B1-1
1640 B2=B2-1
1650 B3=B3-1
1660 GOTO 1790
1670 IF K<>88 THEN 1720
1680 B1=B1+1
1690 B2=B2+1

```

```

1700 B3=B3+1
1710 GOTO 1790
1720 IF K<>83 THEN 1760
1730 B1=B1-1
1740 B3=B3+1
1750 GOTO 1790
1760 IF K<>68 THEN 1790
1770 B1=B1+1
1780 B3=B3-1
1790 IF B2<10 THEN 1840
1800 B1=-10
1810 B2=-10
1820 B3=-10
1830 REM SILMUKKA JA VAAKAKIERRE
1840 IF ABS(B3-B1)<18 THEN 1860
1850 SWAP B1,B3
1860 IF ABS(B3-B1)<8 THEN 1900
1870 B1=B1-1
1880 B2=B2-1
1890 B3=B3-1
1900 IF B1>6 OR B1<-5 THEN 1920
1910 R=17+B1:C=24:COLOR 6:P#=CHR$(219)+CHR$(219):GOSUB 620
1920 IF B2>6 OR B2<-5 THEN 1940
1930 R=17+B2:C=26:P#=CHR$(219)+CHR$(219):COLOR 6:GOSUB 620
1940 IF B3>6 OR B3<-5 THEN 1960
1950 R=17+B3:C=28:P#=CHR$(219)+CHR$(219):COLOR 6:GOSUB 620
1960 IF K<>79 OR PW<10 THEN 1990
1970 PW=PW-10
1980 GOTO 2020
1990 IF K<>80 OR PW>210 THEN 2040
2000 IF FU=0 THEN 2040
2010 PW=PW+10
2020 P#=STR$(PW)
2030 R=21:C=4:GOSUB 570
2040 IF K<>65 OR FL<5 THEN 2070
2050 FL=FL-5
2060 GOTO 2090
2070 IF K<>70 OR FL>25 THEN 2110
2080 FL=FL+5
2090 P#=STR$(FL)
2100 R=22:C=16:GOSUB 570
2110 IF K<>82 OR RD>1 THEN 2140

```

```

2120 RD=RD+1
2130 GOTO 2160
2140 IF K<>87 OR RD<-1 THEN 2190
2150 RD=RD-1
2160 P$=" "+STR$(RD)+" "
2170 R=12:C=17:GOSUB 620
2180 REM LASKUTELINEIDEN LIIKUTUS
2190 IF K<>71 THEN 2280
2200 IF AL<4 THEN 2280
2210 IF GR=1 THEN 2250
2220 GR=1
2230 P$="YLös"
2240 GOTO 2270
2250 GR=0
2260 P$="ALAS"
2270 R=23:C=10:GOSUB 620
2280 RETURN
2290 REM LASKEE UUDEN LENTOSUUNNAN
2300 HD=HD-(B3-B1)*3-RD*2*(VE>2)
2310 M2=M2-1
2320 IF AL>0 OR VE<2 THEN 2340
2330 HD=HD+RD*15
2340 IF HD>359 THEN HD=HD-360
2350 IF HD<0 THEN HD=360+HD
2360 REM TULOSTAA UUDEN LENTOSUUNNAN
2370 P$=STR$(HD)
2380 R=12:C=9:GOSUB 570
2390 IF WN=0 THEN 2440
2400 REM LASKEE TUULEN VAIKUTUKSET
2410 IR=COS(WD/57.296)*WS/160
2420 IC=SIN(WD/57.296)*WS/160
2430 REM LASKEE UUDEN SIJAINNIN
2440 PR=PR-(VE/100)*COS(HD/57.296)+IR
2450 PC=PC+(VE/100)*SIN(HD/57.296)-IC
2460 IF WM=1 THEN 2600
2470 IF PC>40 AND PC<310 AND PR>15 AND
PR<100 THEN 2510
2480 PR=0I
2490 PC=0C
2500 REM NÄYTTÄÄ UUDEN SIJAINNIN
2510 PRESET((INT(OC/10+.5)-1)*8,
(INT(OI/10+.5)-1)*8)
2520 COLOR PS
2530 PRINT#1,CHR$(219)

```

```

2540 PS=POINT((INT(PC/10+.5)-1)*8,
(INT(PR/10+.5)-1)*8)
2550 PRESET((INT(PC/10+.5)-1)*8,
(INT(PR/10+.5)-1)*8)
2560 COLOR 15
2570 PRINT#1,CHR$(43)
2580 OI=PR:OC=PC
2590 GOTO 2700
2600 PRESET((INT(TC+.5)-1)*8,(INT(TR+.5)-1)*8)
2610 COLOR TS
2620 PRINT#1,CHR$(219)
2630 TC=TC+(VE/120)*SIN(HD/57.296)+IC*(AL<1)
*VE/80+IC*(AL>0)
2640 TR=TR-(VE/120)*COS(HD/57.296)-IR*(AL<1)
*VE/80-IR*(AL>0)
2650 IF TR>10 OR TR<2 OR TC>30 OR TC<4 THEN
2960
2660 TS=POINT((INT(TC+.5)-1)*8,
(INT(TR+.5)-1)*8)
2670 R=INT(TR+.5):C=INT(TC+.5)
2680 COLOR 13
2690 P#=CHR$(43):GOSUB 620
2700 IF WM=2 AND AL<2000 AND PC<295 AND
PC>267 AND PR<55 AND PR>41 THEN 2710 ELS
E 2830
2710 IF WM=1 THEN 3010
2720 IF M2>0 THEN 3010
2730 GOSUB 3630
2740 GOSUB 3450
2750 WM=1
2760 M2=3
2770 TR=PR-35
2780 TR=TR/2
2790 TC=PC-264
2800 TS=POINT((INT(TC+.5)-1)*8,
(INT(TC+.5)-1)*8)
2810 GOTO 3010
2820 REM MUUTTAA KARTTAA
2830 IF WM=2 AND AL<2000 AND PC<85 AND
PC>58 AND PR<75 AND PR>61 THEN 2840 ELSE
3010
2840 IF WM=1 THEN 3010
2850 IF M2>0 THEN 3010
2860 GOSUB 3630
2870 GOSUB 3450

```

```

2880 WM=1
2890 M2=3
2900 TR=PR-55
2910 TR=TR/2
2920 TC=PC-55
2930 TS=POINT((INT(TC+.5)-1)*8,
(INT(TR+.5)-1)*8)
2940 GOTO 3010
2950 IF M2>0 THEN 3010
2960 IF WM=2 THEN 3010
2970 GOSUB 3700
2980 WM=2
2990 M2=8
3000 PS=12
3010 IF M2<>0 OR WM<>2 THEN 3040
3020 R=7:C=7:P#=CHR$(219):COLOR 11:GOSUB 620
3030 R=5:C=28:COLOR 11:P#=CHR$(219):GOSUB 620
3040 RETURN
3050 REM TYHJENTÄÄ KUVARUUDUN ALAOSAN
3060 P#=STRING$(30,219):COLOR 1:C=3
3070 FOR R=11 TO 24
3080 GOSUB 620
3090 NEXT R
3100 C=23:P#=STRING$(8,219):COLOR 15
3110 FOR R=12 TO 23
3120 GOSUB 620
3130 NEXT R
3140 R=11:C=22:COLOR 4:P#=STRING$(10,219):
GOSUB 620
3150 COLOR 4:P#=CHR$(219):C=22
3160 FOR R=12 TO 24
3170 GOSUB 620
3180 NEXT R
3190 C=31
3200 FOR R=12 TO 24
3210 GOSUB 620
3220 NEXT R
3230 R=24:C=22:P#=STRING$(10,219):COLOR 4:
GOSUB 620
3240 R=17:C=24:P#=STRING$(6,219):COLOR 6:
GOSUB 620
3250 P#=STRING$(7,219):C=8:COLOR 15
3260 FOR R=14 TO 20
3270 GOSUB 620
3280 NEXT R

```

```

3290 REM TULOSTAA KOJELAUDAN
ALKU-          ARVOISSAAN
3300 P$="SUU":COLOR 11:R=12:C=5:GOSUB 620
3310 P$="BEN":R=14:C=4:GOSUB 620
3320 P$="NOP":R=17:C=4:GOSUB 620
3330 P$="TEH":R=20:C=4:GOSUB 620
3340 P$="LASK ALAS":R=23:C=5:GOSUB 620
3350 P$="PER":R=12:C=13:GOSUB 620
3360 P$="KOR":R=15:C=16:GOSUB 620
3370 P$="NON  ":R=18:C=16:GOSUB 620
3380 P$="SII":R=21:C=16:GOSUB 620
3390 IF WN=0 THEN 3440
3400 COLOR 1
3410 P$="TUULI":R=15:C=9:GOSUB 5130
3420 P$=STR$(WS)+" KN":R=17:C=7:GOSUB 5130
3430 P$=STR$(WD)+" AST":R=19:C=7:GOSUB 5130
3440 RETURN
3450 COLOR 1:P$=STRING$(30,219):R=1:C=3:
GOSUB 620
3460 P$=CHR$(219):C=3
3470 FOR R=2 TO 11
3480 GOSUB 620
3490 NEXT R
3500 C=32
3510 FOR R=2 TO 11
3520 GOSUB 620
3530 NEXT R
3540 C=2
3550 FOR R=1 TO 24:GOSUB 620
3560 NEXT R
3570 R=4:C=7:COLOR 12:P$=STRING$(11,219):
GOSUB 620
3580 R=5:C=7:COLOR 10:GOSUB 620
3590 R=6:C=7:P$=STRING$(11,219):GOSUB 620
3600 R=7:C=7:P$=STRING$(11,219):GOSUB 620
3610 R=8:C=7:COLOR 12:GOSUB 620
3620 RETURN
3630 P$=STRING$(28,219):C=4
3640 COLOR 15
3650 FOR R=2 TO 10
3660 GOSUB 620
3670 NEXT R
3680 RETURN
3690 REM MAASTOKARTTA

```

```

3700 RESTORE
3710 FOR J=2 TO 10
3720 FOR K=4 TO 31
3730 READ CH,RP
3740 IF CH=97 THEN COLOR 1
3750 IF CH=112 THEN COLOR 12
3760 IF CH=120 THEN COLOR 6
3770 IF CH=128 OR CH=129 THEN COLOR 10
3780 P$=STRING$(RP,219)
3790 R=J:C=K:GOSUB 620
3800 K=K+RP-1
3810 NEXT K,J
3820 REM DATA-TIEDOT MAASTOKARTTA
A          VARTEN
3830 DATA 97,2,120,5,112,2,120,2,112,3,120,
1,97,3,120,2,112,8
3840 DATA 97,1,120,2,112,10,120,1,97,6,120,2
3850 DATA 112,5,120,1,120,2,112,10,120,7,97,2
3860 DATA 120,2,112,2,120,2,97,1,112,19
3870 DATA 120,1,97,1,112,3,129,1,112,1,120,2
3880 DATA 112,8,120,2,112,8,97,1,120,1,112,7
3890 DATA 120,1,112,3,129,1,112,2,120,8,97,
5,120,3,112,6
3900 DATA 112,5,120,4,112,2,120,5,97,3,120,
2,112,7
3910 DATA 112,4,120,4,112,7,120,1,97,1,120,3
3920 DATA 112,8,112,2,120,6,112,6,120,4,112,
4,120,6,120,2
3930 DATA 97,2,120,1,112,5,120,2,112,8,120,
4,97,4
3940 RETURN
3950 REM ALKUEHTOJEN ASETUS
3960 FU=500
3970 IF WN=0 THEN 4000
3980 WS=INT(RND(1)*30)+4
3990 WD=INT(RND(1)*360)
4000 OI=PR:OC=PC
4010 RETURN
4020 P$=STRING$(7,219):C=8
4030 COLOR 15
4040 FOR R=14 TO 20
4050 GOSUB 620
4060 NEXT R
4070 RETURN
4080 REM TÖRMÄYS

```

```

4090 SOUND 0,0:SOUND 1,5:SOUND 2,0:
SOUND 3,13
4100 SOUND 4,255:SOUND 5,15:SOUND 6,30:
SOUND 7,0
4110 SOUND 8,16:SOUND 9,16:SOUND 10,16:
SOUND 11,0
4120 SOUND 12,5:SOUND 13,0
4130 FOR DE=1 TO 100:NEXT DE
4140 SOUND 12,56:SOUND 13,0
4150 GOSUB 4020
4160 P$="TÖRMÄYS":R=16:C=8:GOSUB 5130
4170 P$="MAAHAN":R=18:C=8:GOSUB 5130
4180 FOR DE=1 TO 4000:NEXT DE
4190 REM MUUTTUIJEN MOLLAAUS
4200 RO=0:PW=0:VE=0
4210 AL=0:RD=0:FL=0
4220 WN=0:WD=0:WS=0
4230 IR=0:IC=0
4240 B1=0:B2=0:B3=0
4250 GOTO 190
4260 REM TARKISTAA LASKUTELINEET
4270 IF GR=0 THEN 4330
4280 GOSUB 4020
4290 P$="LASKU-":R=15:C=8:GOSUB 5130
4295 P$="TELIN":R=16:C=8:GOSUB 5130
4300 P$="YL-":R=17:C=10:GOSUB 5130
4305 P$="HÄÄLLÄ":R=18:C=8:GOSUB 5130
4310 GOTO 4090
4320 REM TARKISTAA KALLISTUKSEN
4330 IF ABS(B3-B1)<4 THEN 4390
4340 GOSUB 4020
4350 P$="LIIKAA":R=15:C=8:GOSUB 5130
4360 P$="KALLIS-":R=17:C=8:GOSUB 5130
4370 P$="TUSTA":R=19:C=8:GOSUB 5130
4380 GOTO 4090
4390 PLAY "ECF"
4400 GOSUB 4020
4410 P$="OLET ":R=16:C=10:GOSUB 5130
4420 P$="MARSSA":R=18:C=9:GOSUB 5130
4430 FU=500
4440 RO=0
4450 FOR DE=1 TO 16000:NEXT DE
4460 GOSUB 4020
4470 PLAY "CE"

```




```

4480 P$="OLET":R=15:C=10:GOSUB 5130
4490 P$="VALMIS":R=16:C=9:GOSUB 5130
4500 P$="NOU-":R=17:C=10:GOSUB 5130
4510 P$="SUUN":R=18:C=10:GOSUB 5130
4520 P$="":R=19:C=10:GOSUB 5130
4530 FOR DE=1 TO 5000:NEXT DE
4540 GOSUB 4020
4550 GOTO 1460
4560 REM ALKUARVOT
4570 HD=90
4580 PR=62
4590 PC=55
4600 TS=10
4610 GR=0
4620 WM=1
4630 TR=6
4640 TC=8
4650 GOSUB 3960
4660 GOSUB 3060
4670 GOSUB 3450
4680 RETURN
4690 HD=270
4700 PR=70
4710 PC=79
4720 TS=15
4730 GR=0
4740 WM=1
4750 TR=6
4760 TC=30
4770 AL=400
4780 VE=110
4790 PW=100
4800 GOSUB 3960
4810 GOSUB 3060
4820 GOSUB 3450
4830 GOTO 2020
4840 HD=INT(RND(1)*360)
4850 PR=RND(1)*50+40
4860 PC=RND(1)*160+60
4870 WM=2
4880 AL=INT(RND(1)*3000)+4000
4890 VE=INT(RND(1)*50)+100
4900 PW=130
4910 GR=1
4920 GOSUB 3960

```

```
4930 GOSUB 3060
4940 GOSUB 2220
4950 P#=STRING$(30,219)
4960 COLOR 1
4970 R=1:C=3:GOSUB 620
4980 C=3:P#=CHR$(219)
4990 FOR R=2 TO 11
5000 GOSUB 620
5010 NEXT R
5020 C=32
5030 FOR R=2 TO 11
5040 GOSUB 620
5050 NEXT R
5060 C=2
5070 FOR R=1 TO 24
5080 GOSUB 620
5090 NEXT R
5100 GOSUB 3700
5110 PS=POINT(INT(PC/10+.5)*8-1,
INT(PR/10+.5)*8-1)
5120 GOTO 2020
5130 COLOR 1,15
5140 PRESET((C-1)*8,(R-1)*8)
5150 PRINT#1,P#
5160 RETURN
```

KODIN TIETOREKISTERIT



Onko osoitekirjasi sekaisin? Hukkaatko puhelinnumeroita tai unohdatko syntymäpäiviä? Tämä monipuolinen ohjelma auttaa sinua selviytymään tämäntyyppisistä ongelmista. Sen avulla voit luoda ja päivittää tiedostoja, joita ovat esimerkiksi nimi- ja puhelinnumerolistat sekä tiedot syntymäpäivistä, ruokaresepteistä jne. Ohjelman avulla voit etsiä juuri tiettyä tietoa tai voit saada esiin koko tiedoston.

Ohjelmassa on valikko, josta valitaan jokin kahdeksasta toiminnosta.

1. Syötä uusia tietueita – tätä käytetään, kun aloitetaan uusi tiedosto tai lisätään uusia tietueita vanhaan tiedostoon.
2. Muuta tietuetta – tällä muutetaan tietueen nimeä tai siinä olevaa tietoa. Voidaan myös lisätä tietoa tai tuhota koko tietue.
3. Tulosta tietueet kuvaruutuun – kirjoittaa koko tiedoston kuvaruutuun.
4. Kirjoita tietueet paperille – sama kuin edellä, mutta tulostuksen hoitaa kirjoitin.
5. Etsi tietueet – tällä tehokkaalla toiminnolla voit etsiä ja tulostaa tietueita niiden nimen perusteella tai vieläkin kätevämmin voit selata tietueita etsimällä tiettyä hakusanaa, joka voi olla mikä tahansa sana sekä kirjain tai numerojono. Jos et tiedä kenen syntymäpäivä on tänään, voit antaa koneelle päivämäärän ja ohjelma etsii tämän perusteella, missä tietueessa on vastaava nimi ja tulostaa sen sinulle.
6. Lajittele tietueet aakkosjärjestykseen – paljon helpompi ja nopeampi kuin käsin lajittelu.
7. Lataa nauhalta – lataa aikaisemmin tallennetun tiedoston nauhalta.
8. Tallenna nauhalle – käytä tätä aina ennen kuin sammutat tietokoneen!

Vaikka ohjelma onkin kehitetty erikoisesti kotikäyttöön, sitä voidaan käyttää myös monissa pienissä yrityksissä, mm. asiakastietojen ja postitusluetteloiden tekemiseen.

Tietoja ohjelmasta

Rivin n:o

- | | |
|---------|--|
| 160 | Varaa tilaa merkkijonoja varten. |
| 180 | Tietue- ja tietueennimitaulukon tilavaraus |
| 220–340 | Kirjoittaa valikon. |
| 360–390 | Odottaa tietoa toiminnon valinnasta. |
| 400 | Hyppää valittuun ohjelman kohtaan. |
| 420–590 | Uusi tietue |

610-1190 Muuta tietuetta.
 630-730 Etsi tietueen nimi.
 740-780 Tarkista, onko tietue oikea.
 800-920 Mitä tietueelle tehdään?
 940-1030 Tuhoa tietue.
 1050-1080 Vaihda tietueen nimi.
 1100-1130 Muuta tietuetta.
 1150-1190 Lisää uutta tietoa.
 1210-1310 Kirjoita tietueet kuvaruutuun.
 1300-1430 Kirjoita tietueet paperille.
 1450-1840 Tutki tietueita.
 1450-1550 Nimi tai merkkijono?
 1560-1680 Etsi nimi ja kirjoita löydetty tietue.
 1700-1840 Etsi merkkijonoa.
 1860-1960 Tietueet aakkosjärjestykseen
 1980-2100 Lataa tiedosto nauhalta.
 2120-2270 Tallenna tiedosto nauhalle.
 2290-2300 Viive
 2320-2360 Aliohjelma näppäintietojen lukuun

Ohjelma: TIET-JS

```

100 REM *****
110 REM * *
120 REM * KODIN TIETOREKISTERIT *
130 REM * *
140 REM *****
150 REM
160 CLEAR 3000
170 REM TAULUKKOJEN TILAVARAUKSET
TIE-      TUEITA JA NIIDEN NIMIA VARTEN
180 DIM R$(500),N$(500)
190 COLOR 1,15
200 CLS
210 REM TULOSTAA VALIKON
220 LOCATE 9,1
230 PRINT "KODIN TIETOREKISTERIT"
240 LOCATE 0,4
250 PRINT "1. Syötä uusia tietueita":PRINT
260 PRINT "2. Muuta tietuetta":PRINT
270 PRINT "3. Tulosta tietueet kuvaruudull
e":PRINT
280 PRINT "4. Kirjoita tietueet paperille":PRINT
290 PRINT "5. Etsi tietueet":PRINT
  
```

```

300 PRINT "6. Lajittele tietueet":PRINT
310 PRINT "7. Lataa nauhalta":PRINT
320 PRINT "8. Tallenna nauhalle":PRINT
330 LOCATE 0,21
340 PRINT "Valitse oikea numero..."
350 REM ODOTTAA VASTAUSTA JA HYVÄKS
YY      SEN
360 A$=INKEY$
370 IF A$="" THEN 360
380 BEEP
390 IF A$<"1" OR A$>"8" THEN 370
400 ON VAL(A$) GOTO 420,610,1210,1330,1440,
1860,1980,2120
410 REM UUSI TIETUE
420 CLS
430 LOCATE 0,6
440 INPUT "Onko tämä uusi tiedosto?";Q$
450 IF LEFT$(Q$,1)="k" OR LEFT$(Q$,1)="K"
THEN J=1 ELSE J=J+1
460 CLS
470 LOCATE 0,3
480 PRINT "Syötä uudet tietueet - kirjoita
LOPPU, kun lopetat."
490 PRINT:PRINT
500 LINE INPUT "Tietueen nimi? ";N$(J)
510 IF N$(J)="LOPPU" OR N$(J)="loppu" THEN
580
520 PRINT
530 LINE INPUT "Tietosisältö? ";R$(J)
540 IF R$(J)="LOPPU" OR R$(J)="loppu" THEN
580
550 PRINT:PRINT
560 J=J+1
570 GOTO 500
580 J=J-1
590 GOTO 200
600 REM MUUTA TIETUETTA
610 CLS
620 N=0
630 LOCATE 0,3
640 INPUT "Tietueen nimi";T$
650 PRINT:PRINT
660 M=1
670 IF T$=N$(M) THEN N=M:GOTO 740

```

```

680 M=M+1
690 IF M<>J THEN 670
700 IF N<>0 THEN 720
710 PRINT "Tietuetta nimeltä ";T$;" ei löydy"
720 GOSUB 2290
730 GOTO 200
740 PRINT N$(N):PRINT
750 PRINT R$(N):PRINT
760 INPUT "Tämä tietue";Q$
770 IF LEFT$(Q$,1)="K" OR LEFT$(Q$,1)="k"
  THEN 800
780 GOTO 680
790 REM MUUTA TIETUEEN VALIKKO
800 CLS
810 LOCATE 0,4
820 PRINT "1. Tuhoa tietue":PRINT
830 PRINT "2. Muuta tietueen nimi":PRINT
840 PRINT "3. Muuta tietosisältöä":PRINT
850 PRINT "4. Lisää tietoa":PRINT
860 LOCATE 0,14
870 PRINT "Valitse oikea numero...":PRINT
880 A$=INKEY$
890 IF A$="" THEN 880
900 BEEP
910 IF A$<"1" OR A$>"4" THEN 880
920 ON VAL(A$) GOTO 940,1050,1100,1150
930 REM TUHOA TIETUE
940 FOR M=N TO J
950 N$(M)=N$(M+1)
960 R$(M)=R$(M+1)
970 NEXT M
980 N$(J)=""
990 R$(J)=""
1000 J=J-1
1010 PRINT "Tietue tuhottu"
1020 GOSUB 2290
1030 GOTO 200
1040 REM MUUTA NIMEÄ
1050 INPUT "Uusi nimi";N$(N)
1060 PRINT:PRINT"Nimi muutettu"
1070 GOSUB 2290
1080 GOTO 200
1090 REM MUUTA TIETOSISÄLTÖÄ
1100 INPUT "Uusi tietosisältö";R$(N)
1110 PRINT:PRINT"Tietosisältö muutettu"

```

```

1120 GOSUB 2290
1130 GOTO 200
1140 REM LISÄÄ TIETOA
1150 INPUT "Lisätiedot";T$
1160 R$(N)=R$(N)+" - "+T$
1170 PRINT:PRINT"Uusi tieto lisätty"
1180 GOSUB 2290
1190 GOTO 200
1200 REM TULOSTA TIETUEET
1210 CLS
1220 D$=STRING$(36,219)
1230 FOR M=1 TO J
1240 PRINT:PRINT"Tietueen nimi:";N$(M)
1250 PRINT:PRINT R$(M)
1260 PRINT:PRINT D$
1270 FOR DE=1 TO 100:NEXT DE
1280 GOSUB 2320
1290 NEXT M
1300 GOSUB 2290
1310 GOTO 200
1320 REM KIRJOITA TIETUEET PAPERILLE
1330 CLS
1340 LOCATE 0,6
1350 LINE INPUT "Paina enter, kun kirjoitin  
on valmis";NU$
1360 FOR M=1 TO J
1370 LPRINT
1380 LPRINT "Tietueen nimi:";N$(M)
1390 LPRINT
1400 LPRINT R$(M)
1410 LPRINT
1420 NEXT M
1430 GOTO 200
1440 REM ETSI TIETUEET
1450 CLS
1460 LOCATE 0,6
1470 PRINT "1. Tietueen nimen mukaan":PRINT
1480 PRINT "2. Hakusanan mukaan":PRINT
1490 LOCATE 0,12
1500 PRINT "Valitse oikea numero...":PRINT
1510 A$=INKEY$
1520 IF A$="" THEN 1510
1530 BEEP
1540 IF A$<"1" OR A$>"2" THEN 1510
1550 ON VAL(A$) GOTO 1560,1700

```



```

1560 INPUT "Tietueen nimi";T$
1570 PRINT
1580 F=0
1590 FOR M=1 TO J
1600 IF N$(M)<>T$ THEN 1640
1610 PRINT "Tietueen nimi:";N$(M)
1620 PRINT:PRINT R$(M)
1630 F=1
1640 NEXT M
1650 IF F=1 THEN 1670
1660 PRINT:PRINT"Tietuetta ei löydy"
1670 GOSUB 2290
1680 GOTO 200
1690 REM HAKUSANA
1700 INPUT "Etsittävä hakusana";T$
1710 F=0
1720 FOR M=1 TO J
1730 X=INSTR(N$(M),T$)
1740 IF X<>0 THEN 1770
1750 X=INSTR(R$(M),T$)
1760 IF X=0 THEN 1800
1770 PRINT:PRINT"Tietueen nimi:";N$(M)
1780 PRINT:PRINTR$(M):PRINT
1790 F=1
1800 NEXT M
1810 IF F=1 THEN 1830
1820 PRINT:PRINT"Tietuetta ei löydy"
1830 GOSUB 2290
1840 GOTO 200
1850 REM TIETUEIDEN LAJITTELU
1860 CLS
1870 LOCATE 0,6
1880 PRINT "Lajittelu..."
1890 M=1
1900 IF N$(M)<N$(M+1) THEN 1940
1910 SWAP N$(M),N$(M+1)
1920 SWAP R$(M),R$(M+1)
1930 GOTO 1890
1940 M=M+1
1950 IF M<>J THEN 1900
1960 GOTO 200
1970 REM LATAUS NAUHALTA
1980 CLS
1990 LOCATE 0,6

```

```
2000 PRINT "Lataa nauhalta":PRINT
2010 LINE INPUT "Paina enter, kun kassetti
on valmiina";NU$
2020 PRINT:PRINT "Lataus..."
2030 OPEN "KAS:TIET" FOR INPUT AS 1
2040 INPUT #1,J
2050 FOR M=1 TO J
2060 INPUT#1,N$(M)
2070 INPUT #1,R$(M)
2080 NEXT M
2090 CLOSE
2100 GOTO 200
2110 REM TALLENNUS NAUHALLE
2120 CLS
2130 LOCATE 0,6
2140 PRINT "Tallenna nauhalle":PRINT
2150 LINE INPUT "Paina enter, kun kassetti
on valmiina";NU$
2160 PRINT:PRINT"Tallennus..."
2170 MOTOR ON
2180 GOSUB 2290
2190 MOTOR OFF
2200 OPEN "KAS:TIET" FOR OUTPUT AS 1
2210 PRINT#1,J
2220 FOR M=1 TO J
2230 PRINT#1,N$(M)
2240 PRINT#1,R$(M)
2250 NEXT M
2260 CLOSE
2270 GOTO 200
2280 REM UIIVE
2290 FOR DE=1 TO 2500:NEXT DE
2300 RETURN
2310 REM ODOTTAA NÄPPÄIMEN PAINALLUSTA
2320 Z$=INKEY$
2330 IF Z$="" THEN 2360
2340 Z$=INKEY$
2350 IF Z$="" THEN 2340
2360 RETURN
```

PEIKKOLUOLAT



Kaukaisessa maassa, lukemattomien harmaakiviluolien kätköissä asustaa oikein epämiellyttävä hirviö nimeltä Horkki Hirmuinen. Jostain syystä Horkki on ottanut tavakseen syödä pari paikallista asukasta iltapalakseen joka ilta, ja niin on Horkista tullut oikea kiusankappale. Siksi nyt tarvitaankin joku urhoollinen hävittämään tämä hirviö. Se sankari voit olla sinä!

Tehtävä ei ole kuitenkaan aivan yksinkertainen. Kiehuvien laavakuoppien ympäröimiä luolia vartioi lukuisa joukko ilkeitä, vihreitä peikkoja, joiden erikoisherkkua ovat oudot ja pelotto-

mat seikkailijat. Ja mikä pahinta; sinulla on käytettävänäsi luolissa vain kuusi nuolta.

Hyvänä puolena on kuitenkin se, että näillä pahalaisilla on tapana muuttaa luolien väriä seuraavasti:

Kaikki peikkojen vieressä olevat luolat muuttuvat vihreiksi.
Kiehuva laavakuoppa muuttaa luolat punaisiksi.
Hirviö muuttaa luolat sinisiksi.

Tämä on sinulle varoituksena poistua välittömästi luolasta.

Kaikki ohjeet ovat ohjelmassa, joten älä enää viivyttel!

Tietoja ohjelmasta

Rivin n:o	
220-500	Kirjoittaa peliohjeet.
520	Luolatietojen taulukkojen tilavaraus
540-640	Asettaa alkuarvot.
660-890	Pääty- ja sivuseiniä piirtäminen
910-950	Lukee ja tallentaa luolakartan.
970-1000	Hirviön paikka satunnaisesti
1020-1070	Peikkojen paikat
1120-1140	Tarkistaa, asuuko luolassa joku.
1170-1190	Tarkistaa ja värjää luolan.
1230-1290	Suuntana pohjoinen
1310-1370	Suuntana etelä
1390-1450	Suuntana itä
1470-1520	Suuntana länsi
1540-1610	Odottaa suuntatietoa näppäimistöltä.
1630-1660	Tarkistaa kulun esteettömyyden.
1690-1800	Nuolen ampuminen
1820-1890	Tarkistaa osuman.
1900-1950	Ei osunut.
1970-2030	Hirviö sai sinut.
2040-2160	Putosit kuoppaan.
2170-2210	Peikot söivät sinut.
2240-2320	Luolien DATA-tiedot
2340-2420	Osuit peikkoon.
2440-2500	Osuit hirviöön.
2520-2530	Viive
2540-2560	Ilmoittaa nuolien lukumäärän.
2570-2630	Nuolet loppuivat.

Ohjelma: PEIK-JS

```
100 REM *****
110 REM *
120 REM * PEIKKOLUOLAT *
130 REM *
140 REM *****
150 REM
160 CLEAR 500
170 CLS
180 COLOR 1,15
190 R=RND(-TIME)
200 LOCATE 9,2
210 REM OHJEET
220 PRINT "PEIKKOLUOLAT"
230 LOCATE 0,9
240 PRINT "Paina näppäimiä P E I tai L
    kääntyäkse-si haluamaasi ilmansuuntaan.
    Pääset eteenpäin näppäimellä K.":PRINT:
    PRINT
250 PRINT "Muista, että voit liikkua vain
    paina- malla näppäintä K.":PRINT:PRIN
    T
260 LINE INPUT "Jatkuu, paina ENTER...";NU$
270 BEEP
280 CLS
290 LOCATE 0,5
300 PRINT "Luolissa on vaaroja:"
310 LOCATE 0,8
320 PRINT "1. Kiehuvia laavakuoppia -
    muuttavat kaikki viereiset luolat pun
    aisiksi."
330 PRINT
340 PRINT "2. Peikot - muuttavat kaikki
    viereiset luolat vihreiksi."
350 PRINT
360 PRINT "3. Hirviä - muuttaa viereiset
    luolat sinisiksi.":PRINT:PRINT
370 LINE INPUT "Jatkuu, paina ENTER...";
    NU$:CLS
380 BEEP
390 CLS
400 LOCATE 0,3
410 PRINT "Tehtäväsi on etsiä ja tuhota
```

```

    hirviä. Siihen tarkoitukseen sinulla o
n kuusi nuolta."
420 PRINT
430 PRINT "Voit ampua painamalla
    välilyöntinäp- päintä.":PRINT
440 PRINT "Myös peikot voidaan tuhota
    nuolilla.":PRINT
450 PRINT "Jos ammut ohi, juoksevat kaikki
    peikot ja hirviö eri luoliin. (Joku e
    hkä sa- maan luolaan, missä sinä olet.)
460 PRINT
470 PRINT "Häviät pelin, jos menet luolaan,
    jossa on jokin vaaroista tai jos nuo
    lesi lop-puvat."
480 PRINT:PRINT
490 LINE INPUT "Peli alkaa, paina ENTER...";
NO$
500 BEEP
510 OPEN "GRP:" AS 1
520 DIM M(25,60)
530 REM ALKUASEMA
540 R=7
550 C=7
560 REM JA ALKUSUUNTA
570 DR=2
580 CO=2
590 REM NUOLIIEN LUKUMÄÄRÄ
600 AR=6
610 REM PEIKKOJEN LUKUMÄÄRÄ
620 GB=5
630 CLS
640 GOTO 920
650 REM LUOLIIEN PIIRTO
660 COLOR CO
670 DRAW "BM 90,60;R80D80L80U80"
680 PAINT(100,70),CO
690 RETURN
700 COLOR CO
710 DRAW "BM90,60;R80D80L20U5H2U4H3U1H1U15
    H3U10L5H2L8G2L2G1L2G5D10G2D14G1D11L21U80"
720 PAINT(100,70),CO
730 RETURN
740 COLOR CO

```

```

750 DRAW "BM170,60;E50D180H50U80"
760 PRINT(175,90),C0
770 RETURN
780 COLOR C0
790 DRAW "BM 90,60;H50D180E50U80"
800 PRINT(85,70),C0
810 RETURN
820 COLOR C0
830 DRAW "BM 170,60;E50D180H18U9H2U24U10
E1U23H6G3L2G4L1G1FD9G1D9G1D28H12U80"
840 PRINT(175,90),C0
850 RETURN
860 COLOR C0
870 DRAW "BM90,60;H50D180E18U9E2U12E2U12S2
U10E1U23E6F3L2F4L1F6D9F1D9F1D24E8U80"
880 PRINT(80,65),C0
890 RETURN
900 REM LUKEE LUOLIEK KARTTATIEDON
910 RESTORE
920 FOR J=2 TO 10
930 FOR K=2 TO 25
940 READ M(J,K)
950 NEXT K,J
960 REM HIRVIÖN ASEMA
970 P1=INT(RND(1)*8)+1
980 P2=INT(RND(1)*22)+2
990 IF M(P1,P2)<>1 OR (P1=7 AND P2=7) THEN
970
1000 M(P1,P2)=3
1010 REM PEIKKOJEN ASEMA
1020 FOR J=1 TO GB
1030 P1=INT(RND(1)*8)+1
1040 P2=INT(RND(1)*22)+2
1050 IF M(P1,P2)<>1 OR (P1=7 AND P2=7) THEN
1030
1060 M(P1,P2)=4
1070 NEXT J
1080 SCREEN 2
1090 COLOR 15,1,1
1100 CLS
1110 REM TUTKII LUOLAN, JOSSA OLET
1120 IF M(C,R)=2 THEN 2040
1130 IF M(C,R)=4 THEN 2170
1140 IF M(C,R)=3 THEN 1970

```

```

1150 CO=14
1160 REM TARKISTAA LUOLAN VÄRIN
1170 IF M(C,R-1)=2 OR M(C,R+1)=2 OR M(C-1,R)=2 OR M(C+1,R)=2 THEN CO=6
1180 IF M(C,R-1)=3 OR M(C,R+1)=3 OR M(C-1,R)=3 OR M(C+1,R)=3 THEN CO=4
1190 IF M(C,R-1)=4 OR M(C,R+1)=4 OR M(C-1,R)=4 OR M(C+1,R)=4 THEN CO=12
1200 REM VALITSEE SUUNNAN
1210 ON DR GOTO 1230,1310,1390,1470
1220 REM SUUNTA POHJOISEEN
1230 IF M(C,R-1)>0 THEN GOSUB 700 ELSE
GOSUB 660
1240 IF M(C+1,R)>0 THEN GOSUB 820 ELSE
GOSUB 740
1250 IF M(C-1,R)>0 THEN GOSUB 860 ELSE
GOSUB 780
1260 COLOR 15
1270 PRESET(64,180)
1280 PRINT#1,"SUUNTA POHJOISEEN"
1290 GOTO 1540
1300 REM SUUNTA ETELÄÄN
1310 IF M(C,R+1)>0 THEN GOSUB 700 ELSE
GOSUB 660
1320 IF M(C-1,R)>0 THEN GOSUB 820 ELSE
GOSUB 740
1330 IF M(C+1,R)>0 THEN GOSUB 860 ELSE
GOSUB 780
1340 COLOR 15
1350 PRESET(84,180)
1360 PRINT#1,"SUUNTA ETELÄÄN"
1370 GOTO 1540
1380 REM SUUNTA ITÄÄN
1390 IF M(C+1,R)>0 THEN GOSUB 700 ELSE
GOSUB 660
1400 IF M(C,R+1)>0 THEN GOSUB 820 ELSE
GOSUB 740
1410 IF M(C,R-1)>0 THEN GOSUB 860 ELSE
GOSUB 780
1420 COLOR 15
1430 PRESET(84,180)
1440 PRINT#1,"SUUNTA ITÄÄN"
1450 GOTO 1540
1460 REM SUUNTA LÄNTEEN
1470 IF M(C-1,R)>0 THEN GOSUB 700 ELSE

```



```

GOSUB 660
1480 IF M(C,R-1)>0 THEN GOSUB 820 ELSE
GOSUB 740
1490 IF M(C,R+1)>0 THEN GOSUB 860 ELSE
GOSUB 780
1500 COLOR 15
1510 PRESET(84,180)
1520 PRINT#1,"SUUNTA LANTEEN"
1530 REM ODOTTAA TIETOA NÄPPÄIMISTÖLTÄ
1540 A$=INKEY$
1550 IF A$="" THEN 1540
1560 REM SUUNTIEN NUMEROT
1570 IF A$="P" THEN DR=1:GOTO 1100
1580 IF A$="E" THEN DR=2:GOTO 1100
1590 IF A$="I" THEN DR=3:GOTO 1100
1600 IF A$="L" THEN DR=4:GOTO 1100
1610 IF A$="K" THEN 1630 ELSE 1690
1620 REM ONKO LIIKE MAHDOLLINEN
1630 IF DR=1 THEN R=R-1:IF M(C,R)=0 THEN
R=R+1:GOTO 1540
1640 IF DR=2 THEN R=R+1:IF M(C,R)=0 THEN
R=R-1:GOTO 1540
1650 IF DR=3 THEN C=C+1:IF M(C,R)=0 THEN
C=C-1:GOTO 1540
1660 IF DR=4 THEN C=C-1:IF M(C,R)=0 THEN
C=C+1:GOTO 1540
1670 GOTO 1100
1680 REM NUOLEN AMPUMINEN
1690 IF AR<1 THEN 1540
1700 IF A$=CHR$(32) THEN 1710 ELSE 1540
1710 SOUND 6,15
1720 SOUND 7,7
1730 SOUND 8,16
1740 SOUND 9,16
1750 SOUND 10,16
1760 SOUND 11,0
1770 SOUND 12,16
1780 SOUND 13,4
1790 FOR DE=1 TO 1000:NEXT DE
1800 AR=AR-1
1810 REM TARKISTAA OSUMAN
1820 IF DR=1 AND M(C,R-1)=3 THEN M(C,R-1)=
1:GOTO 2440
1830 IF DR=1 AND M(C,R-1)=4 THEN M(C,R-1)=
1:GOTO 2340

```

```

1840 IF DR=2 AND M(C,R+1)=3 THEN M(C,R+1)=
1:GOTO 2440
1850 IF DR=2 AND M(C,R+1)=4 THEN M(C,R+1)=
1:GOTO 2340
1860 IF DR=3 AND M(C+1,R)=3 THEN 2440
1870 IF DR=3 AND M(C+1,R)=4 THEN M(C+1,R)=
1:GOTO 2340
1880 IF DR=4 AND M(C-1,R)=3 THEN 2440
1890 IF DR=4 AND M(C-1,R)=4 THEN M(C-1,R)=
1:GOTO 2340
1900 COLOR 15
1910 PRESET(50,70)
1920 PRINT#1,"          ET OSUNUT!"
1930 GOSUB 2540
1940 GOSUB 2520
1950 GOTO 910
1960 GOTO 2650
1970 COLOR 15
1980 PRESET(50,70)
1990 PRINT#1,"HIRVIÖ SAI SINUT!"
2000 SOUND 7,56
2010 PLAY "O2GFEDCO1BAGFED"
2020 GOSUB 2520
2030 GOTO 1960
2040 COLOR 15
2050 PRESET(30,70)
2060 PRINT#1,"PUTOSIT LAAVAKUOPPAANI!"
2070 FOR J=48 TO 255
2080 SOUND 7,254
2090 SOUND 8,15
2100 SOUND 0,J
2110 SOUND 1,0
2120 NEXT J
2130 SOUND 8,0:SOUND 7,56
2140 PLAY "L9;02;D"
2150 GOSUB 2520
2160 GOTO 1960
2170 COLOR 15
2180 PRESET(65,70)
2190 PRINT#1,"PEIKOT SÄIVÄT SINUT!"
2200 SOUND 7,56
2210 GOTO 2010
2220 GOSUB 2520
2230 REM LUOLIIEN DATA-TIEDOT
2240 DATA 0,0,0,0,2,2,0,0,0,2,0,0,0,0,0,0,

```

```

2,0,0,0,0,2,0,0
2250 DATA 0,0,1,1,1,1,0,0,0,1,0,0,0,0,0,0,
1,1,1,1,0,1,0,0
2260 DATA 0,0,0,0,1,1,0,0,0,1,0,1,1,1,1,1,
1,0,1,1,0,1,1,2
2270 DATA 0,0,0,1,1,1,0,0,1,1,1,1,0,1,0,0,
0,0,1,0,0,1,0,0
2280 DATA 0,0,0,1,0,1,1,0,1,0,0,0,0,1,0,0,
0,0,1,1,1,1,1,2
2290 DATA 2,1,1,1,1,1,1,1,1,1,1,1,1,1,1,0,
1,1,1,1,1,0,1,0
2300 DATA 0,0,0,1,0,1,1,0,1,0,0,1,2,0,0,0,
0,0,0,0,1,0,0,0
2310 DATA 0,0,0,1,1,1,0,0,1,1,1,1,0,0,0,0,
0,0,0,1,1,1,2,0
2320 DATA 0,0,0,0,2,2,0,0,0,2,0,1,1,1,1,1,
1,1,1,1,0,1,2,0
2330 REM MAALINA PEIKKO
2340 PRESET (50,70)
2350 COLOR 15
2360 PRINT#1,"OSUIT PEIKKOON!"
2370 SOUND 7,56
2380 PLAY "T240;04CDEFGAB05CDEFGAB"
2390 GB=GB-1
2400 GOSUB 2540
2410 GOSUB 2520
2420 GOTO 1080
2430 REM MAALINA HIRVI6
2440 COLOR 15
2450 PRESET(50,70)
2460 PRINT#1,"OSUIT HIRVI66N!"
2470 SOUND 7,56
2480 PLAY "T150F03A02DL12G03B02DL4G03B02D
L150"
2490 GOSUB 2520
2500 GOTO 1960
2510 REM UIIVERUTIINI
2520 FOR DE=1 TO 1900:NEXT DE
2530 RETURN
2540 PRESET(65,20)
2550 PRINT#1,"NUOLIA JÄLJELLÄ";AR
2560 IF AR<>0 THEN 2640
2570 GOSUB 2520
2580 SCREEN 0

```

```
2590 LOCATE 6,8
2600 PRINT "NUOLESI LOPPUIVAT!"
2610 GOSUB 2520
2620 LOCATE 0,21
2630 GOTO 1960
2640 RETURN
2650 END
```

PINGVIINI



Näyttää siltä, että pingviinit ovat pulassa kaukaisilla Etelämantereen erämailla. Sadat rosvoilevat linnut yrittävät jatkuvasti syöksyillen tuhota pingviinien munat.

Vaikka näillä onnettomilla luontokappaleilla onkin uusin MSX-laser-ase käytössään, on heillä tiukat paikat, joten sinun apuasi tarvitaan.

Liiku kohdistimen ohjausnäppäimillä vasemmalle ja oikealle ja ammu välilyöntinäppäimellä. Linnut voidaan tuhota joko laser-aseella tai vanhanaikaisemmalla, mutta yhtä tehokkaalla tavalla – juoksemalla niitä päin.

Tähtää tarkasti, sillä laser-aseen lataamiseen kuluu pari sekuntia laukauksen jälkeen.

Tietoja ohjelmasta

Rivin n:o	
200	Pingviinirivi
210-220	Aliohjelmat keskeytyksille
250-260	Odottaa tietoa näppäimistöltä; ajantasausaliohjelma, jos mitään ei paineta.
280-300	Liikuttaa pingviiniä sivusuunnassa.
310-410	Testaa laserin valmiuden, ääniefektit.
430-470	Liikuttaa laser-spritea.
500-610	Kirjoittaa otsikon ja peliohjeet.
690-810	Pingviini-, laser- ja lintu-spritet
870-930	Piirtää maiseman.
950-960	Lisää auringon.
980-1030	Lisää lokit.
1040-1050	Keskeytykset päälle
1090-1130	Piirtää munat.
1180-1450	Pääohjelasilmukka; liikuttaa lintuja.
1470-1570	Kaikki munat menetetty.
1580-1770	Pelaajan nimi; parhaat tulokset
1790-1870	Kysyy uutta peliä.
1900-2000	Asettaa lintujen paikan satunnaisesti.
2020-2100	Kirjoittaa tuloksen ja munien määrän.
2140-2420	Sprite-törmäys; osuma lintuun
2430-2450	Laser-lippu alkuarvoonsa

Ohjelma: PING-JS

```

100 REM *****
110 REM *
120 REM * PINGVIINI *
130 REM *
140 REM *****
150 REM
160 R=RND(-TIME)
170 CLS
180 OPEN "GRP:" AS 1
190 REM PINGVIINIRIVI
200 PR=174
210 ON SPRITE GOSUB 2150
220 ON INTERVAL=50 GOSUB 2440

```

```

230 GOTO 500
240 REM ODOTTAA TIETOA NÄPPÄIMISTÖLTÄ
250 K$=INKEY$
260 IF K$="" THEN GOSUB 2110:GOTO 480
270 REM PINGVIININ LIIKE
280 IF K$=CHR$(29) AND PC>20 THEN PC=PC-8
290 IF K$=CHR$(28) AND PC<240 THEN PC=PC+8
300 PUT SPRITE 1,(PC,PR),1,1
310 IF K$<>CHR$(32) THEN 480
320 REM AMPUMINEN LASER-ASEELLA
330 IF FF=1 THEN 480
340 SOUND 6,15
350 SOUND 7,7
360 SOUND 8,16
370 SOUND 9,16
380 SOUND 10,16
390 SOUND 11,0
400 SOUND 12,16
410 SOUND 13,0
420 REM LIIKUTTAA LASER-SADETTÄ
430 FOR Z=PR-8 TO 8 STEP -8
440 PUT SPRITE 2,(PC,Z),10,2
450 NEXT Z
460 PUT SPRITE 2,(0,209),,2
470 FF=1
480 RETURN
490 REM OHJEET
500 COLOR 1,15
510 CLS
520 LOCATE 12,3
530 PRINT "PINGVIINI"
540 LOCATE 0,7
550 PRINT "Ei ole hauskaa olla nykyisin
pingviini.On suojattava kuutta munaa ryc
stelevil-tä linnuilta yhdellä ja yhteisellä
ä laser-pyssyllä.
560 PRINT:PRINT
570 PRINT "Kuinka pärjäisit pingviininä?"
580 PRINT:PRINT
590 PRINT "Liiku kohdistinnäppäimillä
vasemmalle ja oikealle ja tulitus välilyö
ntinäp- päimellä."
600 PRINT:PRINT
610 PRINT "Peli alkaa, paina jotain näppäintä!"

```

```

620 A$=INKEY$
630 IF A$="" THEN 620
640 BEEP
650 SCREEN 2
660 COLOR 1,15,5
670 CLS
680 REM LUKEE SPRITE-DATAN J
A      MÄÄRITTELEE SPRITET
690 FOR K=1 TO 4
700 P$=""
710 FOR J=0 TO 7
720 READ D$
730 P$=P$+CHR$(VAL("&H"+D$))
740 NEXT J
750 SPRITE$(K)=P$
760 NEXT K
770 REM SPRITE DATA
780 DATA 1C,38,3C,66,66,66,3C
790 DATA 08,08,08,08,08,08,08,08
800 DATA 80,82,CE,FC,38,78,6E,47
810 DATA 01,41,73,3F,1C,1E,76,E2
820 FOR J=1 TO 6
830 M(J)=0
840 NEXT J
850 SC=0
860 REM PIIRTAA VUORET
870 PRESET(0,100),5
880 DRAW "E8 F16 E12 F20 R2 E30 R3 F22 R2
E17 F40 R10 E33 R9 F32"
890 PRESET(47,106),5
900 DRAW "E10 F10"
910 PRESET(146,106),5
920 DRAW "E30 F30"
930 PAINT (0,0),5
940 REM AURINKO
950 CIRCLE(190,42),13,10,,,1.3
960 PAINT(190,42),10
970 REM LOKIT
980 PRESET(140,30),15
990 DRAW "F2 E2"
1000 PRESET(134,26),15
1010 DRAW "F2 E2"
1020 PRESET(122,32),15
1030 DRAW "F2 E2"

```



```
1040 SPRITE ON
1050 INTERVAL ON
1060 REM PINGVIINISARAKE
1070 PC=125
1080 REM MUNIEN ASEMAT
1090 FOR J=1 TO 6
1100 EC(J)=J*40
1110 CIRCLE(J*40,188),2,6,,,1.3
1120 PAINT(J*40,188),6
1130 NEXT J
1140 REM MUNIEN LUKUMAÄRÄ
1150 EL=6
1160 GOSUB 2020
1170 REM PÄÄOHJELMASILMUKKA
1180 FOR J=1 TO 6
1190 GOSUB 1900
1200 GOSUB 250
1210 IF BC<=EC(WE) THEN 1260
1220 BC=BC-8
1230 SS=3
1240 PUT SPRITE 4,(0,209),,4
1250 GOSUB 250
1260 IF BC>=EC(WE) THEN 1310
1270 BC=BC+8
1280 SS=4
1290 PUT SPRITE 3,(0,209),,3
1300 GOSUB 250
1310 IF BR>181 THEN 1330
1320 BR=BR+8
1330 PUT SPRITE SS,(BC,BR),1,SS
1340 GOSUB 250
1350 IF BC=EC(WE) AND BR>181 THEN 1360 ELSE
1440
1360 SOUND 7,56
1370 PLAY "L4 02 D"
1380 PRESET(EC(WE)-8,184),15
1390 PRINT#1,CHR$(219)+CHR$(219)
1400 PUT SPRITE SS,(0,209),,SS
1410 EL=EL-1
1420 GOSUB 2020
1430 GOTO 1450
1440 GOTO 1210
1450 NEXT J
1460 REM PELI LOPPUU
```

```
1470 SOUND 7,56
1480 PLAY "T210 04 BAGFEDC 03 BAGFEDC 02
    BAGFEDC"
1490 PRESET(50,140)
1500 COLOR 3
1510 PRINT#1," P E L I   O H I "
1520 FOR DE=1 TO 2500:NEXT DE
1530 SCREEN 0
1540 COLOR 1,15
1550 CLS
1560 PRINT "SAIT";SC;"PISTETTA
1570 PRINT
1580 INPUT "MIKÄ ON NIMESI ";NM$
1590 IF LEN(NM$)>15 THEN PRINT "NIMI ON
    LIIAN PITKÄ!":PRINT:GOTO 1580
1600 REM PARHAAT TULOKSET TAULUKKONA
1610 S(10)=SC
1620 N$(10)=NM$
1630 FOR J=10 TO 2 STEP-1
1640 IF S(J)<S(J-1) THEN 1670
1650 SWAP S(J),S(J-1)
1660 SWAP N$(J),N$(J-1)
1670 NEXT J
1680 CLS
1690 PRINT TAB(8);"PARAS TULOS"
1700 PRINT:PRINT
1710 FOR M=1 TO 9
1720 PRINT " ";M;" ";S(M);" ";N$(M)
1730 NEXT M
1740 LOCATE 2,19
1750 PRINT "UUSI PELI, PAINA 'P'"
1760 PRINT "  LOPETUS,   PAINA 'L'"
1770 PRINT
1780 REM ODOTTAA VASTAUSTA
1790 K$=INKEY$
1800 IF K$="" THEN 1790
1810 BEEP
1820 IF K$<>"P" THEN 1870
1830 SC=0
1840 SCREEN 2
1850 CLS
1860 GOTO 820
1870 IF K$<>"L" THEN 1790
1880 GOTO 2460
```


```
1890 REM LINTUJEN ASEMA
1900 BC=INT(RND(1)*28)+3
1910 BC=BC*8
1920 BR=INT(RND(1)*17)+2
1930 BR=BR*8
1940 FF=0
1950 WE=INT(RND(1)*6)+1
1960 GOSUB 250
1970 IF M(WE)=1 THEN 1950
1980 M(WE)=1
1990 IF BC/8>16 THEN SS=4 ELSE SS=3
2000 RETURN
2010 REM TULOS KUVARUUDULLE
2020 PRESET(0,0)
2030 COLOR 5
2040 PRINT#1,STRING$(32,219)
2050 PRESET(30,0)
2060 COLOR 15
2070 PRINT#1,"MUNIA:";EL
2080 PRESET(150,0),15
2090 PRINT#1,"TULOS:";SC
2100 RETURN
2110 FOR DE=1 TO 50
2120 NEXT DE
2130 RETURN
2140 REM SPRITE-KESKEYTYS
2150 SPRITE OFF
2160 REM RÄJÄHDYS
2170 SOUND 0,0
2180 SOUND 1,5
2190 SOUND 2,0
2200 SOUND 3,13
2210 SOUND 4,255
2220 SOUND 5,15
2230 SOUND 6,30
2240 SOUND 7,0
2250 SOUND 8,16
2260 SOUND 9,16
2270 SOUND 10,16
2280 SOUND 11,0
2290 SOUND 12,5
2300 SOUND 13,0
2310 FOR DE=1 TO 30:NEXT DE
2320 SOUND 12,56
```

```

2330 SOUND 13,0
2340 SC=SC+10
2350 M(WE)=0
2360 PUT SPRITE SS,(0,209),,SS
2370 Z=8
2380 PUT SPRITE 2,(0,209),,2
2390 GOSUB 2020
2400 GOSUB 1900
2410 SPRITE ON
2420 RETURN
2430 REM KESKEYTYS
2440 FF=0
2450 RETURN
2460 END

```

TILIKIRJA



Jos olet ajatellut ostaa levyaseman tai kirjoittimen lisätäksesi MSX-järjestelmääsi, saat tämän ohjelman avulla tietää, kestääkö pankkitilisi tätä.

Ohjelman avulla on mahdollista pitää kaikki tilitietosi ajan tasalla ja tallentaa ne nauhalle tai kirjoittaa paperille.

Ohjelma on jaettu seitsemään osaan, jotka voidaan valita valikon avulla seuraavasti:

1. Uusi tili – tätä käytetään tilin alkuarvojen antamiseen, kun ohjelmaa käytetään ensimmäisen kerran. Ohjelma kysyy tilin nimeä ja muita tietoja. Kaikissa toiminnoissa erät, kuten esimerkiksi kestotilaukset ja suoraveloitukset, täytyy ilmoittaa omilla tilinumeroillaan.
2. Tilin päivitys – tätä toimintoa käytetään lisätietojen antamiseksi jo avattuun tiliin.
3. Tilítietojen tulostus – kirjoittaa kaikki tilítiedot kuvaruutuun. Tulostuksen voi pysäyttää painamalla mitä tahansa näppäintä ja saada uudelleen päälle painamalla uudestaan jotain näppäintä.
4. Tilítietojen kirjoitus – kirjoittaa kaikki tilítiedot kirjoittimella paperille.
5. Saldo – antaa tämänhetkisen saldon kaikista tietokoneessa olevista tileistä.
6. Lataa nauhalta – lataa tilítiedot nauhalta.
7. Tallenna nauhalle – tallentaa kaikki tilítiedot nauhalle. Käytä tätä toimintoa aina ennen kuin sammutat tietokoneen!

Ohjelmalla pystytään käsittelemään kussakin tiedostossa kolmea eri tiliä, joissa kussakin voi olla 100 tilítapahtumaa. Näitä arvoja voidaan kuitenkin helposti suurentaa, tietokoneen muistin asettamissa rajoissa, muuttamalla taulukoiden tilavarauksia rivillä 180 sekä muuttamalla vastaavasti rivejä 720, 840, 1030, 1380, 1830, 2270 ja 2530.

Tätä ohjelmaa voidaan helposti käyttää kaikessa kodin tilinpidossa ja jopa pienissä yrityksen tiliasioissa.

Tietoja ohjelmasta

Rivin n:o

180

Tilítietojen taulukoiden tilivaraus

190

'Print using' -lausekkeen muotoilu

230–340

Kirjoittaa valikon.

350–390

Odottaa valintaa näppäimistöltä.

410

Hyppää oikeaan ohjelmakohtaan.

430–850

Uusi tili

450-480	Kirjoittaa otsikot.
510-530	Etsii välilyönnin.
540-560	Välilyöntiä ei löytynyt.
570-730	Maksetut laskut
740-840	Saadut suoritukset
870-1090	Tilin päivitys
960-980	Etsii nimetyn tiedoston.
990-1010	Nimeä ei löytynyt.
1030-1050	Etsii edellisten tapahtumien lopun.
1110-1550	Tilitietojen tulostus
1130-1160	Kirjoittaa otsikot.
1170	Asettaa alkuarvot.
1180-1210	Mikä tili?
1230-1250	Etsii nimetyn tiedoston.
1260-1290	Nimeä ei löytynyt.
1320	Tarkistaa esillä olevan tilin.
1330-1370	Kirjoittaa otsikot.
1390	Tarkistaa tilitietojen lopun.
1400-1410	Tarkistaa maksusuoritusten antamisen.
1430-1470	Kirjoittaa tiedot.
1480	Viive lukemista varten
1490	Odottaa näppäimen painallusta.
1570-1980	Kirjoittaa tilitiedot paperille samassa muodossa kuin kuvaruutuunkin.
2000-2120	Kirjoittaa kaikkien tilien saldon.
2140-2340	Lataa tilitiedot nauhalta.
2360-2600	Tallentaa tilitiedot nauhalle.
2460-2480	Rutiini tallennusta varten
2620-2630	Viive
2640-2690	Tarkistaa näppäimen painalluksen.

Ohjelma: TILI-JS

```

100 REM *****
110 REM *
120 REM *   TILIKIRJA   *
130 REM *
140 REM *****
150 REM
160 CLEAR 2000
170 REM TILAVARAUS TILITÄULUKOILLE
180 DIM NM$(3),AM(3,100),P$(3,100),BA(3),
CN(3,100),BF(3,100)
190 D$="#####.##"

```

```

200 COLOR 1,15
210 CLS
220 REM KIRJOITTAA VALIKON
230 LOCATE 11,1
240 PRINT " TILIKIRJA "
250 LOCATE 0,4
260 PRINT "1. Uusi tili.":PRINT
270 PRINT "2. Tilin päivitys.":PRINT
280 PRINT "3. Tilitietojen tulostus.":PRINT
290 PRINT "4. Tilitietojen kirjoitus.":PRINT
300 PRINT "5. Saldo.":PRINT
310 PRINT "6. Lataa nauhalta.":PRINT
320 PRINT "7. Tallenna nauhalle."
330 LOCATE 0,21
340 PRINT "Valitse oikea vaihtoehto..."
350 A$=INKEY$
360 IF A$="" THEN 350
370 BEEP
380 REM TARKISTAA SYÖTTÖTIEDON
390 IF A$<"1" OR A$>"7" THEN 350
400 REM HYPPÄÄ VALITTUUN OHJELMAKOHTAAN
410 ON VAL(A$) GOTO 430,870,1110,1570,2000,
2140,2360
420 REM UUSI TILI
430 CLS
440 LOCATE 11,3
450 PRINT " TILIKIRJA "
460 LOCATE 12,5
470 PRINT "Uusi tili"
480 PRINT:PRINT
490 K=1
500 REM ETSII SEURAAVAN VAPAAN TILIN
510 FOR M=1 TO 3
520 IF NM$(M)="" THEN J=M:M=3:NEXT M:GOTO
570
530 NEXT M
540 PRINT "Tilitiedosto on täynnä."
550 GOSUB 2620
560 GOTO 210
570 INPUT "Tilin nimi";NM$(J)
580 PRINT
590 INPUT "Viimeisin tiedossa oleva saldo";
BA(J)
600 PRINT

```



```

610 PRINT "Uudet tilitiedot...":PRINT
620 INPUT "Tilinumero";CN(J,K)
630 PRINT
640 INPUT "Maksun saaja";P$(J,K)
650 PRINT
660 INPUT "Markkamäärä";AM(J,K)
670 PRINT
680 IF AM(J,K)=0 THEN 740
690 BA(J)=BA(J)-AM(J,K)
700 BF(J,K)=BA(J)
710 K=K+1
720 IF K=100 THEN BEEP:PRINT "VAROITUS:
Viimeinen lisäys tähän tiedostoon
- tallenna tiedot nauhalleja aloita uusi
tiedosto."
730 GOTO 620
740 PRINT:PRINT
750 PRINT "Tulotiedot...":PRINT:PRINT
760 CN(J,K)=-1:PRINT
770 INPUT "Mistä saatu";P$(J,K)
780 PRINT
790 INPUT "Markkamäärä";AM(J,K)
800 IF AM(J,K)=0 THEN 210
810 BA(J)=BA(J)+AM(J,K)
820 BF(J,K)=BA(J)
830 K=K+1
840 IF K=100 THEN BEEP:PRINT "VAROITUS:
Viimeinen lisäys tähän tiedostoon
- tallenna tiedot nauhalleja aloita uus
i tiedosto."
850 GOTO 760
860 REM TILIN PÄIVITYS
870 CLS
880 LOCATE 11,3
890 PRINT " TILIKIRJA "
900 LOCATE 10,5
910 PRINT "Tilin päivitys"
920 PRINT:PRINT
930 INPUT "Halutun tilin nimi";T$
940 PRINT
950 REM ETSII VALITUN TIEDOSTON
960 FOR M=1 TO 3
970 IF T$=NM$(M) THEN J=M:M=3:NEXT M:GOTO
1030

```

```

980 NEXT M
990 PRINT "Tiliä nimeltä ";T$;" ei löydy."
1000 GOSUB 2620
1010 GOTO 210
1020 REM ETSII VIIMEISEN TIETUEEN
1030 FOR M=1 TO 100
1040 IF AM(J,M)=0 THEN K=M:M=100
1050 NEXT M
1060 PRINT:PRINT
1070 PRINT "Uudet tilítiedot..."
1080 PRINT
1090 GOTO 620
1100 REM TILITIE TOJEN TULOSTUS
1110 CLS
1120 LOCATE 11,3
1130 PRINT " TILIKIRJA "
1140 LOCATE 6,5
1150 PRINT "Tilitietojen tulostus"
1160 PRINT:PRINT
1170 J=1:K=1:F=0
1180 INPUT "Tilin nimi tai KAIKKI";T$
1190 PRINT:PRINT
1200 REM OTTAA HUOMIOON KAIKKI TILIT
1210 IF T$="KAIKKI" OR T$="Kaikki" OR
   T$="kaikki" THEN F=1:GOTO 1310
1220 REM ETSII VALITUN TIEDOSTON
1230 FOR M=1 TO 3
1240 IF T$=NM$(M) THEN J=M:M=3:NEXT M:GOTO
1330
1250 NEXT M
1260 BEEP
1270 PRINT "Tiliä nimeltä ";T$;" ei löydy."
1280 GOSUB 2620
1290 GOTO 210
1300 REM TULOSTAA KOLMEN TILIN TIEDOT
1310 FOR J=1 TO 3
1320 IF AM(J,1)=0 THEN 1530
1330 PRINT "Tilin nimi: ";NM$(J)
1340 PRINT:PRINT
1350 PRINT "Tilinumero.";TAB(16);"Maksun
   saaja"
1360 PRINT " Markkamäärä";TAB(18);"Saldo"
1370 PRINT
1380 FOR K=1 TO 100

```

```

1390 IF AM(J,K)=0 THEN K=100:GOTO 1500
1400 IF CN(J,K)<>-1 THEN 1430
1410 PRINT "Tulot";
1420 GOTO 1440
1430 PRINT CN(J,K);
1440 PRINT TAB(16);P$(J,K)
1450 PRINT USING D$;AM(J,K);
1460 PRINT TAB(17);
1470 PRINT USING D$;BF(J,K)
1480 FOR DE=1 TO 100:NEXT DE
1490 GOSUB 2650
1500 NEXT K
1510 PRINT:PRINT
1520 IF F<>1 THEN 1540
1530 NEXT J
1540 GOSUB 2620:GOSUB 2620
1550 GOTO 210
1560 REM TILITIETOJEN KIRJOITUS
1570 CLS
1580 LOCATE 11,3
1590 PRINT " TILIKIRJA "
1600 LOCATE 6,5
1610 PRINT "Tilitietojen kirjoitus"
1620 PRINT:PRINT
1630 J=1:K=1:F=0
1640 INPUT "Tilin nimi tai KAIKKI";T$
1650 PRINT:PRINT
1660 IF T$="KAIKKI" OR T$="Kaikki" OR
T$="kaikki" THEN F=1:GOTO 1740
1670 FOR M=1 TO 3
1680 IF T$=NM$(M) THEN J=M:M=3:NEXT M:GOTO
1760
1690 NEXT M
1700 BEEP
1710 PRINT "Tiliä nimeltä ";T$;" ei lyödy."
1720 GOSUB 2620
1730 GOTO 210
1740 FOR J=1 TO 3
1750 IF AM(J,1)=0 THEN 1970
1760 LPRINT "Tilin nimi: ";NM$(J)
1770 LPRINT:LPRINT
1780 LPRINT "Tilinumero.";
1790 LPRINT TAB(19);"Maksun saaja";
1800 LPRINT TAB(38);"Markkamäärä";

```

```

1810 LPRINT TAB(62);"Saldo"
1820 LPRINT
1830 FOR K=1 TO 100
1840 IF AM(J,K)=0 THEN K=100:GOTO 1940
1850 IF CN(J,K)<>-1 THEN 1880
1860 LPRINT "Tulot";
1870 GOTO 1890
1880 LPRINT CN(J,K);
1890 LPRINT TAB(18);P$(J,K);
1900 IF CN(J,K)=-1 THEN LPRINT TAB(46);
ELSE LPRINT TAB(36);
1910 LPRINT USING D$;AM(J,K);
1920 LPRINT TAB(60);
1930 LPRINT USING D$;BF(J,K)
1940 NEXT K
1950 LPRINT:LPRINT
1960 IF F<>1 THEN 1980
1970 NEXT J
1980 GOTO 210
1990 REM TULOSTAA SALDON
2000 CLS
2010 LOCATE 11,3
2020 PRINT " TILIKIRJA "
2030 LOCATE 9,5
2040 PRINT "Nykyinen saldo"
2050 PRINT:PRINT
2060 FOR J=1 TO 3
2070 PRINT NM$(J);": ";
2080 PRINT USING D$;BA(J)
2090 PRINT
2100 NEXT J
2110 GOSUB 2620:GOSUB 2620
2120 GOTO 210
2130 REM LATAA TIEDOSTON NAUHALTA
2140 CLS
2150 LOCATE 11,3
2160 PRINT " TILIKIRJA "
2170 LOCATE 9,5
2180 PRINT "Lataus nauhalta"
2190 PRINT:PRINT
2200 LINE INPUT "Paina enter, kun kassetti
on valmiina.";NU$
2210 PRINT:PRINT
2220 PRINT "Lataus..."


```

```
2230 OPEN "KAS:TILI" FOR INPUT AS 1
2240 FOR J=1 TO 3
2250 INPUT#1,NM$(J)
2260 INPUT#1,BA(J)
2270 FOR K=1 TO 100
2280 INPUT#1,CN(J,K)
2290 INPUT#1,P$(J,K)
2300 INPUT#1,AM(J,K)
2310 INPUT#1,BF(J,K)
2320 NEXT K,J
2330 CLOSE
2340 GOTO 210
2350 REM TIEDOSTON TALLENUS NAUHALLE
2360 CLS
2370 LOCATE 11,3
2380 PRINT " TILIKIRJA "
2390 LOCATE 7,5
2400 PRINT "Tallennus nauhalle"
2410 PRINT:PRINT
2420 LINE INPUT "Paina enter, kun kassetti
on valmiina.";NU$
2430 PRINT:PRINT
2440 PRINT "Tallennus..."
2450 REM ODOTTAA TALLENNUKSEN ALKUA
2460 MOTOR ON
2470 GOSUB 2620:GOSUB 2620
2480 MOTOR OFF
2490 OPEN "KAS:TILI" FOR OUTPUT AS 1
2500 FOR J=1 TO 3
2510 PRINT#1,NM$(J)
2520 PRINT#1,BA(J)
2530 FOR K=1 TO 100
2540 PRINT#1,CN(J,K)
2550 PRINT#1,P$(J,K)
2560 PRINT#1,AM(J,K)
2570 PRINT#1,BF(J,K)
2580 NEXT K,J
2590 CLOSE
2600 GOTO 210
2610 REM VIIIVE
2620 FOR DE=1 TO 2500:NEXT DE
2630 RETURN
2640 REM ODOTTAA NÄPPÄIMEN PAINALLUSTA
2650 Z$=INKEY$
```

120

```
2660 IF Z$="" THEN 2690
2670 Z$=INKEY$
2680 IF Z$="" THEN 2670
2690 RETURN
```

PIKARI



Ajaudut kohtia Kalmarin temppelin raunioita. Raunioiden takana kohoaa suunnattoman arvokas kultainen pikari, jota vartioi ihmissyöjähirviö nimeltä Gork, joka on juuri nyt kovin nälkäinen!

Näihinkin raunioihin, kuten kaikkiin hyviin raunioihin vaikuttaa taikalumous, joka kuljettaa sinut ulos raunioista törmätessäsi raunion seiniin.

Yö on pimeä ja myrskyinen. Vain satunnaiset salamanvälähdykset opastavat sinua, joten tarvitset hyvän muistin ja nopeat refleksit saadaksesi pikarin.

Voit liikkua kohdistimen ohjausnäppäimillä.

Tietoja ohjelmasta

Rivin n:o

170-180	Asettaa äänen tempon ja nuottien keston sekä varaa tilaa kuvaruututaulukolle.
210-230	Odottaa tietoa näppäimistöltä.
250-270	Liike ylös
290-340	Tarkistaa törmäyksen.
360-280	Liike alas
420-450	Liike oikealle
470-500	Liike vasemmalle
550-700	Alkuarvot ja vaikeusaste
720-900	Määrittelee spritet.
920-990	Piirtää rauniot.
1000-1040	Sijoittaa pikarin.
1080-1220	Liikuttaa hirviötä.
1240-1290	Salamanvälähdykset
1300-1340	Onnistuit.
1350-1380	Epäonnistuit.
1400-1430	Uusi peli?

Ohjelma: PIKA-JS

```

100 REM *****
110 REM * *
120 REM * PIKARI *
130 REM * *
140 REM *****
150 REM
160 REM MUSIIKIN TEMPO JA NUOTIN PITUUS
170 PLAY "T255;L64"
180 DIM M(24,32)
190 GOTO 550
200 REM TUTKII NÄPPÄIMISTÖÄ
210 K$=INKEY$
220 IF K$="" THEN 530
230 KY=ASC(K$)
240 REM YLÖS
250 IF KY<>30 THEN 360
260 PR=PR-1
270 IF PR<1 THEN PR=1
280 REM CHECK FOR IMPACT

```



```
290 IF M(PR,PC)=0 THEN 510
300 IF M(PR,PC)=3 THEN 1300
310 PLAY "04 C"
320 PR=23
330 PUT SPRITE 1,((PC-1)*8,(PR-1)*8),4,1
340 GOTO 530
350 REM ALAS
360 IF KY<>31 THEN 420
370 PR=PR+1
380 IF PR>24 THEN PR=24
390 REM TARKISTAA UUDELLEEN
400 GOTO 290
410 REM OIKEALLE
420 IF KY<>28 THEN 470
430 PC=PC+1
440 IF PC>31 THEN PC=31
450 GOTO 290
460 REM VASEMMALLE
470 IF KY<>29 THEN 510
480 PC=PC-1
490 IF PC<3 THEN PC=3
500 GOTO 290
510 PUT SPRITE 1,((PC-1)*8,(PR-1)*8),4,1
520 PLAY "05 D"
530 RETURN
540 REM ALKUASETUKSET
550 R=RND(-TIME)
560 SCREEN 0
570 COLOR 1,15
580 CLS
590 LOCATE 12,11
600 PRINT " P I K A R I "
610 PR=23
620 PC=16
630 MR=4
640 MC=31
650 LOCATE 0,21
660 INPUT "Vaikeusaste (1 - 10)";SK
670 IF SK<1 OR SK>10 THEN BEEP:PRINT:GOTO
660
680 SCREEN 1
690 COLOR 1,15,15
700 CLS
710 REM SPRITE-MÄÄRITTELY
720 FOR K=1 TO 3
```

```

730 P$=""
740 FOR J=0 TO 7
750 READ D$
760 P$=P$+CHR$(VAL("&H"+D$))
770 NEXT J
780 SPRITE$(K)=P$
790 NEXT K
800 DATA 38,44,38,7C,BA,38,44,C6
810 DATA 41,41,41,3E,1C,08,08,1C
820 DATA 1C,3E,6F,FF,77,07,7E,3C
830 FOR J=1 TO 8
840 READ D$
850 E1$=E1$+CHR$(VAL("&H"+D$))
860 READ D$
870 E2$=E2$+CHR$(VAL("&H"+D$))
880 NEXT J
890 DATA 1C,38,3E,7C,6F,F6,FF,FF
900 DATA 77,EE,07,E0,7E,7E,3C,3C
910 REM RAUNIOT
920 FOR J=1 TO 22
930 FOR K=3 TO 32
940 IF RND(1)<.9-SK/40 THEN 990
950 IF J=2 AND K=16 THEN 990
960 M(J,K)=1
970 LOCATE K-3,J-1:COLOR 1
980 PRINT CHR$(219);
990 NEXT K,J
1000 PUT SPRITE 2,(120,8),11,2
1010 M(2,16)=3
1020 LOCATE 13,2:COLOR 15
1030 PRINT CHR$(32)
1040 M(3,16)=0
1050 PUT SPRITE 1,((PC-1)*8,(PR-1)*8),4,1
1060 GOTO 1200
1070 GOSUB 210
1080 IF RND(1)>.03+SK/100 THEN 1240
1090 IF MC<=PC THEN 1120
1100 MC=MC-1
1110 SPRITE$(3)=E1$
1120 IF MR<PR THEN 1140
1130 MR=MR-1
1140 IF MR>=PR THEN 1160
1150 MR=MR+1
1160 IF MC>=PC THEN 1190

```

```
1170 SPRITE$(3)=E2$
1180 MC=MC+1
1190 PUT SPRITE 3,((MC-1)*8,(MR-1)*8),6,3
1200 PLAY "Q3D"
1210 REM JAIT KIINNI
1220 IF MR=PR AND MC=PC THEN 1350
1230 REM SALAMOINTI
1240 IF RND(1)<.95+SK/210 THEN 1070
1250 COLOR 1,15,15
1260 FOR DE=1 TO 50
1270 NEXT DE
1280 COLOR 1,1,1
1290 GOTO 1070
1300 PLAY "T240;03CDEFGAB04CDEFGAB05CDEFGAB"
1310 COLOR 1,15,15
1320 LOCATE 7,11
1330 PRINT "PIKARI ON SINUN!!"
1340 GOTO 1390
1350 PLAY "T240;03BAGFEDC02BAGFEDC"
1360 COLOR 1,15,15
1370 LOCATE 4,11
1380 PRINT "HIRVIÖ SÖI SINUT!"
1390 LOCATE 0,24
1400 INPUT "UUSI PELI? (K/E)";Q$
1410 IF LEFT$(Q$,1)="K" OR LEFT$(Q$,1)="k"
  THEN RUN
1420 SCREEN 0
1430 END
```

1. The first part of the report is a general introduction to the subject of the study. It discusses the importance of the study and the objectives of the research.

2. The second part of the report is a detailed description of the methodology used in the study. It includes information about the sample size, the data collection methods, and the statistical analysis techniques.

3. The third part of the report is a presentation of the results of the study. It includes tables, figures, and text describing the findings.

4. The fourth part of the report is a discussion of the results and their implications. It discusses the strengths and limitations of the study and provides suggestions for future research.

5. The fifth part of the report is a conclusion. It summarizes the main findings of the study and provides a final statement on the importance of the research.

6. The sixth part of the report is a list of references. It includes all the sources used in the study.

7. The seventh part of the report is an appendix. It includes any additional information that is relevant to the study but is not included in the main text.

8. The eighth part of the report is a glossary. It includes definitions of key terms used in the study.

9. The ninth part of the report is a list of figures. It includes all the figures used in the study.

10. The tenth part of the report is a list of tables. It includes all the tables used in the study.

PELASTUS



Oletko koskaan kuvitellut olevasi haarniskapukuinen ritari? Nyt sinulla on siihen tilaisuus! Ilkeä noita on vanginnut kau-
niin neidon linnaansa, ja sinä olet neidon ainoa toive päästä
vapauteen.

Noidan linna on kuvaruudun oikeassa yläkulmassa. Suorit-
taaksesi pelastustehtävän sinun täytyy kulkea linnaan lähtö-
asemastasi, vasemmasta alakulmasta. Kuulostaa helpolta,
mutta sitä se ei ole! Noita on äkeissään, koska olet tuppautu-
nut hänen suunnitelmiensa tielle, ja heittelee valliltaan sinua

isoilla, vihreillä kivilohkareilla. Nämä taikalohkareet tappavat sinut, jos törmäät niihin tai jäät niiden alle.

Voit liikkua kohdistimen ohjausnäppäimillä.

Tietoja ohjelmasta

Rivin n:o	
200-250	Tulostaa otsikkosivun.
260-320	Kysyy vaikeusastetta.
400-470	Määrittelee ritari- ja linna-spriteja.
490-510	DATA-tiedot spriteja varten
530-610	Lohkareiden alkuasema
620-720	Asettaa loppukuvaruudun.
730-740	Alkuarvot ritarille
780-950	Liikuttaa ritaria.
880-900	Tarkistaa törmäyksen tai allejäännin.
970-1090	Lohkareen heitto
1100-1140	Pääohjelmasilmut
1150-1230	Onnistuit.
1260-1280	Tyhjentää näppäinpuskurin.
1290-1360	Uusi peli?
1380-1520	Törmäsit taikalohkareeseen.
1540-1810	Jäit lohkarin alle.
1820-1840	Viive
1850-1880	Soittaa melodian.
1890-1960	Kuvaruudun tyhjennys ja pelin lopetus

Ohjelma: PELA-JS

```

100 REM *****
110 REM *
120 REM * PELASTUS *
130 REM *
140 REM *****
150 REM
160 DEFINT A-Z
170 SCREEN 0
180 KEY OFF
190 REM OTSIKKO JA VAIKEUSASTE
200 COLOR 13,1
210 CLS
220 LOCATE 11,9
230 PRINT "P E L A S T U S"
```

```

240 GOSUB 1850
250 GOSUB 1820
260 LOCATE 0,20
270 COLOR 2
280 PRINT "Vaikkeusaste (1 - 5)?"
290 A$=INKEY$
300 IF A$="" THEN 290
310 IF A$<"1" OR A$>"5" THEN 290
320 SK=VAL(A$)-1
330 CLS
340 R=RND(-TIME)
350 SCREEN 2
360 COLOR 15,1,1
370 OPEN "GRP:" AS 1
380 ON SPRITE GOSUB 1150
390 REM SPRITE-MÄÄRITTELY
400 FOR K=1 TO 3
410 P$=""
420 FOR J=0 TO 7
430 READ D$
440 P$=P$+CHR$(VAL("&H"+D$))
450 NEXT J
460 SPRITE$(K)=P$
470 NEXT K
480 REM SPRITE DATA
490 DATA 08,08,1C,2A,2A,08,14,14
500 DATA AA,FF,FF,B6,FF,FF,B6,FE
510 DATA AA,FE,FE,DA,FE,DA,7E,7E
520 REM KIVENLOHKAREIDEN ALKUASEMA
530 FOR J=1 TO SK*40
540 R1=INT(RND(1)*32)
550 R2=INT(RND(1)*24)
560 PRESET (R1*8,R2*8)
570 COLOR 12
580 PRINT#1,CHR$(1);CHR$(&H43)
590 PRESET (R1*8,R2*8)
600 PRINT#1,CHR$(1);CHR$(&H42)
610 NEXT J
620 PRESET (240,0)
630 COLOR 1
640 PRINT#1,CHR$(219);CHR$(219)
650 PUT SPRITE 2,(240,0),10,2
660 PUT SPRITE 3,(248,0),10,3
670 PUT SPRITE 1,(16,184),4,1

```

```

680 FOR J=1 TO 4
690 PRESET (16,152+J*8)
700 COLOR 1
710 PRINT#1,CHR$(219)
720 NEXT J
730 MR=23:RM=23
740 MC=2:CM=2
750 SPRITE ON
760 GOTO 1100
770 REM RITARIN LIIKE
780 K$=INKEY$
790 IF K$="" THEN 950
800 K=ASC(K$)
810 MR=RM
820 MC=CM
830 IF K=28 THEN MC=MC+1
840 IF K=29 THEN MC=MC-1
850 IF K=30 THEN MR=MR-1
860 IF K=31 THEN MR=MR+1
870 IF MC<1 OR MC>31 OR MR<0 OR MR>23 THEN
950
880 X=POINT(MC*8+4,MR*8+4)
890 IF X=12 THEN 1380
900 IF MR=RM AND MC=CM THEN 950
910 PUT SPRITE 1,(MC*8,MR*8),4,1
920 PLAY "O5C"
930 RM=MR
940 CM=MC
950 RETURN
960 REM KIVEN HEITTAMINEN
970 R1=INT(RND(1)*12)+RM-6
980 IF R1>23 OR R1<0 THEN 970
990 R2=INT(RND(1)*12)+CM-6
1000 IF R2>31 OR R2<0 THEN 990
1010 IF R1=RM AND R2=CM THEN 1540
1020 IF (R1=0 AND R2=30) OR (R1=0 AND R2=31)
THEN 1090
1030 PLAY "O2D"
1040 PRESET (R2*8,R1*8)
1050 COLOR 12
1060 PRINT#1,CHR$(1);CHR$(&H42)
1070 PRESET (R2*8,R1*8)
1080 PRINT#1,CHR$(1);CHR$(&H43)
1090 RETURN

```



```
1100 FOR L=1 TO 10
1110 GOSUB 780
1120 NEXT L
1130 GOSUB 970
1140 GOTO 1100
1150 GOSUB 1850
1160 REM ONNISTUIT
1170 PRESET (85,88)
1180 COLOR 1
1190 PRINT#1,STRING$(10,219)
1200 PRESET (85,88)
1210 COLOR 15
1220 PRINT#1,"ONNISTUIT!"
1230 GOSUB 1820
1240 PRESET(16,184)
1250 COLOR 1
1260 FOR Z=1 TO 20
1270 A$=INKEY$
1280 NEXT Z
1290 PRINT#1,STRING$(16,219)
1300 PRESET (16,184)
1310 COLOR 15
1320 PRINT#1,"UUSI PELI? (K/E)"
1330 A$=INKEY$
1340 IF A$="" THEN 1330
1350 IF A$="K" OR A$="k" THEN RUN
1360 GOTO 1890
1370 REM TÖRMÄYS KIVENLOHKAREESEN
1380 FOR J=1 TO 10
1390 PUT SPRITE 1,(0,209)
1400 FOR DE=1 TO 30:NEXT DE
1410 PUT SPRITE 1,(CM*8,RM*8),1,4
1420 FOR DE=1 TO 30:NEXT DE
1430 NEXT J
1440 PRESET (70,88)
1450 COLOR 1
1460 PRINT#1,STRING$(15,219)
1470 PRESET (70,88)
1480 COLOR 15
1490 PRINT#1,"T Ö R M Ä S I T"
1500 PLAY "05GFEDC04BAGFED03BAGFEDC"
1510 GOSUB 1820
1520 GOTO 1240
1530 REM KIVENLOHKARE OSUU
```

```
1540 PRESET (80,88)
1550 COLOR 1
1560 PRINT#1,STRING$(13,219)
1570 PRESET (80,88)
1580 COLOR 15
1590 PRINT#1,"L Ä T S I S !"
1600 SOUND 0,0:SOUND 1,5
1610 SOUND 2,0:SOUND 3,13
1620 SOUND 4,255:SOUND 5,15
1630 SOUND 6,30:SOUND 7,0
1640 SOUND 8,16:SOUND 9,16
1650 SOUND 10,16:SOUND 11,0
1660 SOUND 12,5:SOUND 13,0
1670 FOR DE=1 TO 30:NEXT DE
1680 SOUND 12,56:SOUND 13,0
1690 FOR J=1 TO 10
1700 PRESET (CM*8,RM*8)
1710 COLOR 1
1720 PRINT#1,CHR$(219)
1730 PRESET(CM*8,RM*8)
1740 COLOR 12
1750 PRINT#1,CHR$(1);CHR$(&H42)
1760 PRESET (CM*8,RM*8)
1770 PRINT#1,CHR$(1);CHR$(&H43)
1780 NEXT J
1790 GOSUB 1020
1800 SOUND 7,56
1810 GOTO 1240
1820 FOR D=1 TO 300
1830 NEXT D
1840 RETURN
1850 PLAY "T120L403CEDF"
1860 GOSUB 1820
1870 PLAY "T240L64"
1880 RETURN
1890 SCREEN 0
1900 KEY ON
1910 COLOR 15,4,4
1920 CLS
1930 LOCATE 11,9
1940 PRINT "Kiitos pelistä!"
1950 LOCATE 0,22
1960 END
```

MUUKALAISRYNNÄKKÖ



Maailman pelastaminen on sinun käsissäsi. Joudut taisteluun loppumatonta muukalaisjoukkiota vastaan, joka pudottelee pommeja päällesi.

Kuinka peliä pelataan

Voit liikuttaa laser-alustasi vasemmalle ja oikealle käyttämällä näppäimiä O ja P. Tulitus tapahtuu välilyöntinäppäimellä. Jos vihollinen pääsee laskeutumaan aluksesi päälle tai siihen osuu pommi, menetät yhden kolmesta aluksestasi. Kun kaikki alukset on menetetty, on peli lopussa. Pelissä on kolme vaikeusastetta, jotka vaikuttavat laser-aseesi tulitusno-

peuteen. Ole varovainen, sillä kolmas vaikeusaste on todella vaikea!

Ohjelmointikikkoja

Ohjelma kostuu kolmesta aliohjelmasta, joita kutsutaan peräkkäin. Ohjelma on näin saatu melko nopeaksi tarvitsematta turvautua konekieleen. On hämmästyttävää miten paljon tietokone voi tehdä pelkästään BASICin avulla.

Tietoja ohjelmasta

Rivin n:o	
190	Äänen tempon ja nuotin keston asetus
220-240	Otsikko
250-270	Vaikeusaste
300-460	Liikuttaa alustasi ja tulittaa.
480-530	Liikuttaa muukalaisia alaspäin.
550-700	Liikuttaa pommeja.
720-790	Pääohjelma, joka kutsuu aliohjelmia.
850-930	Määrittelee tarvittavat spritet.
950-980	DATA-tiedot spriteja varten
1040-1240	Piirtää pelin taustan.
1250-1260	Aliohjelma laserin lataamiseksi
1280-1620	Sprite-törmäys
1290-1440	Muukalaiset osuivat.
1450-1600	Sinä osuit.
1640-1690	Alukset tuhottu
1720-1870	Laskee ja tulostaa parhaat tulokset.
1880-2030	Uusi peli?
2050-2490	Ääniefektit laser-aseelle, räjähdykselle ja pommin putoamiselle

Ohjelma: MURY-JS

```

100 REM *****
110 REM * *
120 REM * MUUKALAISSRYNNÄKKÖ *
130 REM * *
140 REM *****
150 REM
160 DEFINT A-L,N-Y
170 KEY OFF
180 REM RYTMII JA NUOTIN KESTO
190 PLAY "T240 L64"
```

```

200 SCREEN 0
210 COLOR 13,1
220 CLS
230 LOCATE 8,7
240 PRINT "MUUKALAISSRYNNÄKKÖ"
250 LOCATE 0,18
260 INPUT "VAIKEUSASTE (1 - 3)";SK
270 ON INTERVAL=SK*20 GOSUB 1250
280 GOTO 800
290 REM ALUKSEN LIIKUTTAMINEN
300 K$=INKEY$
310 IF SK>2 THEN GOSUB 550
320 IF SK>1 THEN GOSUB 550
330 IF K$="" THEN 450
340 IF K$="O" AND B>1 THEN B=B-1
350 IF K$="P" AND B<31 THEN B=B+1
360 IF K$<>" " THEN 450
370 IF FF=1 THEN 450
380 GOSUB 2050
390 IF AR>22 THEN 450
400 FOR J=20 TO AR-1 STEP -1
410 PUT SPRITE 4,(B*8,J*8),11,4
420 NEXT J
430 PUT SPRITE 4,(0,209),,4
440 FF=1
450 PUT SPRITE 1,(B*8,184),4,1
460 RETURN
470 REM MUUKALAISTEN LIIKE
480 AR=AR+1
490 IF AR<23 THEN 520
500 AR=1
510 AC=INT(RND(1)*29)+1
520 PUT SPRITE 2,(AC*8,AR*8),12,2
530 RETURN
540 REM POMMIEN LIIKE
550 MR=MR+1
560 MC=MC+ZC
570 IF MC>31 OR MC<1 THEN 590
580 IF MR<24 THEN 690
590 PUT SPRITE 3,(0,208),,3
600 IF RND(1)>.5 THEN 700
610 IF AR>18 OR AR<1 THEN 700
620 MR=AR+2
630 MC=AC
640 ZC=0

```

```

650 IF RND(1)>.5 THEN 680
660 IF AR<3 THEN 680
670 ZC=.25*SGN(B-AC)
680 GOSUB 2420
690 PUT SPRITE 3,(MC*8,MR*8),5,3
700 RETURN
710 REM PAAOHJELMASILMUKKA
720 GOSUB 300
730 GOSUB 550
740 GOSUB 300
750 GOSUB 480
760 GOSUB 300
770 GOSUB 550
780 GOSUB 300
790 GOTO 720
800 R=RND(-TIME)
810 SCREEN 2,1
820 OPEN "GRP:" AS 1
830 ON SPRITE GOSUB 1280
840 REM SPRITE-MÄÄRITTELY
850 RESTORE
860 FOR K=1 TO 4
870 P$=""
880 FOR J=0 TO 7
890 READ D$
900 P$=P$+CHR$(VAL("&H"+D$))
910 NEXT J
920 SPRITE$(K)=P$
930 NEXT K
940 REM SPRITE DATA
950 DATA 18,18,FF,FF,FF,FF,FF,FF
960 DATA C3,FF,DB,FF,C3,BD,18,66
970 DATA 00,00,00,18,18,18,00,00
980 DATA 08,08,08,08,08,08,08,08
990 REM ALKUARVOT
1000 B=15:BS=3
1010 AR=1
1020 AC=INT(RND(1)*29)+1
1030 REM TAUSTAN PIIRTO
1040 COLOR 15,1,1
1050 CLS
1060 CIRCLE (90,22),13,11,,,1.3
1070 PAINT (90,22),11
1080 CIRCLE (90,22),16,6,5.5,3.85,.1

```

```

1090 CIRCLE (90,23),16,13,5.5,3.85,.1
1100 CIRCLE (190,33),6,6,,,1.3
1110 PAINT (190,33),6
1120 FOR J=1 TO 16
1130 R1=INT(RND(1)*240)+10
1140 R2=INT(RND(1)*150)+10
1150 CIRCLE (R1,R2),1,15,,,1.3
1160 NEXT J
1170 CIRCLE(50,70),16,9,,,1.3
1180 PAINT(50,70),9
1190 PRESET(30,0)
1200 COLOR 15
1210 PRINT#1,"ALUKSET: 3      TULOS: 0"
1220 SPRITE ON
1230 INTERVAL ON
1240 GOTO 720
1250 FF=0
1260 RETURN
1270 REM SPRITE-TÖRMÄYS
1280 SPRITE OFF
1290 IF AR=23 OR MR>22 THEN 1450
1300 IF J<AR-2 OR J>AR+2 THEN 1610
1310 GOSUB 2220
1320 SC=SC+10
1330 FOR J=2 TO 4
1340 PUT SPRITE J,(0,209),,J
1350 NEXT J
1360 PRESET(190,0)
1370 COLOR 1
1380 PRINT#1,STRING$(5,219)
1390 COLOR 15
1400 PRESET (190,0)
1410 PRINT#1,SC
1420 AR=24
1430 MR=24
1440 GOTO 1610
1450 BS=BS-1
1460 GOSUB 2220
1470 PUT SPRITE 1,(0,209),,1
1480 AR=1
1490 B=15
1500 AC=INT(RND(1)*29)+1
1510 PUT SPRITE 3,(0,209)
1520 PUT SPRITE 2,(0,209)

```

```

1530 PRESET (90,0)
1540 COLOR 1
1550 PRINT#1,STRING$(4,219)
1560 COLOR 15
1570 PRESET(90,0)
1580 PRINT#1,BS
1590 IF BS=0 THEN 1630
1600 FOR DE=1 TO 500:NEXT DE
1610 SPRITE ON
1620 RETURN
1630 REM PELIN LOPPU
1640 SOUND 7,56
1650 PLAY "05BAFEDC04BAFEDC03BAFEDC"
1660 PRESET(50,99)
1670 COLOR 4
1680 PRINT#1," P E L I      O H I "
1690 FOR DE=1 TO 2500:NEXT DE
1700 SCREEN 0
1710 COLOR 15,1
1720 PRINT "SAIT";SC;"PISTETTA":PRINT
1730 INPUT "MIKÄ ON NIMESI";NM$
1740 IF LEN(NM$)>15 THEN PRINT "NIMI ON
    LIIAN PITKÄ!":PRINT:GOTO 1730
1750 REM PARHAAT TULOKSET TAULUKKONA
1760 S(10)=SC
1770 N$(10)=NM$
1780 FOR J=10 TO 2 STEP-1
1790 IF S(J)<S(J-1) THEN 1820
1800 SWAP S(J),S(J-1)
1810 SWAP N$(J),N$(J-1)
1820 NEXT J
1830 CLS
1840 PRINT TAB(8);"PARAS TULOS":PRINT:PRINT
1850 FOR M=1 TO 9
1860 PRINT " ";M;" ";S(M);" ";N$(M)
1870 NEXT M
1880 LOCATE 2,19
1890 PRINT "UUSI PELI? PAINA 'P'"
1900 PRINT "  LOPETUS?   PAINA 'L'"
1910 PRINT
1920 K$=INKEY$:IF K$="" THEN 1920
1930 IF K$<>"P" AND K$<>"p" THEN 1980
1940 SC=0
1950 MR=24
1960 SCREEN 2,1

```




```
1970 GOTO 830
1980 IF K$<>"1" AND K$<>"L" THEN 1920
1990 SCREEN 0
2000 KEY ON
2010 COLOR 15,4,4
2020 CLS
2030 GOTO 2500
2040 REM 4ANITEHOSTEET
2050 SOUND 0,250
2060 SOUND 1,0
2070 SOUND 4,150
2080 SOUND 5,0
2090 SOUND 6,15
2100 SOUND 7,42
2110 SOUND 8,15
2120 SOUND 9,15
2130 SOUND 10,16
2140 SOUND 11,20
2150 SOUND 12,0
2160 SOUND 13,12
2170 FOR DE=1 TO 30:NEXT DE
2180 SOUND 8,0
2190 SOUND 9,0
2200 SOUND 10,0
2210 RETURN
2220 SOUND 0,0
2230 SOUND 1,5
2240 SOUND 2,0
2250 SOUND 3,13
2260 SOUND 4,255
2270 SOUND 5,15
2280 SOUND 6,30
2290 SOUND 7,0
2300 SOUND 8,16
2310 SOUND 9,16
2320 SOUND 10,16
2330 SOUND 11,0
2340 SOUND 12,5
2350 SOUND 13,0
2360 FOR DE=1 TO 30:NEXT DE
2370 SOUND 12,56
2380 SOUND 13,0
2390 SOUND 6,0
2400 FOR DE=1 TO 30:NEXT DE
```

140

```
2410 RETURN
2420 FOR DE=48 TO 56
2430 SOUND 7,254
2440 SOUND 8,15
2450 SOUND 0,DE
2460 SOUND 1,0
2470 NEXT DE
2480 SOUND 8,0
2490 RETURN
2500 END
```

URUT




Tässä on mahdollisuutesi kokeilla MSX-tietokoneesi musiikki-mahdollisuuksia sekä opetella pianonsoittoa ja tulla vaikkapa Mozartin kaltaiseksi mestariksi.

Ylärivin kirjaimia (alkaen Q:sta) painamalla saadaan 10 nuottia. Voit itse halutessasi laajentaa nuottivalikoimaa noudattamalla rivien 150–340 mukaista tapaa ja käyttämällä muita näppäinrivejä.

Ohjelma: URUT-JS

```
100 REM *****
110 REM *      *
120 REM * URUT *
130 REM *      *
140 REM *****
150 REM
160 PLAY "T120L4"
170 SCREEN 0
180 COLOR 1,15
190 CLS
200 LOCATE 3,8
210 PRINT "Soittaminen tapahtuu näppäimill
a      ...QWERTYUIOP..."
220 LOCATE 0,23
230 A$=INKEY$
240 IF A$="" THEN GOTO 230
250 IF A$="Q" THEN PLAY "O2F"
260 IF A$="W" THEN PLAY "O2G"
270 IF A$="E" THEN PLAY "O2A"
280 IF A$="R" THEN PLAY "O2B"
290 IF A$="T" THEN PLAY "O3C"
300 IF A$="Y" THEN PLAY "O3D"
310 IF A$="U" THEN PLAY "O3E"
320 IF A$="I" THEN PLAY "O3F"
330 IF A$="O" THEN PLAY "O3G"
340 IF A$="P" THEN PLAY "O3A"
350 GOTO 230
```

PAKOILU



Tässä pelissä neljä ilkeän näköistä, vihreää kummitusta ajaa sinua takaa ja yrittää tuhota sinut. Kuten tiedämme, kummitukset voivat kulkea helposti seinien läpi, mutta sinä et siihen pysty.

Voit kulkea sokkelossa käyttäen kohdistimen ohjausnäppäimiä. Pisteitä saadaan pysymällä hengissä. Sinun täytyy käyttää järkeäsi ja salamannopeita refleksejä liikkuessasi, sillä tämä on vaativa peli. On saavutus saada yli 1 500 pistettä.

Tietoja ohjelmasta

Rivin n:o	
180-210	Kuvaruudun alkuasetus
220-270	Otsikko
280	Avaa tiedoston hienografiikkaa varten.
300	Asettaa äänen tempon ja nuotin keston.
330-350	Odottaa ohjetta näppäimistöltä.
360-420	Uusi asema
430	Tarkistaa osuman seinään tai kummitukseen.
460-500	Liike
530	Valitsee kummituksen satunnaisesti.
550-620	Liikuttaa kummitusta.
690-760	Määrittelee spritet.
780-790	DATA-tiedot spriteja varten
810-1150	Piirtää sokkelon seinät.
1170-1290	Alkuasemat ja paikat spriteja varten
1310-1410	Pääohjelma - kutsuu aliohjelmat ja lisää tulosta.
1430-1480	Kummitukset saivat sinut.
1510-1700	Laskee ja tulostaa parhaat pisteet.
1710-1780	Uusi peli?
1800-1820	Viive
1830-1850	Kuvaruutu alkutilaan

Ohjelma: PAKO-JS

```

100 REM *****
110 REM *           *
120 REM * PAKOILU *
130 REM *           *
140 REM *****
150 REM
160 REM KUVARUUDUN ALKUARVOT JNE.
170 DEFINT A-Z
180 COLOR 1,15
190 SCREEN 0
200 CLS
210 KEY OFF
220 LOCATE 13,9
230 PRINT "PAKOILU"
240 LOCATE 0,21
250 PRINT "Peli alkaa, kun painat jotain näppäintä"
```

```

260 A$=INKEY$
270 IF A$="" THEN 260
280 OPEN "GRP: " AS 1
290 REM ASETTAA RYTMIN JA NUOTIN KESTON
300 PLAY "T255L64"
310 GOTO 640
320 REM LIIKE
330 K$=INKEY$
340 IF K$="" THEN GOSUB 1800:GOTO 510
350 K=ASC(K$)
360 MR=RM
370 MC=CM
380 IF K=28 THEN MC=MC+1
390 IF K=29 THEN MC=MC-1
400 IF K=30 THEN MR=MR-1
410 IF K=31 THEN MR=MR+1
420 IF MC<3 OR MC>31 OR MR<1 OR MR>24 THEN
510
430 X=POINT(MC*8+4,MR*8+4)
440 IF X=12 THEN 1430
450 IF X=4 THEN 510
460 IF MR=RM AND MC=CM THEN 510
470 PLAY "O5C"
480 PUT SPRITE 1,(MC*8,MR*8),13,1
490 RM=MR
500 CM=MC
510 RETURN
520 REM KUMMITUKSIEN LIIKE
530 R=INT(RND(1)*4)+1
540 IF R=5 THEN R=1
550 IF NS(R,1)=MR THEN 590
560 NS(R,1)=NS(R,1)+SGN(MR-NS(R,1))
570 PUT SPRITE R+1,(NS(R,2)*8,NS(R,1)*8),
12,2
580 PLAY "O3D"
590 IF NS(R,2)=MC THEN 620
600 NS(R,2)=NS(R,2)+SGN(MC-NS(R,2))
610 PUT SPRITE R+1,(NS(R,2)*8,NS(R,1)*8),
12,2
620 IF MR=NS(R,1) AND MC=NS(R,2) THEN 1430
630 RETURN
640 CLS
650 COLOR 15,1,1
660 SCREEN 2
670 RR=RND(-TIME)

```

```

680 REM SPRITE-MÄÄRITTELY
690 FOR K=1 TO 2
700 P$=""
710 FOR J=0 TO 7
720 READ D$
730 P$=P$+CHR$(VAL("&H"+D$))
740 NEXT J
750 SPRITE$(K)=P$
760 NEXT K
770 REM SPRITE DATA
780 DATA 38,44,38,6C,AA,38,44,C6
790 DATA 38,7C,BA,FF,FF,C6,38,38
800 REM SOKKELON PIIRTO
810 FOR J=8 TO 20 STEP 6
820 PRESET (24,J*8)
830 COLOR 4
840 PRINT#1,STRING$(10,219)
850 PRESET (112,J*8)
860 PRINT#1,STRING$(9,219)
870 PRESET (192,J*8)
880 PRINT#1,STRING$(7,219)
890 NEXT J
900 FOR J=4 TO 16 STEP 6
910 PRESET (24,J*8)
920 PRINT#1,STRING$(15,219)
930 PRESET (152,J*8)
940 PRINT#1,STRING$(12,219)
950 NEXT J
960 FOR J=6 TO 18 STEP 6
970 PRESET (24,J*8)
980 PRINT#1,STRING$(6,219)
990 PRESET (80,J*8)
1000 PRINT#1,STRING$(16,219)
1010 PRESET (216,J*8)
1020 PRINT#1,STRING$(4,219)
1030 NEXT J
1040 PRESET (24,176)
1050 PRINT#1,STRING$(28,219)
1060 PRESET (24,16)
1070 PRINT#1,STRING$(28,219)
1080 FOR J=2 TO 22
1090 PRESET (16,J*8)
1100 PRINT#1,CHR$(219)
1110 NEXT J

```



```

1120 FOR J=2 TO 22
1130 PRESET (248,J*8)
1140 PRINT#1,CHR$(219)
1150 NEXT J
1160 REM ALKUASEMAT
1170 NS(1,1)=3:NS(1,2)=5
1180 NS(2,1)=3:NS(2,2)=30
1190 NS(3,1)=21:NS(3,2)=5
1200 NS(4,1)=21:NS(4,2)=30
1210 PRESET (80,0)
1220 COLOR 15
1230 PRINT#1,"TULOS ="
1240 FOR J=1 TO 4
1250 PUT SPRITE J+1,(NS(J,2)*8,NS(J,1)*8) 12,2
1260 NEXT J
1270 MR=11:MC=18
1280 RM=11:CM=18
1290 PUT SPRITE 1,(MC*8,MR*8),13,1
1300 REM PÄÄOHJELMASILMUKKA
1310 GOSUB 330
1320 GOSUB 530
1330 GOSUB 330
1340 SC=SC+10
1350 PRESET (144,0)
1360 COLOR 1
1370 PRINT#1,STRING$(8,219)
1380 PRESET (144,0)
1390 COLOR 15
1400 PRINT#1,SC
1410 GOTO 1310
1420 REM JÄIT KIINNI
1430 SOUND 7,56
1440 PLAY "O5BAFEDC04BAFEDC03BAFEDC"
1450 PRESET(50,90)
1460 COLOR 3
1470 PRINT#1," P E L I   O H I "
1480 FOR DE=1 TO 2500:NEXT DE
1490 SCREEN 0
1500 COLOR 15,1
1510 PRINT "SAIT";SC;"PISTETTA":PRINT
1520 INPUT "MIKÄ ON NIMESI";NM$
1530 IF LEN(NM$)>15 THEN PRINT "NIMI ON
LIIAN PITKÄ!":PRINT:GOTO 1520
1540 REM PARHAAT TULOKSET TAULUKKONA

```

```
1550 S(10)=SC
1560 N$(10)=NM$
1570 FOR J=10 TO 2 STEP-1
1580 IF S(J)<S(J-1) THEN 1610
1590 SWAP S(J),S(J-1)
1600 SWAP N$(J),N$(J-1)
1610 NEXT J
1620 CLS
1630 PRINT TAB(8);"PARAS TULOS":PRINT:PRINT
1640 FOR M=1 TO 9
1650 PRINT " ";M;" ";S(M);" ";N$(M)
1660 NEXT M
1670 LOCATE 2,19
1680 PRINT "UUSI PELI? PAINA 'P'"
1690 PRINT " LOPETUS? PAINA 'L'"
1700 PRINT
1710 K$=INKEY$:IF K$="" THEN 1710
1720 PLAY "OSC"
1730 IF K$<>"P" AND K$<>"L" THEN 1770
1740 SC=0
1750 SCREEN 2
1760 GOTO 810
1770 IF K$="I" OR K$="L" THEN 1830
1780 GOTO 1710
1790 REM UIIVE
1800 FOR DE=1 TO 25
1810 NEXT DE
1820 RETURN
1830 COLOR 15,4,4
1840 KEY ON
1850 CLS
1860 END
```


MSX



Vince Apps

MSX on moniin kotitietokoneisiin sopiva standardi. Tämä pelikokoelma on kirjoitettu tyydyttämään MSX-standardin synnyttämää kysyntää.

Voit testata reaktioitasi ja sorminäppäryyttäsi ARCADE-peleillä. Pääset mukaan nokkeluutta ja muistia kehittäviin SEIKKAILUIHIN. Ratkaiset ARVOITUKSIA. Pelien joukosta löydät myös ohjelmointitaitojasi parantavia rutiineja.